TeC 教科書

Ver. 3.2.2

徳山工業高等専門学校
情報電子工学科

目 次

第1章　はじめに

1.1　この科目で学ぶこと

現代，コンピュータと呼ばれているものはスーパーコンピュータと呼ばれる高価で高性能なものから，マイコンと呼ばれ炊飯器やエアコンに組み込まれている小型のものまで，全て同じ原理に基づき動作しています．

この原理は，1946 年に米国の数学者フォン・ノイマン (Von Neumann) が提案したと言われています．現在のコンピュータの，ほぼ，全てのものは，ノイマンの提案した同じ原理を使用しており「ノイマン型コンピュータ」と呼ばれます．

「ノイマン型コンピュータ」が出現して既に約 70 年の時間が経過しましたが，未だにこれを上回る「自動機械」の構成方法は発明されていません．だから，パソコンのような一般の人がコンピュータだと考えている装置だけでなく，炊飯器やエアコンの制御装置のようなコンピュータらしくない装置まで，「自動機械」が必要なところには全て「ノイマン型コンピュータ」が使用されているのです．恐らく，あと数十年は,「ノイマン型コンピュータ」の時代が続くでしょう．

この教科書では，教育用コンピュータ (TeC) を用いて，この「ノイマン型コンピュータ」の動作原理を，しっかり勉強できるようになっています．ここでしっかり勉強しておけば，将来，どんな新型のコンピュータに出会ったとしても，恐れることはありません．所詮,「ノイマン型コンピュータ」の一種ですから，皆さんはその正体を簡単に見抜くことができるはずです．

「ノイマン型コンピュータ」の原理をきちんと理解しておくことは，皆さんにとって，寿命の長いエンジニアになるための大切なステップとなります．しっかり，がんばって下さい．

1.2　勉強の進め方

この教科書は，高専や工業高校の 1 年生と 2 年生が，教育用コンピュータ TeC を教材に，ノイマン型コンピュータの基本を学ぶことを想定して作ってあります．1 章から 5 章の途中までを 1 年で学び，残りを 2 年生 (半年〜1 年間) で学びます．

1 年生では，(1) コンピュータの内部で情報を表現する方法，(2)TeC の組み立て，(3)TeC の基本的なプログラミングを学びます．2 年生では，TeC を用いた高度なプログラミングを学びます．

1.3　教材用コンピュータ

私達の身近にあるパーソナルコンピュータ (パソコン) や携帯電話のようなコンピュータシステムは，高度で複雑すぎて「ノイマン型コンピュータ」の原理を学ぶための教材としては適していません．

原理を学ぶのに適した単純で小さなコンピュータ (マイコン) を，専用に開発しました．このマイコンは TeC(Tokuyama Educational Computer： 徳山高専教育用コンピュータ) と呼ばれます．

TeC の特徴は単純で小さなことです．単純で小さなことで，次のようなメリットがあります．

単純　ノイマン型コンピュータの本質的な部分だけを勉強しやすい．現在のパソコン等は，勉強するには難しすぎる．

小型　自宅に持ち帰り宿題ができる．授業時間外でも，納得がいくまで色々と試してみることができる．

TeC には 2003 年から使用している TeC6 と，2011 年から使用を開始した TeC7 の二種類があります．本書では TeC7 を基本に解説しています．TeC6 と TeC7 で異なる場合は TeC6 の場合を追記するようにしています．

第2章　情報の表現

この章では，コンピュータ内部でどのように情報が表現されているのか，その情報をどのような回路で扱うことができるのか，簡単に紹介します．

2.1　コンピュータ内部の情報表現

人は，情報を音声や文字，絵等で表現することができます．コンピュータも表面的には音声や文字，絵を扱うことができますが，コンピュータの内部は電子回路で構成されているので，最終的には電圧や電流で情報を表現せざるを得ません．

$$\text{情報の表現} = \left(\begin{array}{l} \text{人 : 音声，文字，絵,...} \\ \text{コンピュータ : 電圧，電流} \end{array} \right)$$

2.1.1　電気を用いた情報の表現

電圧や電流で情報表現する方法には，いろいろなアイデアがあります．しかし，現在のコンピュータは，電圧が**ある/ない (ON/OFF)** か，電流が**流れている/流れていない (ON/OFF)** のような，二つの状態だけを用いて情報を表現する方法を使っています．(デジタル・コンピュータ)

コンピュータの情報表現
　電圧が**ある/ない**
　電流が**流れる/流れない**
　　↓
　ON と OFF の二つの状態だけを用いて
　情報を表現する．(回路が作りやすい)

次の例を見て下さい．電気の「ON/OFF」で「おおかみが来たか/来ていないか」のどちらの状態なのかを伝達できる「情報の表示装置」を実現することができます．電気の「ON/OFF」を用いて情報を表現することができました．

例 おおかみが来た情報表示装置
　ランプ点灯：おおかみが来た
　ランプ消灯：おおかみが来ていない



2.1.2　ビット

前の例では，ランプの「ON/OFF」を用いて「二つの状態のどちらなのか」を表しました．このような,「二つのどちらか」を表す情報が「情報の最小単位」になります．情報の最小単位のことを**「ビット (bit)」**と呼びます．

on/off のどちらか → 情報の最小単位 (ビット)

通常，ビットの値は「ON/OFF」ではなく,「1/0」で書きます．

$$\left(\begin{array}{rcl} \text{ON} & : & 1 \\ \text{OFF} & : & 0 \end{array} \right)$$

「おおかみが来た情報」は，ビットの値に次のように対応付けできます．

ビット値	おおかみが来た情報
0(off)	おおかみがきていない
1(on)	おおかみが来た !!!

2.1.3　より複雑な情報の表現

二つの状態では表現できない，もっと複雑な情報はどうやって表現したら良いでしょうか．

前の例で，やって来たおおかみの頭数により牧場の人が対応を変化させたい場合が考えられます．そのためには,「来たか/来なかったか」だけの情報だけでは十分ではなく，**「たくさん来たか」**を知らせる必要があります．それには，複数のビットを組み合わせて使用します．

以下に，複数のビットの組み合わせて表現した「拡張おおかみが来た情報」を示します．

ビット値	拡張おおかみが来た情報	
00	おおかみがきていない	平気
01	おおかみが 1 頭来た	戦う
10	おおかみが 2 頭来た	？
11	おおかみがたくさん来た	逃げる

例 拡張おおかみが来た情報表示装置



この例で見たように，2 ビットを用いれば 4 種類の情報を表現することができます．

一般に，n ビットを用いると 2^n 種類の情報を表現することができます．システム内で必要なビット数を決めて，それの組合せに意味付けをすれば，どんな情報だって表現できます．

ビット数	ビットの組合せ	組合せ数
1	0 1	2
2	00 01 10 11	4
3	000 001 010 011 100 101 110 111	8
…	…	
n		2^n

ビットだけでは情報の単位として小さすぎるので，4 ビットまとめたもの，8 ビットまとめたものにも名前があります．

「4 ビット」　＝　「1 ニブル」
「8 ビット」　＝　「1 バイト」

2.2　数値の表現

ビットの組合せをどのように意味付けするかは，前節の例のように「システムが扱う必要のある情報」により，毎回，約束すればよいのです．しかし，どのコンピュータでも同じ方法で意味付けされている情報もあります．それの一つが数値の表現方法です．

2.2.1　2 進数

コンピュータの内部では，数値を 2 進数で表すのが普通です．我々が普段使用している 10 進数と，2 進数の特徴比較を次に示します．

10 進数と 2 進数の比較

10 進数の特徴
(1) 0 〜 9 の 10 種類の数字を使用する．
(2) 1 桁毎に 10 倍の重みをもつ．

2 進数の特徴
(1) 0 と 1 の 2 種類の数字を使用する．
(2) 1 桁毎に 2 倍の重みをもつ．

コンピュータの内部では，ビットが情報の表現に使用されています．そこで，ビット値の 0/1 をそのまま 2 進数の 1 桁と考えれば数値が表現できます．

例えば，4 ビット用いると 0 〜 15 の数が次のように表現できます．

ビット 3 (b_3)	ビット 2 (b_2)	ビット 1 (b_1)	ビット 0 (b_0)	意味
0	0	0	0	0
0	0	0	1	1
0	0	1	0	2
0	0	1	1	3
0	1	0	0	4
0	1	0	1	5
0	1	1	0	6
0	1	1	1	7
1	0	0	0	8
1	0	0	1	9
1	0	1	0	10
1	0	1	1	11
1	1	0	0	12
1	1	0	1	13
1	1	1	0	14
1	1	1	1	15

一般に n ビットで 0 〜 $2^n - 1$ の範囲の数を表現することができます．

2.2.2　2 進数と 10 進数の相互変換

前の 4 ビットの例なら，2 進数と 10 進数の対応を暗記することが可能です．しかし，8 ビットの場合ならどうでしょう？

組合せは 256 もあり，とても暗記できそうにありません．対応を計算で求める必要があります．

1. 2 進から 10 進への変換

   2 進数の桁ごとの重みは，桁の番号を n とすると 2^n になります．

   $$\begin{array}{cccc} b_3 & b_2 & b_1 & b_0 \\ 2^3 = 8 & 2^2 = 4 & 2^1 = 2 & 2^0 = 1 \end{array}$$

   2 進数の数値は，その桁の重みと桁の値を掛け合わせたものの合計です．例えば 2 進数の 1010_2

は，2^3 の桁が 1，2^2 の桁が 0，2^1 の桁が 1，2^0 の桁が 0 ですから，次のように計算できます．

$$\begin{aligned}1010_2 &= 1\times 2^3 + 0\times 2^2 + 1\times 2^1 + 0\times 2^0\\ &= 1\times 8 + 0\times 4 + 1\times 2 + 0\times 1\\ &= 8 + 0 + 2 + 0\\ &= 10_{10}\end{aligned}$$

2. 10 進から 2 進への変換

次に，10 進数を 2 進数に変換する方法を考えます．ここで着目するのは桁の移動です．

10 進数では，値を 10 で割ると右に 1 桁移動します．2 進数では，2 で割ると右に 1 桁移動します．どちらの場合でも，割算をした時の余りは，最下位の桁からはみ出した数になります．

つまり，数値を 2 で割った時の余りは 2 進数を右に 1 桁移動したときはみ出してきた数を表しています．

そこで，2 で割る操作を繰り返しながらはみ出して来た数を記録すれば，もとの数を 2 進数で表したときの 0/1 の並びが分かります．

$$\begin{array}{rcl} & 12_{10} & = & 1100_2 \\ 1/2\downarrow & & & \\ & 6^{\cdots 0}_{10} & = & 0110^{\cdots 0}_{2} \\ 1/2\downarrow & & & \\ & 3^{\cdots 0}_{10} & = & 0011^{\cdots 0}_{2} \\ 1/2\downarrow & & & \\ & 1^{\cdots 1}_{10} & = & 0001^{\cdots 1}_{2} \\ 1/2\downarrow & & & \\ & 0^{\cdots 1}_{10} & = & 0000^{\cdots 1}_{2} \end{array}$$

計算方法と計算例

$$\begin{array}{l} 2)\ \underline{100} \\ 2)\ \ \underline{50}\cdots 0 \\ 2)\ \ \underline{25}\cdots 0 \\ 2)\ \ \underline{12}\cdots 1 \\ 2)\ \ \underline{6}\cdots 0 \\ 2)\ \ \underline{3}\cdots 0 \\ 2)\ \ \underline{1}\cdots 1 \\ \quad 0\cdots 1 \end{array}$$

余りを右から順に並べると 1100100_2

2.2.3 16 進数

2 進数は，桁数が多くなり書き表すのに不便です．そこで，2 進数 4 桁をまとめて 16 進数一桁で書き表します．9 より大きな数字が無いので，16 進数ではアルファベットを数字の代用にします．

2 進数	16 進数	10 進数
0000_2	0_{16}	0_{10}
0001_2	1_{16}	1_{10}
0010_2	2_{16}	2_{10}
0011_2	3_{16}	3_{10}
0100_2	4_{16}	4_{10}
0101_2	5_{16}	5_{10}
0110_2	6_{16}	6_{10}
0111_2	7_{16}	7_{10}
1000_2	8_{16}	8_{10}
1001_2	9_{16}	9_{10}
1010_2	A_{16}	10_{10}
1011_2	B_{16}	11_{10}
1100_2	C_{16}	12_{10}
1101_2	D_{16}	13_{10}
1110_2	E_{16}	14_{10}
1111_2	F_{16}	15_{10}

n 進数の表記

日常生活では，数値を書き表すときは「いつも 10 進数」で書きます．しかし，コンピュータの世界では 2 進数，10 進数，16 進数を場合により使い分けます．そのため，何進数で書いてあるのか分からなくて困ることがあります．

そこで，上の表のように数値の右に小さな字で何進数かを書き加えます．ときには，数字の代わりに「2 進数='b'」，「16 進数='H'」を加えることもあります．

例えば，「01100100_2 を 01100100_b」，「64_{16} を 64H」のように書きます．

問題

1. 上の「2 進数，16 進数，10 進数対応表」を暗記しなさい．
2. 10 進数の 16，50，100，127，130 を，2 進数 (8 桁)，16 進数 (2 桁) で書き表しなさい．
3. 2 進数の 00011100，00111000，11100000 を 16 進数 (2 桁)，10 進数で書き表しなさい．
4. 16 進数の 1F と AA を 2 進数 (8 桁) で書き表しなさい．また，10 進数で書き表しなさい．

2.3　負数の表現

拡張おおかみが来た情報表示装置では,「表示装置の二つのビット (二つのランプ) を，あのように読む」ことを約束しました．つぎに，数値の場合は，「n 個のビットを 2 進数として読む」ことを約束しました．

今度は，負の数が必要になりました．そこで，ビットの新しい読み方を約束します．この節で出てくるビットは「符号付き数値」を表しています．以下では,「符号付き数値を表すビット」をどのように読むかを説明します．

2.3.1　符号付き絶対値表現

使用できるビットのうち一つを符号を表すために使用します．これを「符号ビット」と呼ぶことにします．通常，符号ビットには最上位 (左端) のビットを使用します．

次の例のように 4 ビット使用して，-7 〜 $+7$ の範囲を表すことができます．この表現方法は分かりやすくて都合が良いのですが，実際に使われることはあまりありません．

符号付き絶対値表現 (4 ビット) の例	
-7	1111_2
-6	1110_2
-5	1101_2
...	...
-1	1001_2
-0	1000_2
$+0$	0000_2
$+1$	0001_2
...	...
$+5$	0101_2
$+6$	0110_2
$+7$	0111_2

2.3.2　補数表現

n 桁の b 進数でにおいて b^n から x を引いた数 y を x に対する「b の補数」と言います．n 桁の b 進数でにおいて $b^n - 1$ から x を引いた数 z を x に対する「(b-1) の補数」と言います．

$$y = b^n - x \quad (y \text{ は } x \text{ に対する } b \text{ の補数})$$
$$z = b^n - 1 - x \quad (z \text{ は } x \text{ に対する } (b-1) \text{ の補数})$$

例えば，10 進数の世界で次のような例があります．

2 桁の 10 進数で補数の例		
$b = 10$ 進数		
$n = 2$ 桁	100	
$b^n = 100$	-25	75 は 25 に対する 10 の補数
$x = 25$	75	
$b = 10$ 進数		
$n = 2$ 桁	99	
$b^n - 1 = 99$	-25	74 は 25 に対する 9 の補数
$x = 25$	74	

2 進数の世界では次のような例があります．

4 桁の 2 進数で補数の例		
$b = 2$ 進数		0110_2 は
$n = 4$ 桁	10000_2	1010_2 に
$b^n = 10000_2$	-1010_2	対する
$x = 1010_2$	0110_2	2 の補数
$b = 2$ 進数		0101_2 は
$n = 4$ 桁	1111_2	1010_2 に
$b^n - 1 = 1111_2$	-1010_2	対する
$x = 1010_2$	0101_2	1 の補数

2.3.3　1 の補数による負数の表現

上の**「4 桁の 2 進数で補数の例」**にあるように，$11...1_2$ からもとの数 (x) を引いた数を x に対する「1 の補数」と呼びます．次の表は 0 〜 7 に対する 1 の補数を計算したものです．

4 ビット 2 進数の 1 補数	
0	$1111_2 - 0000_2 = 1111_2$
1	$1111_2 - 0001_2 = 1110_2$
2	$1111_2 - 0010_2 = 1101_2$
3	$1111_2 - 0011_2 = 1100_2$
4	$1111_2 - 0100_2 = 1011_2$
5	$1111_2 - 0101_2 = 1010_2$
6	$1111_2 - 0110_2 = 1001_2$
7	$1111_2 - 0111_2 = 1000_2$

「0001_2 に対する 1 の補数を -0001_2 を表現するために使用する.」,「0010_2 に対する 1 の補数を -0010_2 を表現するために使用する.」つまり，**「1 の補数を負の数を表現するために使用する.」**と約束すれば，次の表のように -7 〜 $+7$ の範囲の数を表現できます.

1 の補数を用いた符号付き数値									
-7	1000_2	−	−	−	−	−	−	−	+
-6	1001_2	−	−	−	−	−	−	+	|
-5	1010_2	−	−	−	−	−	+	|	|
-4	1011_2	−	−	−	−	+	|	|	|
-3	1100_2	−	−	−	+	|	|	|	|
-2	1101_2	−	−	+	|	|	|	|	|
-1	1110_2	−	+	|	|	|	|	|	|
-0	1111_2	+	|	|	|	|	|	|	|
$+0$	0000_2	+	|	|	|	|	|	|	|
$+1$	0001_2	−	+	|	|	|	|	|	|
$+2$	0010_2	−	−	+	|	|	|	|	|
$+3$	0011_2	−	−	−	+	|	|	|	|
$+4$	0100_2	−	−	−	−	+	|	|	|
$+5$	0101_2	−	−	−	−	−	+	|	|
$+6$	0110_2	−	−	−	−	−	−	+	|
$+7$	0111_2	−	−	−	−	−	−	−	+

次のように 1 の補数 (z) を求める式を $x =$ の形に変形しても同じ形になるので，x と z は「互いに 1 の補数」です.

$$z = 2^n - 1 - x \quad (z \text{ は } x \text{ に対する 1 の補数})$$
$$x = 2^n - 1 - z \quad (x \text{ は } z \text{ に対する 1 の補数})$$

なお，表で $+7$ と -7 が結んであるのは,「互いに 1 の補数」であることを表すためです．また,「互いに 1 の補数」の 2 数を見比べるとビットの 0/1 が入れ替わった関係になっていることが分かります．1 の補数は引算だけではなく**「ビット反転」**でも計算できます.

1 の補数を使用した方法も，実際に使われることはあまりありません．コンピュータの内部で実際に使用されるのは，次に説明する 2 の補数による表現です.

2.3.4 2 の補数による負数の表現

前出の**「4 桁の 2 進数で補数の例」**にあるように，$100...0_2$ からもとの数 (x) を引いた数を x に対する「2 の補数」と呼びます．次の表は 0 〜 8 に対する 2 の補数を計算したものです．5 ビットで表現さている部分もありますが，四角で囲んだ 4 ビットに注目してください.

4 ビット 2 進数の 2 補数			
0	$1\boxed{0000}_2 - \boxed{0000}_2 =$	$1\boxed{0000}_2$
1	$1\boxed{0000}_2 - \boxed{0001}_2 =$	$\boxed{1111}_2$
2	$1\boxed{0000}_2 - \boxed{0010}_2 =$	$\boxed{1110}_2$
3	$1\boxed{0000}_2 - \boxed{0011}_2 =$	$\boxed{1101}_2$
4	$1\boxed{0000}_2 - \boxed{0100}_2 =$	$\boxed{1100}_2$
5	$1\boxed{0000}_2 - \boxed{0101}_2 =$	$\boxed{1011}_2$
6	$1\boxed{0000}_2 - \boxed{0110}_2 =$	$\boxed{1010}_2$
7	$1\boxed{0000}_2 - \boxed{0111}_2 =$	$\boxed{1001}_2$
8	$1\boxed{0000}_2 - \boxed{1000}_2 =$	$\boxed{1000}_2$

「0001_2 に対する 2 の補数を -0001_2 を表現するために使用する.」,「0010_2 に対する 2 の補数を -0010_2 を表現するために使用する.」つまり**「2 の補数を負の数を表すために使用する.」**と約束すれば，次の表のように -8 〜 $+7$ の範囲の数を表現できます.(ただし 1000_2 は -8 を表現することとします．4 ビットでは $+8$ を表現することはできません.)

2 の補数を用いた符号付き数値									
-8	1000_2								
-7	1001_2	−	−	−	−	−	−	−	+
-6	1010_2	−	−	−	−	−	−	+	|
-5	1011_2	−	−	−	−	−	+	|	|
-4	1100_2	−	−	−	−	+	|	|	|
-3	1101_2	−	−	−	+	|	|	|	|
-2	1110_2	−	−	+	|	|	|	|	|
-1	1111_2	−	+	|	|	|	|	|	|
0	0000_2	+	|	|	|	|	|	|	|
1	0001_2	−	+	|	|	|	|	|	|
2	0010_2	−	−	+	|	|	|	|	|
3	0011_2	−	−	−	+	|	|	|	|
4	0100_2	−	−	−	−	+	|	|	|
5	0101_2	−	−	−	−	−	+	|	|
6	0110_2	−	−	−	−	−	−	+	|
7	0111_2	−	−	−	−	−	−	−	+

次のように 2 の補数 (y) を求める式を $x =$ の形に変形しても同じ形になるので，x と y は「互いに 2 の補数」です．

$$y = 2^n - x \quad (y \text{ は } x \text{ に対する 2 の補数})$$
$$x = 2^n - y \quad (x \text{ は } y \text{ に対する 2 の補数})$$

一般に，2 の補数表現を用いた n ビット符号付き 2 進数が表現できる数値の範囲は次の式で計算できます．

$$-2^{n-1} \ \sim \ 2^{n-1} - 1$$

2 の補数表現を用いると，2 進数の足し算・引き算が，正負のどちらの数でも同じ手順で計算できます (詳しくは後述)．「手順が同じ＝演算回路が同じ」ことになりますので，実際にコンピュータを製作する上では非常に都合がよく，現在のコンピュータは，ほとんどの機種で 2 の補数表現を採用しています．

2.3.5　2 の補数を求める手順

$-x$ を n ビット 2 の補数表現 (y) で表します．2 の補数の定義より，y は次のように計算できます．

$$\begin{aligned} y &= 2^n - x \\ &= (2^n - 1 - x) + 1 \\ &= (x \text{ の 1 の補数}) + 1 \end{aligned} \tag{2.1}$$

「x に対する 1 の補数」はビット反転で簡単に求めることができます．「x に対する 2 の補数」は「x をビット反転した後，1 を加える」ことにより簡単に求めることができます．

ビット反転した後，1 を加える．

2.3.6　2 の補数から元の数を求める手順

次に，逆変換について考えます．既に 2.3.4 で触れたように 2 の補数と元の数は互いに「2 の補数」ですから，2 の補数を求めるのと同じ計算で，2 の補数から元の数を求めることができます．

ビット反転した後，1 を加える．

2.4　2 進数の計算

ここでは，2 進数の和と差の計算方法を学びます．

2.4.1　正の数の計算

2 進数の計算も 10 進数と同じ要領です．桁上がりが 10 ではなく，2 で発生することに注意してください．

もちろん，2 進数で計算しても 10 進数で計算しても同じ計算結果になります．

2 進数と 10 進数の計算を比較

10 進数		2 進数		
07		0111_2		(10 進の 7)
+ 05		+ 0101_2		(10 進の 5)
12		1100_2		(10 進の 12)
12		1100_2		(10 進の 12)
− 05		− 0101_2		(10 進の 5)
07		0111_2		(10 進の 7)

何進数で計算しても同じ結果になる．

2.4.2　負の数を含む計算

2 の補数表現を用いると，正の数だけのときと同じ要領で負の数を含む計算ができます．ここでは和の計算を例に説明します．

正の数と負の数の和

正の数 (X) と負の数 ($-A$)(A の 2 の補数 (B)) の和の計算を考えます．

$X + B$ の計算 ($X + (-A)$)

1. 結果が負の場合 ($|A| > |X|$)
   解は $-(A - X)$ になるはず！

$$\begin{aligned} X + B &= X + (2^n - A) \\ &= 2^n - (A - X) \end{aligned} \tag{2.2}$$

(2.2) 式は，正解 $-(A - X)$ の 2 の補数表現になっています．

例　$3 + (-5) = -2$ (4 ビット 2 の補数表現)

3 を 2 進数に変換すると 0011_2
-5 を 2 進数に変換すると 1011_2

「和を計算する」
$0011_2 + 1011_2 = 1110_2 = -2_{10}$

2. 結果が正またはゼロの場合 ($|X| \geq |A|$)
解は $(X - A)$ になるはず！

$$\begin{aligned} X + B &= X + (2^n - A) \\ &= 2^n + (X - A) \qquad (2.3) \\ &= X - A \end{aligned}$$

(2.3) 式の 2^n は，桁あふれにより結果に残らないので，正解 $(X - A)$ となります．

例　$5 + (-3) = 2$ (4 ビット 2 の補数表現)

5 を 2 進数に変換すると 0101_2
-3 を 2 進数に変換すると 1101_2

「和を計算する」
$0101_2 + 1101_2 = \boxed{1}0010_2 = 2_{10}$
(5 ビット目の 1 は桁あふれで消滅)

負の数と負の数の和

負の数 $(-A_1)$(2 の補数 (B_1)) と
負の数 $(-A_2)$(2 の補数 (B_2)) の和
$B_1 + B_2$ の計算 $((-A_1) + (-A_2))$
解は $-(A_1 + A_2)$ になるはず！

$$\begin{aligned} B_1 + B_2 &= (2^n - A_1) + (2^n - A_2) \\ &= 2^n + (2^n - (A_1 + A_2)) \qquad (2.4) \\ &= 2^n - (A_1 + A_2) \qquad (2.5) \end{aligned}$$

(2.4) 式の最初の 2^n は，桁あふれにより消滅する．(2.5) 式は正解 $(-(A_1 + A_2))$ の 2 の補数表現になっている．

例　$(-1) + (-3) = (-4)$ (4 ビット 2 の補数表現)

-1 を 2 進数に変換すると 1111_2
-3 を 2 進数に変換すると 1101_2

「和を計算する」
$1111_2 + 1101_2 = \boxed{1}1100_2 = -4_{10}$
(5 ビット目の 1 は桁あふれで消滅)

以上のように，2 の補数表現を使用すると負の数を含んだ足し算が簡単に計算できます．ここでは省略しますが，引き算も同様に正負の区別をすることなく計算できます．

MSB と LSB

「2 の補数表現を用いると，最上位ビットが 1 なら負の数と分かります.」のように最上位ビット と言う言葉をよく使います.

「最上位ビット」と言うのは長いので，これを英語にした「Most Significant Bit」の頭文字を取り MSB と略すことが良くあります．憶えておいて下さい.

同様に「最下位ビット」のことは英語で「Least Significant Bit」なので，略して LSB と呼びます．これも，憶えておきましょう.

問題

1. 次の数を「2 の補数表現を用いた 4 ビット符号付き 2 進数」で書き表しなさい.
3_{10}，-3_{10}，5_{10}，-5_{10}，6_{10}，-6_{10}

2. 次の数を「2 の補数表現を用いた 8 ビット符号付き 2 進数」で書き表しなさい.
3_{10}，-3_{10}，8_{10}，-8_{10}，
30_{10}，-30_{10}，100_{10}，-100_{10}

3. 次の「2 の補数表現を用いた 4 ビット符号付き 2 進数」を 10 進数で書き表しなさい.
1100_2，1011_2，0100_2，1101_2

4. 次の 2 進数は「2 の補数表現を用いた符号付き 2 進数」です．空欄を埋めなさい.

1)　$0011\ 1100_2$ + $0010\ 1101_2$ → $\square_{10}$ + $\square_{10}$ = $\square_{10}$; $\square_2$

2)　$0110\ 0100_2$ + $1000\ 0001_2$ → $\square_{10}$ + $\square_{10}$ = $\square_{10}$; $\square_2$

3)　$1110\ 0100_2$ + $0100\ 0001_2$ → $\square_{10}$ + $\square_{10}$ = $\square_{10}$; $\square_2$

4)　$1110\ 0100_2$ + $1100\ 0001_2$ → $\square_{10}$ + $\square_{10}$ = $\square_{10}$; $\square_2$

2.5　小数の表現

この節では，小数を 2 進数で表現する方法を紹介します．

2.5.1　2 進数による小数の表現

小数を含む数を，コンピュータ内部で 2 進数を用いて表現する方法は何種類かあります．ここでは，その中でも最も基本的な，固定小数点形式を紹介します．

これまでの 2 進数は，暗黙のうちに小数点が右端にあると考えました．小数点の位置を右端以外だと約束すれば，小数を含む数の表現も可能になります．

例　4 ビットで小数点が 2 桁目なら

$00.00_2 = 0.0_{10}$
$00.01_2 = 0.25_{10}$
$00.10_2 = 0.5_{10}$
$00.11_2 = 0.75_{10}$
$01.00_2 = 1.0_{10}$
$01.01_2 = 1.25_{10}$
$01.10_2 = 1.5_{10}$
$01.11_2 = 1.75_{10}$
$10.00_2 = 2.0_{10}$
...
$11.11_2 = 3.75_{10}$

小数点を中心に，左は 1 桁毎に 2 倍，右は 1 桁毎に 1/2 倍になります．(10 進数では，左は 1 桁毎に 10 倍，右は 1 桁毎に 1/10 倍になりましたね．)

2.5.2　10 進数との相互変換

1. 2 進数から 10 進数への変換
   整数の場合と同様に桁の重みを加算すれば 10 進数に変換できます．

   例　2 進 10 進変換
   $10.01_2 = 2 + 1/4 = 2.25_{10}$

2. 10 進数から 2 進数への変換
   整数の場合は，1/2 倍することで右にシフト (桁移動) し，小数点を横切った 0/1 を判定しました．(小数点を 1 が横切ると，余りが出ていました．)

   小数点以下の数値は，2 倍することで左にシフトし，小数点を横切った 0/1 を判定すれば 2 進数に変換できます．

例 小数を表す 10 進数から 2 進数への変換

2 進数	×2 は	10 進数
0.101_2	左シフト	0.625
↙	トと同じ	× 2
1.010_2		$\underline{1}.250$

2 進数	×2 は	10 進数
0.010_2	左シフト	0.250
↙	トと同じ	× 2
0.100_2		$\underline{0}.500$

2 進数	×2 は	10 進数
0.100_2	左シフト	0.500
↙	トと同じ	× 2
1.000_2		$\underline{1}.000$

10 進数で計算したとき，小数点を横切って整数部に出てきた数を小数点の右に順番に並べる．

$$0.\underline{101}_2$$

3. 整数部と小数部の両方がある場合
   整数部分と小数部分を分離して，別々に計算します．

問題

1. 次の 2 進数を 10 進数に変換しなさい．
   (a) 0.1001_2
   (b) 0.0101_2
   (c) 11.11_2
2. 次の 10 進数を 2 進数に変換しなさい．
   (a) 0.0625_{10}
   (b) 0.1875_{10}
   (c) 0.4375_{10}
3. 次の 10 進数を 2 進数に変換しなさい．(難しい問題)
   (a) 0.8_{10}
   (b) 0.7_{10}

2.6 文字の表現

この節では，文字を 2 進数で表現する方法を紹介します．(文字情報を表しているビットの読み方を約束する．)

2.6.1 文字コード

文字の場合，数値のような規則性を期待することはできません．そこで，使用する文字の一覧表を作成し，1 文字毎に対応する 2 進数 (コード) を決めます．この一覧表のことを「文字コード表」と呼びます．文字コード表を各自 (コンピュータメーカ等) が勝手に定義すると，コンピュータの間でのデータ交換に不便です．そこで，規格として制定してあります．

2.6.2 ASCII コード

ASCII(American Standard Code for Information Interchange) コードは，1963 年にアメリカ規格協会 (ANSI) が定めた情報交換用の文字コードです．英字，数字，記号等が含まれます．現在のパーソナルコンピュータ等で使用される文字コードは，ASCII コードが基本になっています．

ASCII 文字コード表

(上位3ビット)

(下位4ビット)	0	1	2	3	4	5	6	7
0	NUL	DLE	(SP)	0	@	P	`	p
1	SOH	DC1	!	1	A	Q	a	q
2	STX	DC2	”	2	B	R	b	r
3	ETX	DC3	#	3	C	S	c	s
4	EOT	DC4	$	4	D	T	d	t
5	ENQ	NAK	%	5	E	U	e	u
6	ACK	SYN	&	6	F	V	f	v
7	BEL	ETB	’	7	G	W	g	w
8	BS	CAN	(	8	H	X	h	x
9	HT	EM	)	9	I	Y	i	y
A	LF	SUB	*	:	J	Z	j	z
B	VT	ESC	+	;	K	[	k	{
C	FF	FS	,	<	L	\	l	|
D	CR	GS	−	=	M	]	m	}
E	SO	RS	.	>	N	^	n	∼
F	SI	US	/	?	O	—	o	DEL

文字コード表で 00H 〜 1FH と 7FH は，機能 (改行等) を表す特殊な文字になっています．20H(SP) は空白を表す文字です．

ASCII 文字コードは 7 ビットで表現されます．しかし，コンピュータの内部では 1 バイト (8 ビット) 単位の方が扱いやすいので，最上位に 0_2 を補って 8 ビットで扱うことがほとんどです．

2.6.3 JIS 文字コード

日本では JIS(Japan Industrial Standard：日本工業規格) の一部として，8 ビットコード (英数記号＋カナ) と，16 ビットコード (英数記号＋カナ＋ひらがな＋カタカナ＋漢字...) が定められています．

JIS 8 ビットコードは，ASCII コードに半角カタカナを追加したものです．記号，数字，英字の部分は同じ並びになっています．

2.7 補助単位

長さや重さを書くとき，1,000m を 1km，1,000g を 1kg のように書きました．この k のような記号を補助単位と呼びます．

コンピュータの世界でよく使用される補助単位には，k(キロ=10^3)，M(メガ=10^6)，G(ギガ=10^9)，T(テラ=10^{12}) 等があります．(1,000 倍毎に補助単位があります．) パソコンのカタログに「CPU のクロックは 1GHz」のような記述を見かけますね．

記憶容量を表す場合の補助単位は，$1,000$ の代わりに $2^{10} = 1,024$ を使用します．1,000 も 1,024 も k で表現すると紛らわしいので，k の代わりに ki と書き表すこともあります．つまり，$1kB = 1kiB = 2^{10}B = 1,024B$，$1MB = 1MiB = 2^{20}B = 1,048,576B$ となります．(「B」はバイトを表す．)

「H.D.D. の容量は 500GB」のように表示されている場合は，記憶容量を表しているので $G = 2^{30}$ で表示されています．同じことを「H.D.D. の容量は 500GiB」のように表示することもあります．

補助単位まとめ

一般的に			記憶容量		
値	記号	読み方	値	記号	読み方
10^3	k	キロ	2^{10}	ki	キビ
10^6	M	メガ	2^{20}	Mi	メビ
10^9	G	ギガ	2^{30}	Gi	ギビ
10^{12}	T	テラ	2^{40}	Ti	テビ

2.8　コンピュータの基本回路

前の節までで，コンピュータ内部での情報の表現方法を勉強しました．コンピュータの内部では，どんな情報も電気の「ON/OFF」で表現しているのでしたね．

この節では，電気の「ON/OFF」を使用して，計算や記憶をする回路を勉強します．

2.8.1　論理演算と論理回路

論理 (Yes/No，真/偽，True/False) を対象とする演算 (計算) のことを 論理演算 と呼びます．論理の「真/偽」をビットの「1/0」に対応させることにより，論理も電気の「ON/OFF」で表現することができます．

ここでは，論理演算と，論理演算をする回路を紹介します．論理演算をする回路のことを 論理回路 と呼びます．ここで紹介する論理回路が，コンピュータを構成する基本回路になります．

2.8.2　基本的な論理回路

基本的な論理回路を 4 種類，組合せてできたものを 2 種類，紹介します．

1. 論理積 (AND)
   二つのビットを入力し，両方が 1 のときだけ出力が 1 になるような論理演算です．

入力		出力
A	B	X
0	0	0
0	1	0
1	0	0
1	1	1

ANDの真理値表

$X = A \cdot B$

ANDの論理式



ANDの回路記号

2. 論理和 (OR)
   二つのビットを入力し，どちらかが 1 のとき出力が 1 になるような論理演算です．

入力		出力
A	B	X
0	0	0
0	1	1
1	0	1
1	1	1

ORの真理値表

$X = A + B$

ORの論理式



ORの回路記号

3. 否定 (NOT)
   一つのビットを入力し，入力と逆の論理を出力する論理演算です．

入力	出力
A	X
0	1
1	0

NOTの真理値表

$X = \overline{A}$

NOTの論理式



NOTの回路記号

4. 排他的論理和 (XOR)
   二つのビットを入力し，二つが異なるとき出力が 1 になるような論理演算です．

入力		出力
A	B	X
0	0	0
0	1	1
1	0	1
1	1	0

XORの真理値表

$X = A \oplus B$

XORの論理式



XORの回路記号

5. NAND
   否定 (NOT) と論理積 (AND) を組み合わせた演算です．

入力		出力
A	B	X
0	0	1
0	1	1
1	0	1
1	1	0

NANDの真理値表

$X = \overline{A \cdot B}$

NANDの論理式



NANDの回路記号

6. NOR
否定 (NOT) と論理和 (OR) を組み合わせた演算です.

入力		出力
A	B	X
0	0	1
0	1	0
1	0	0
1	1	0

NORの真理値表

$$X = \overline{A + B}$$

NORの論理式



NORの回路記号

2.8.3 演算回路

「基本的な論理回路」を組み合わせることにより, 足し算／引き算等のより複雑な演算を行う回路を実現できます.

1. 半加算器
例えば, 1 ビットの足し算回路は次の例のように実現できます. この例では A, B の 2 ビットを入力し, 和 (S) と上の桁への桁上がり (C) を出力する回路を示しています. このような回路のことを「半加算器」と呼びます.

例 1 ビット半加算器

```
  A     0     0     1     1
+ B   + 0   + 1   + 0   + 1
-----------------------------
C S   0 0   0 1   0 1   1 0
```

入力		出力	
A	B	C	S
0	0	0	0
0	1	0	1
1	0	0	1
1	1	1	0

半加算器の真理値表



半加算器の回路図

2. 全加算器
半加算器は桁上がりを出力しますが, 自分自身は下の桁からの桁上がりを入力することができません. 桁上がりの入力を備えた 1 ビット足し算器を「全加算器」と呼びます.

次の例のように, 全加算器は半加算器を組み合わせて実現することができます.

例 1 ビット全加算器

入力			出力	
A	B	I	C	S
0	0	0	0	0
0	0	1	0	1
0	1	0	0	1
0	1	1	1	0
1	0	0	0	1
1	0	1	1	0
1	1	0	1	0
1	1	1	1	1

A:計算の入力
B:計算の入力
I:桁上がりの入力

全加算器の真理値表



全加算器の回路図

3. n ビット加算器
半加算器と全加算器を組み合わせることで, 任意ビットの加算器を実現することができます. 次に 4 ビット加算器の例を示します.

例 4 ビット加算器



4ビット加算器の回路図

4. n ビット 1 の補数器
   1 の補数を作る回路です．1 の補数はビット反転によってできるので，NOT でできます．次に，4 ビットの例を示します．簡単ですね．

例 4 ビット 1 の補数器



4ビット1の補数器の回路図

5. n ビット 2 の補数器
   2 の補数を作る回路です．2 の補数は，1 の補数に 1 足すとできるので，1 の補数器の出力に半加算器を組み合わせるとできます．

例 4 ビット 2 の補数器



4ビット2の補数器の回路図

演算回路を数種類紹介しました．ここに紹介しませんでしたが，引き算回路等も同様に製作可能です．

2.8.4　記憶回路

「基本的な論理回路」を組み合わせることにより，記憶機能を実現することもできます．ここでは，最も簡単な RS-FF(RS フリップフロップ) を紹介します．

RS-FF は S(Set) と R(Reset) の二つの入力と，状態 $(Q, \bar{Q})$ の出力を持つ回路です．S=0,R=1 を入力することにより，回路はリセットされ出力 Q は 0 になります．S=1,R=0 を入力することにより，回路はセットされ出力 Q は 1 になります．$\bar{Q}$ には，常に Q の否定が出力されます．

S=0,R=0 を入力すると，回路は直前の値を「記憶」します．S=1,R=1 は入力してはいけない組合せです．

RS-FF は，回路をリセット／セットした後で「記憶」状態にすることにより,「値を記憶する回路 (＝記憶回路)」として働きます．

RS-FF

入力		出力
S	R	Q
0	0	記憶
0	1	0
1	0	1
1	1	禁止

RS-FFの動作



RS-FFの回路

論理回路素子

回路を製作するには，基本的な論理回路を内蔵した集積回路 (論理 IC) を用いることができます．

様々な論理 IC が販売されていますが，ここでは，74 シリーズのものを紹介します．下の例に示したのは，NAND ゲートを四つ内蔵した 7400 と呼ばれる IC です．同様な IC で，AND，OR，XOR，NOT 等の回路を内蔵したものがあります．

このような IC 同士を配線して組み合わせることにより，本文で紹介した演算回路や記憶回路を実現することができます．

論理 IC



外観



論理ICの内部

第3章　組み立て

この章では，TeC7 の組み立て方を説明します．何年間も使用するマイコンなので，慎重に組み立てて下さい．(TeC6 の組み立て方は本書の古い版に掲載されています．TeC の Web サイトから PDF をダウンロードして御覧下さい．)

3.1　ハンダ付け

マイコン本体は基板に部品をハンダ付けして組み立てます．ハンダ付けの要領を図 3.1 に示します．上手なハンダ付けができると，ハンダが富士山型になります．ハンダ付けの経験がある人にコツを習って，少し練習して下さい．



図 3.1: ハンダ付け手順

ハンダ付けの順序は，次のような基準で決めます．

1. 背の低い部品
2. 基板中心の部品
3. 壊れにくい部品

3.2　部品の取り付け

組み立ては，部品をハンダ付けして基板に取り付ける作業です．以下の説明では，取り付ける部品が表になっています．表の部品の「記号」と同じ「記号」が，基板の上に白い文字で書いてあるので，その場所に部品をハンダ付けします．

3.2.1　抵抗器

抵抗器は図 3.2 のような外観をした部品です．抵抗値は，部品のカラーコードで見分けます．正しい抵抗値の抵抗器を選んで，足を曲げて基板に差し込みます．このとき，部品と基板の間に隙間ができないように気を付けて下さい．



(1) カラーコードの読み方



(2) 実装方法

図 3.2: 抵抗器

抵抗器の一覧を表 3.1 に示します．カラーコードを良く確認して，間違えないように部品を選択して下さい．抵抗器に向きはありませんが，カラーコードの読みやすさを考えて方向を統一する方が良いでしょう．

表 3.1: 抵抗器

記号	カラーコード	説明
R1,15	黄紫茶金	470 Ω 1/4W カーボン
R2-6	赤紫茶金	270 Ω 1/4W カーボン
R7	茶緑黒茶茶	1.5k Ω 1/4W 金属皮膜
R8	橙黒黒茶茶	3.0k Ω 1/4W 金属皮膜
R9-12	黄紫赤金	4.7k Ω 1/4W カーボン
R13	橙橙茶金	330 Ω 1/4W カーボン
R14	茶緑茶金	150 Ω 1/4W カーボン

3.2.2　積層セラミック・コンデンサ

積層セラミックコンデンサは 4 種類，合計 31 個あります．そのうち，25 個は同じ部品です．一覧を表 3.2 に示します．向きはありませんが，完成したとき表面の文字が読みやすい方向に取り付けると良いでしょう．図 3.3 を良く見て，ハンダ付けして下さい．

表 3.2: 積層セラミック・コンデンサ

記号	型番	説明
C1,2	47	$47pF$
C3,4,11-15,17-34	104	$0.1\mu F$
C35	101	$100pF$
C6,8,10	475	$4.7\mu F$



図 3.3: 積層セラミック・コンデンサ

3.2.3　IC

表 3.3 の部品を基板にハンダ付けします．IC には向きがありますので，間違えないように注意して下さい．図 3.4 の説明を良く読んで，方向を間違えないように取り付けて下さい．この時，図 3.5 のように**「IC の足を修正して」**基板の穴に差し込んで下さい．

部品が傾いたり，浮き上がったりしないようにするには，IC の足 2ヵ所程度を**「仮にハンダ付けし，問題が無いか確認して」**から他の足をハンダ付けすると良いでしょう．足の多い部品を間違ってハンダ付けしてしまうと，取り外すのが**絶望的に大変です**ので，くれぐれも間違えないように注意して下さい．

表 3.3: IC

記号	型番	説明
U3	K516	水晶発振 IC
U6	LM339	電圧比較 IC



図 3.4: IC 取付の説明図



図 3.5: IC 足の調整方法

3.2.4 集合抵抗器とラダー抵抗器

集合抵抗器とラダー抵抗器は図 3.6 のような部品です．この部品にも向きがありますので，注意して下さい．IC と同じ要領で，傾きや浮き上がりが無いよう注意深くハンダ付けしましょう．

表 3.4: 集合抵抗器

記号	型番	説明
RA1,2	M9-1-471 (L91S 471)	470 Ω (8 素子)
RA3	M9-1-391 (L91S 391)	390 Ω (8 素子)
RA4	M5-1-391 (L51S 391)	390 Ω (4 素子)
RA5	8L103	ラダー抵抗器



(1)方向の見分け方



(2)内部の構造

図 3.6: 集合抵抗器とラダー抵抗器

3.2.5 フェライトビーズ

フェライトビーズは電源のノイズを弱くする働きのある部品です．図 3.7 の部品をハンダ付けします．この部品は向きはありません．リード線の先を基板に差し込んでハンダ付けします．

表 3.5: フェライトビーズ

記号	型番	説明
FB1,2,3	なし	なし



図 3.7: フェライトビーズ

3.2.6 ジャンパ

表 3.6 の部品をハンダ付けします (1 個だけです)．熱に弱い部品なので手早くハンダ付けしてください．ハンダ付けができたら，図 3.8 のように上半分の部品を差し込みます．図をよく見て上下を間違えないように注意してください．

差し込む位置は TeC の位置です．図 3.8 を確認して下さい．

表 3.6: ジャンパ

記号	型番	説明
J1	なし	なし



(1) 組みたて方



(2) 差し込み位置

図 3.8: ジャンパ

3.2.7　圧電スピーカ

圧電スピーカは TeC が音を出すためのスピーカの役割をします．表 3.7 の部品です．向きはありません．基板との隙間ができないように気を付けて下さい．

表 3.7: 圧電スピーカ

記号	型番	説明
BZ1	なし	圧電スピーカ



図 3.9: 圧電スピーカ

3.2.8　電解コンデンサ

表 3.8 の部品をハンダ付けします．部品の形状は図 3.10 のようなものです．この部品も，向きがありますので取り付け方向に注意が必要です．長い足の方が「＋」になります．図 3.10 を良く見て，向きを間違えないようにしてください．

表 3.8: 電解コンデンサ

記号	型番	説明
C0,C5,C7,C9,C16	$47\mu F25V$	$47\mu F$



図 3.10: 電解コンデンサ

3.2.9　LED(ランプ)

LED(ランプ) は，マイコンを使用するとき，いつも注視する部品です．傾いていたりすると，とても気になりますので綺麗に取り付けて下さい．LED は，赤，緑，黄色の 3 種類があります．LED をハンダ付けする要領は次の通りです．

1. 同じ色を一斉に**アノード (＋)** だけハンダ付けする．(黄→緑→赤の順に中心から取り付ける．)
2. LED が垂直になっているか確認する．(垂直になっていない場合は，再度温めて修正する．)
3. LED が奥までささっているか確認する．
4. カソードをハンダ付けする．
5. リード線を切る．

アノード (+)，カソード (−) の見分け方は図 3.11 (1) の通りです．方向を間違えないようにしてください．3 種類の LED は，図 3.11 (2) のように配置してください．



(1) 方向の見方



(2) LEDの配置

図 3.11: LED の取り付け

3.2.10 スイッチ

スイッチには，**トグルスイッチ**と**プッシュスイッチ**の 2 種類があります．トグルスイッチは上下に倒すタイプのスイッチ，プッシュスイッチは押しボタンになったスイッチです (図 3.12 参照)．



図 3.12: スイッチ

スイッチは，マイコンを操作するときいつも触る部品です．スイッチが傾いていたりすると操作性が悪くなりますので，まっすぐになるよう慎重にハンダ付けしてください．また，スイッチは温まりにくいので，ハンダ付けするときは少し余計に熱するようにしてください．スイッチをハンダ付けする手順は以下の通りです．

1. 足を穴にしっかり差し込み，ハンダ面に 7 本の足が均等に出ていることを確認する．
2. 足のうち 1 本をハンダ付けする．
3. スイッチが傾いていないか確認する．
   (傾いていた場合は，温め直して修正する．)
4. 他の足をハンダ付けする．

スイッチの配置は図 3.13 の通りです．間違えないように配置してください．



図 3.13: スイッチの配置

3.2.11 JTAG コネクタ

マイコン内部の回路を書き換えるときに使用するコネクタです．表 3.9 の部品を使用します．この部品にも向きがあります．図 3.14 を良く見て，向きを間違えないように取り付けて下さい．

表 3.9: JTAG コネクタ

記号	型番	説明
CN4	なし	小さい 14 ピンのコネクタ



図 3.14: JTAG コネクタ

3.2.12 入出力ポートコネクタ

マイコンの外部に追加の回路を接続するときに使用するコネクタです．表 3.10 の部品を使用します．この部品にも向きがあります．図 3.15 を良く見て，向きを間違えないように取り付けて下さい．

表 3.10: 入出力ポートコネクタ

記号	型番	説明
CN5	なし	大きい 20 ピンのコネクタ



図 3.15: 入出力ポートコネクタ

3.2.13　電源コネクタ

表 3.11 の電源コネクタは，図 3.16 (1) のような形状の部品です．このコネクタからマイコンに電源を供給します．パソコンの USB ポートと接続してパソコンとの通信に使用することもあります．図 3.16 (2) の基板上の取り付け穴にハンダ付けします．付属のコネクタ部品によっては手前の二つの穴を使用しないこともあります．このコネクタの取り付けでは，失敗するとひどい火傷をする恐れがあります．以下の手順を良く読んで慎重に作業してください．

表 3.11: 電源コネクタ

記号	型番	説明
CN1	なし	USB-B コネクタ



(1) 形状　　(2) 取り付け穴

図 3.16: 電源コネクタ

1. 穴にしっかり差し込む．
2. 大きな穴とコネクタの端子をハンダゴテで十分熱する．
3. 大きな穴が塞がるまで，ハンダをどんどん融かし込む．
4. **十分に冷めるのを待つ．**
5. 部品が傾いていないか確認する．
6. 小さな穴に部品の足をハンダ付けする．

3.2.14　PS/2 コネクタ

表 3.12 はマイコンにパソコン用のキーボードを接続するコネクタです．外観は図 3.17 (1) の通りです．ハンダ付けの要領は **電源コネクタ** と同様です．火傷に気を付けてハンダ付けして下さい．

表 3.12: PS/2 コネクタ

記号	型番	説明
CN2	なし	PS/2 コネクタ



(1) 形状　　(2) 取り付け穴

図 3.17: PS/2 コネクタ

3.2.15　VGA コネクタ

表 3.13 はマイコンにパソコン用のディスプレイを接続するコネクタです．外観は図 3.18 (1) の通りです．取り付け穴の配置を図 3.18 (2) に示します．大きな穴に部品の足を差し込むとロックされる仕組みになっています．大きな穴はハンダ付けしません．

1. ロックされるまで穴にしっかり差し込む．
2. 小さなピンをハンダ付けする．

表 3.13: VGA コネクタ

記号	型番	説明
CN3	なし	VGA コネクタ



(1) 形状　　(2) 取り付け穴

図 3.18: VGA コネクタ

3.2.16 ゴム足の取り付け

基板の裏側に足をつけます．足は，両面テープの付いた黒い円盤状の部品です．足を取り付ける場所は，図 3.19 の通りです．取り付け場所の裏側 (基板の表側) に○印が付いています．

まず，アルコールで周辺をきれいに拭いてから，鉄板とシールを剥がして，両面テープで指定の場所に貼り付けて下さい．



図 3.19: ゴム足の取り付け位置

3.2.17 プッシュスイッチの頭の取り付け

最後に，プッシュスイッチに頭を取り付けます．図 3.20 を良く見て取り付けて下さい．



図 3.20: プッシュスイッチの頭の取り付け方

3.3 完成

3.3.1 命令表の貼り付け

ケースのフタの裏に，命令表を貼り付けて下さい．

3.3.2 目視確認

ご苦労さまでした．これでマイコンが完成しました．でも，もう一度，部品の種類，向きが間違っていないか目視確認して下さい．

3.3.3 設計データの書き込み

間違えがなかったら，教員に頼んでマイコンの設計データを書き込んでもらって下さい．(設計データの書き込みに JTAG コネクタが使用されます．) これで，マイコンとして動作するようになったはずです．

第4章　マイコンの操作

この章では，マイコンの操作方法を学習します．併せてマイコンの内部にあるレジスタや記憶装置も憶えましょう．この章に掲載している写真は TeC7 のものです．TeC6 とはコネクタの構成が異なりますが，コンソールパネルの構成，操作方法，内部構造は共通です．

4.1　各部の名称

図 4.1 に，マイコン各部の名称 (または役割り) を示します．各部の役割りは次の通りです．

1. コンソールパネル
   マイコンにプログラムを入力したり，マイコンの内部を観察したり，プログラムの実行を指示したりするために使用する部分です．データの入力や表示は 2 進数を用います．本マイコンの顔です．

2. リセットスイッチ
   マイコンをリセットするための押しボタンスイッチです．

3. ジャンパ
   マイコンのモードを切替えます．1，2 年生は，「TeC」の位置にジャンパーを挿して「TeC モード」で使用して下さい．高学年で「TaC モード」を使用します.「DEMO1 モード」,「DEMO2 モード」は，電子オルゴールのデモプログラムが予め入力された状態になるモードです．

4. スピーカー
   コンソールパネルを操作したとき，操作を確認するための音を出します．また，マイコンに入力したプログラムで音を出すこともできます．

5. 電源コネクタ
   マイコンに電力を供給するコネクタです．付属のアダプタに USB ケーブルを用いて接続します．USB ケーブルでパソコンと接続して通信させることも可能です．

6. キーボードコネクタ
   パソコン用の PS/2 キーボードを接続することができます．この機能は,「TaC モード」で使用します．

7. ディスプレイコネクタ
   パソコン用の ディスプレイを接続することができます．この機能は,「TaC モード」で使用します．

8. μSD スロット
   携帯電話などで使用する μSD メモリカードを挿入します．この機能は,「TaC モード」で使用します．

9. JTAG コネクタ
   マイコン本体の回路を変更するとき使用します．

10. 入出力ポートコネクタ
    マイコンの外部に追加の回路を接続するために使用します．

4.2　コンソールパネル

図 4.2 にコンソールパネルの様子を示します．この図を見ながら以下を読んで下さい．

4.2.1　コンソールパネルの構成

コンソールパネルは，以下のランプやスイッチにより構成されています．コンソールパネルが，マイコンボードの約半分の面積を使用していますが，これは，操作性を考慮した結果です．これ以上，小さくするとスイッチの間隔が狭くなり過ぎ，操作性が悪くなります．

1. アドレスランプ
   最上段の 8 個の赤ランプはアドレスランプです．アドレスランプは，主記憶を操作するとき，操作対象となる主記憶の番地を表示します．

2. データランプ
   上から 2 段目の 8 個の緑ランプは，選択されたレジスタまたは主記憶の内容を表示します．



図 4.1: 各部の名称

3. ロータリースイッチ
   上から 3 段目の 6 個の黄色ランプと左右の押しボタンスイッチは，ロータリースイッチの代用です．通常のロータリースイッチは背が高く，ビデオカセットケースにマイコンを収めるための障害になるのでこのようにしました．

   ランプのうち一つが点灯し，レジスタ (G0，G1，G2，SP，PC) または主記憶 (MM) のどれが選択されているか表示します．左右の押しボタンスイッチ (←，→) によって，点灯するランプを変更することができます．ロータリースイッチで選択されたものの内容が，データランプに表示されます．

4. データスイッチ
   8 個のトグルスイッチは，データやアドレスの値を入力するためのものです．

5. 機能スイッチ
   一番下の列の 8 個のスイッチには，様々な機能が割り当てられています．左から順に，**ブレークポイントモード (BREAK)**，**ステップ実行モード (STEP)**，**プログラム実行 (RUN)**，**プログラム停止 (STOP)**，**アドレスセット (SETA)**，**アドレスを進める (INCA)**，**アドレスを戻す (DECA)**，**書き込み (WRITE)** の機能を持ちます．押しボタンスイッチは，長押しでリピート機能が働きます．

6. 実行ランプ
   右上の RUN ランプは，CPU がプログラムを実行中であることを表します．CPU が正しくない命令を実行した場合は点滅して異常を知らせます．

7. フラグランプ
   実行ランプ下の三つの黄色ランプ C(Carry)，S(Sign)，Z(Zero) は，フラグの値を表示します．

4.2.2　コンソールパネルの操作方法

コンソールパネルから，プログラムを実行させたり実行を途中で止めたりすることができます．また，図 4.3 に示す TeC 内部の情報をアクセスすることができます．TeC の内部には 3 ビットのフラグ，8 ビットのレジスタが 5 個，8 ビット × 256 のメモリが内蔵されています．

コンソールパネルのランプが，レジスタやフラグの値を表示します．ランプは点灯している状態が 2 進数の’1’，消灯している状態が 2 進数の’0’ を表現しています．

データを入力するときはトグルスイッチで値を表現します．トグルスイッチ (上下に倒すスイッチ) は，



図 4.2: コンソールパネル

上に倒した状態が 2 進数の'1'，下に倒した状態が 2 進数の'0' に対応します．以下では，コンソールパネルの操作方法を目的別に説明します．

フラグの表示

フラグの状態は C，S，Z の三つのランプに常時表示されています．フラグの値を表示するために何も操作をする必要はありません．フラグの値をコンソールパネルから変更する方法は用意されていません．

レジスタの表示と書き換え

五つのレジスタ (G0，G1，G2，SP，PC) を表示する方法と，書き換える方法を説明します．レジスタの値は，緑色のデータランプに表示されます．8 個のランプで 8 ビットの情報を表現します．どのレジスタをデータランプに表示するかは，ロータリースイッチで選択します．

レジスタの値を書き換える手順は，次の通りです．

1. ロータリースイッチを操作し，書き換えたいレジスタの値が表示される状態にする．
2. 書き込むデータをデータスイッチにセットする．
3. **WRITE スイッチ**を押す．

主記憶 (メモリ) の表示と書き換え

図 4.3 を見ると分かるように，メモリには 00_{16} 〜 FF_{16} (2 進数で表現すると 0000 0000_2 〜 1111 1111_2) の番地 (アドレス) が付けられています．メモリへ対する表示や書き換え等の操作は，1 アドレス (8 ビット) 毎に行いますので，操作対象となるメモリのアドレスを指定した上で行う必要があります．メモリ・アドレスは，**アドレスランプ**にセットします．アドレスランプにアドレスをセットする方法は次の通りです．

1. ロータリースイッチを MM(Main Memory) に合わせる．
2. アドレスをデータスイッチにセットする．
3. **SETA(Set Address) スイッチ**を押す．
   (データスイッチにセットした値がアドレスランプに表示される．)
4. アドレスを次の番地に進めたいときは，**INCA(Increment Address) スイッチ**を押す．

フラグ		計算結果を表す目印
レジスタ	G0,G1,G2	計算対象となるデータや 計算結果を一時的に置く場所
	SP	G0,G1,G2同様な用途に使える。 特別な使い方もできる。
	PC	実行するプログラムのアドレス
主記憶		プログラムや変数を置く場所



図 4.3: TeC の内部

5. アドレスを前の番地に戻したいときは，**DECA(Decrement Address) スイッチ**を押す．

アドレスの指定方法が分かったところで，メモリ内容を表示する手順を説明します．

1. ロータリースイッチを MM(Main Memory) に合わせる．
2. 操作の対象となるアドレスをアドレスランプで指定する．
3. データランプに該当メモリ番地の内容が表示される．
4. 別の番地の内容を表示したいときは，SETA，INCA，DECA 等のスイッチで，アドレスランプの表示を変更する．

次に，メモリ内容を書き換える手順を説明します．

1. 上の手順に従い，書き換えたいメモリ番地の内容がデータランプに表示された状態にする．
2. データスイッチに書き込みたい値をセットする．
3. **WRITE スイッチ**を押す．
4. メモリにデータが書き込まれる．

なお，**WRITE スイッチ**を押したとき，アドレスが自動的に次の番地に変化します．これは，連続したデータの書き込みに便利だからです．

プログラムの実行

ノイマン型のコンピュータは，**「プログラム内蔵方式 (ストアードプログラム方式)」**を採用しているのが特徴です．「プログラム内蔵方式」とは，データだけでなくプログラムもメモリに記憶する方式です．TeC もプログラム内蔵方式ですので，メモリにデータのようにプログラムを書き込んで実行します．つまり，コンソールパネルからデータを書き込む方法を知っていれば，プログラムを書き込むこともできます．

ノイマン型コンピュータのもう一つの特徴として，**「逐次実行」**があげられます．「逐次実行」とは，メモリに記憶したプログラムの命令をアドレス順に一つ一つ順番に実行することを言います．次に実行する命令のアドレスは，PC レジスタが記憶しています．

TeC でプログラムを実行する手順は，次の通りです．

1. プログラムをメモリに書き込む．
2. プログラムの開始番地を PC に書き込む．
3. **BREAK，STEP スイッチ**を下に倒す．
4. **RUN スイッチ**を押す．(RUN ランプ点灯)
5. プログラムの実行終了．(RUN ランプ消灯)

次に，とても短いプログラムの打ち込みと実行の例を示します．

例 最も簡単なプログラムの実行

下は，何もしない命令 (NO 命令) を三つ実行した後，停止命令 (HALT 命令) を実行して停止するプログラムです．このプログラムを打ち込んで実行します．

最も簡単なプログラムのリスト

番地	命令	命令の種類
00_{16}	00_{16}	NO
01_{16}	00_{16}	NO
02_{16}	00_{16}	NO
03_{16}	FF_{16}	HALT

プログラム打ち込み手順

1. ロータリースイッチを MM に合わせる．
2. アドレスランプに 00_{16} をセットする．
3. データスイッチに 00_{16} をセットする．
4. WRITE スイッチを 3 回押し，メモリ 00_{16} 番地〜02_{16} 番地の 3 番地に 00_{16} を書き込む．
5. データスイッチに FF_{16} をセットする．
6. WRITE スイッチを押し，メモリ 03_{16} 番地に FF_{16} を書き込む．

プログラム実行手順

1. ロータリースイッチを PC に合わせる．
2. データスイッチに 00_{16} をセットする．
3. WRITE スイッチを押し，PC を 00_{16} にする．
4. BREAK，STEP スイッチを下に倒す．
5. RUN スイッチを押し実行を開始する．
6. 一瞬でプログラム実行が終了する．
7. PC は，HALT 命令実行後なので，その次の番地 04_{16} を指している．
8. 何もしない命令と，停止命令しか実行していないので，PC 以外のレジスタ，フラグ，メモリに変化はない．

プログラムのステップ実行

プログラム中の命令を，1 命令ずつ実行することを**ステップ実行**と言います．作ったプログラムが予定通りに動かないとき，原因を調べるために使用します．(このような作業を**デバッグ**と言います．) ステップ実行の手順は次の通りです．

1. プログラムをメモリに書き込む．
2. プログラムの開始番地を PC レジスタに書き込む．
3. **STEP スイッチ**が上に倒れていることを確認する．(倒れていなかった場合は倒す．)
4. **RUN スイッチ**を押す．
5. PC にセットしてあった番地の 1 命令が実行され，プログラムが停止する．
6. レジスタやメモリの状態が予想通りか確認する．
7. 確認が終わったら RUN スイッチを押して，次の命令を実行させる．
8. プログラムの終わりまで以上の操作を繰り返す．

次に，ステップ実行の例を示します．

例 プログラムのステップ実行

前の例と同じプログラムを，ステップ実行モードで実行してみます．

プログラムの実行手順

1. 前の例と同様に，プログラムを打ち込む．
2. STEP スイッチを上に倒して (ステップ実行モードにして)，前の例と同様にプログラムの実行を開始する．
3. ステップ実行モードなので，00_{16} 番地の命令だけを実行し停止する．このとき，PC は次命令の番地 01_{16} を指している．
4. 再度 RUN スイッチを押すと，01_{16} 番地の命令を実行し，PC が 02_{16} を指して停止する．
5. 以後，RUN スイッチを押す毎に，1 命令実行しては停止する．

ブレークポイントを使用した実行

デバッグをするときに目的の命令まで一気に進んでから，ステップ実行をしたい場合があります．そのとき，一気に目的の命令まで進むのにブレークポイントを使用できます．ブレークポイントとは，プログラムの実行を停止する場所 (アドレス) のことです．次のような手順になります．

1. プログラムをメモリに書き込む．
2. プログラムの開始番地を PC レジスタに書き込む．
3. **BREAK スイッチが上**に **STEP スイッチが下**に倒れていることを確認する．
   (倒れていなかった場合は倒す．)
4. プログラムの実行を停止させる命令のアドレスをデータスイッチにセットする．
5. **RUN スイッチ**を押す．
6. データスイッチで指定したアドレスの命令を**実行後**，停止する．

例 プログラムのブレークポイントモードでの実行
前の例と同じプログラムを，ブレークポイントモードで実行してみます．

プログラムの実行手順

1. 前の例と同様にプログラムを打ち込む．
2. 実行を停止したい命令のアドレス (今回は，01_{16}) をデータスイッチにセットする．
3. STEP スイッチを**下**に，BREAK スイッチを**上**に倒して (ブレークポイントモードにして)，前の例と同様にプログラムの実行を開始する．
4. ブレークポイントモードなので 01_{16} 番地の命令を実行したら停止する．このとき，PC は次命令の番地 02_{16} を指している．
5. 再度 RUN スイッチを押すと 02_{16} 番地からプログラムの実行を再開する．
6. 03_{16} 番地には HALT 命令が格納されているので実行が停止する．

4.3　リセットスイッチ

基板の左下にある RESET と書かれた押しボタンスイッチのことです．TeC をリセットしたいときに使用します．リセットスイッチを押すと，フラグ，レジスタ，アドレスランプ，TeC に内蔵された入出力装置の状態が (‘0’ に) クリアされます．メモリの内容と，ロータリースイッチの状態はクリアされません．

4.4　操作音を小さくする

押しボタンスイッチを押すと，ボタンが押されたことが確認できるように,「ピッ」と短い電子音が鳴るようになっています．しかし，静かな部屋で使用する場合，この音が邪魔になります．

このようなときは，**STOP スイッチ**を押しながら **RESET スイッチ**を押して下さい．電子音が，かすかに聞こえる「プッ」という音に変わります．

演習

1. TeC のジャンパーを DEMO1 の位置に差し込んで電源を投入し，00_{16} 番地から電子オルゴールの曲データを入力し実行しなさい．曲データの作り方は付録Aを参照しなさい．デモモード 1 (DEMO1) に付いては，CD–ROM の ReadMe.txt ファイルに参考になる情報があります．
2. 付録Aの電子オルゴールプログラムを入力／実行しなさい．
3. 付録Aの電子オルゴールプログラムの曲データを変更して実行しなさい．

第5章　プログラミング

この章では，TeC のプログラミングを学習します．まず，TeC の内部構成を勉強し，ノイマン型コンピュータの例として，TeC が適切であることを確認します．次に，TeC の命令と，それを使用したプログラムの作成を勉強します．

5.1　コンピュータの構成

プログラミングの対象となるコンピュータの構成を確認します．ここでは，一般のコンピュータの構成と，TeC の構成の両方を説明します．TeC の構成は一般のコンピュータの構成を簡単化したもので，原理的には同じものだと分かります．

5.1.1　一般的なコンピュータの構成

一般的にパソコン等の構成は，図 5.1 のようになっています．各部分の役割は次の通りです．

1. CPU(Central Processing Unit:中央処理装置)
   命令を読み込んで，命令に従い計算をするコンピュータの心臓部です．計算機能，制御機能を持っています．CPU が他の部分に積極的に働きかけることにより，コンピュータが働きます．
2. 主記憶装置 (メモリ)
   プログラムやデータを記憶する記憶装置です．CPU が高速に読み書きをすることができます．
3. 入出力インタフェース
   入出力装置をコンピュータに接続するための回路です．入出力装置の種類や機種毎に，専用のインターフェース回路が必要になります．
4. 入出力装置
   時計やコンピュータの外部とデータのやりとりをする装置のことです．タイマ，キーボード，マウス，ディスプレイ，プリンタや通信装置等がこれにあたります．ハードディスクドライブや CD-ROM ドライブ等も入出力装置の仲間ですが，データを記憶する装置であるので少し性格が異なります．そこで，これらは補助記憶装置と呼ばれることがあります．
5. バス
   CPU と主記憶装置や入出力インタフェースを接続する配線のことです．アドレスやデータ，制御信号等がバスで伝達されます．CPU はバスを制御することにより，バスに接続された装置とデータのやりとりをします．

5.1.2　TeC の構成

TeC7 の構成は図 5.2 のようになっています．接続されている入出力装置が少ない，メモリの容量が小さい等の違いはありますが，基本的には一般的なコンピュータと同じになっています．各部分の役割りは次の通りです．

1. CPU
   TeC の CPU も計算機能，制御機能を持っています．図 4.3 に示した**フラグ**と**レジスタ**は CPU の内部にあります．
2. 主記憶装置 (メモリ)
   TeC のメモリは，図 4.3 に示した 8 ビット構成，256 アドレスのものです．一つのアドレスに 8 ビットの情報を記憶することができます．
3. 入出力インタフェース
   入出力インターフェース回路を通してバスと入出力装置が接続されます．
4. 入出力装置
   入出力装置として，シリアル通信回路，スピーカ，コンソールパネルのデータスイッチ，タイマ，入出力ポートが搭載されています．(TeC6 には，タイマと入出力ポートがありません．)
5. バス
   CPU と主記憶装置や入出力インタフェースの間で次の情報を伝達します．
   (a) 8 ビットのアドレス
   (b) 8 ビットのデータ
   (c) いくつかの制御情報



図 5.1: 一般的なコンピュータの構成



図 5.2: TeC の構成

5.2 機械語プログラミング

機械語命令の羅列がプログラムです．機械語命令の羅列を作る作業がプログラミングです．

5.2.1 機械語命令

コンピュータのCPUは，メモリから命令を取り出し，取り出した命令を解釈し，決められた計算等を行います．CPUが解釈できる状態の命令を，「CPU＝機械」が解釈できる命令なので**「機械語命令」**と呼びます．機械語命令は2進数でメモリに書き込みます．

2進数を用いた機械語の表現方法だけでは人間にとって分かり難いので，命令を意味する英語を簡略化した短い綴で表します．これを**「ニーモニック」**と呼びます．

機械語命令	ニーモニック	意味
$0000\ 0000_2$	NO	No Operation
$1111\ 1111_2$	HALT	Halt

5.2.2 ノイマン型コンピュータの特徴

TeCもノイマン型コンピュータの一種です．ノイマン型コンピュータの特徴として次の3点があげられますが，これは，TeCにも当てはまります．

1. プログラム内蔵 (ストアードプログラム) 方式
   データだけでなく，プログラムもメモリに記憶する方式です．TeCでもプログラムはメモリに書き込んで実行します．
2. 逐次実行
   プログラムの命令を，メモリのアドレス順に一つ一つ順番に実行することを言います．
3. 2進法
   コンピュータの内部では，プログラムやデータの表現に2進数を使います．

5.2.3 機械語プログラミング

ノイマン型コンピュータ上での機械語プログラミングとは，逐次実行により実行される順に機械語命令の羅列を作る作業のことです．作ったプログラムは，2進数にしてメモリに書き込んで実行します．

TeCにどのような機械語命令があるかを，次の節から種類別に説明します．

5.3 特殊な命令

まず，TeCの機械語命令の中で特別なものから説明します．

5.3.1 NO(No Operation) 命令

意味：何もしない．
ニーモニック：NO
命令フォーマット：NO命令は，1バイト長の命令です．

第1バイト	
OP	GR XR
0000_2	$00\ 00_2$

命令フォーマットは，その機械語命令が2進数でどのように表現されるかを表します．NO命令の場合，機械語が1バイトであること，その1バイトの内容が，OPフィールド4ビットが0000_2，GRフィールド2ビットが00_2，XRフィールド2ビットが00_2であることが命令フォーマットの図から分かります．NO命令以外の命令も，フィールドのビット数は同じです．

5.3.2 HALT(Halt) 命令

意味：プログラム実行の終了．
ニーモニック：HALT
命令フォーマット：HALT命令は，1バイト長の命令です．

第1バイト	
OP	GR XR
1111_2	$11\ 11_2$

5.4　データ転送命令

TeC の CPU とメモリの間でデータ転送をする機械語命令を説明します.

5.4.1　LD(Load) 命令

Load は,「荷物等を積み込む」と言う意味の英語です. LD は，Load の綴を縮めたものです.

意味： 主記憶 (メモリ) からレジスタへデータを転送します. (メモリからレジスタへ値をコピーします. メモリの値は変化しません.)

フラグ： 変化しません.

ニーモニック： `LD GR,A[,XR]`
`GR` は，転送先レジスタを表します. `A` はデータの置いてあるメモリのアドレス (番地) を表します. `[,XR]` は，省略可能です. 将来，きちんと説明するまで，`[,XR]` は常に省略 (この部分を書かない) してください.

命令フォーマット： LD 命令は 2 バイトの長さを持ちます. 命令の各ビットは次の通りです.

第 1 バイト		第 2 バイト
OP	GR XR	
0001_2	$GR\ XR$	aaaa aaaa

- OP フィールド (4 ビット) は LD 命令を示す 0001_2 にします.
- GR フィールド (2 ビット) はデータを格納するレジスタを表します. GR フィールドの値は次の表のような意味を持ちます.

GR	意味
00	G0
01	G1
10	G2
11	SP

- XR フィールド (2 ビット) は，将来，詳しく説明するまでは常に”00” にします.
- 第 2 バイト (*aaaa aaaa*) は，A(番地) を 2 進数で格納します.

フローチャート： フローチャートでは，LD 命令を次のように描きます. `[` と `]` を書き忘れないように，注意して下さい.



使用例： LD 命令を使用したプログラムの例を示します. なお，ニーモニック欄の数値は 10 進数ですので，注意してください.

番地	機械語	ラベル	ニーモニック
00_{16}	$10_{16}\ 0A_{16}$		`LD GO,10`
02_{16}	$14_{16}\ 0B_{16}$		`LD G1,11`
04_{16}	FF_{16}		`HALT`

例の中で機械語は 16 進数で書いてありますが, 2 進数に変換すると 0 番地の命令は $0001\ 0000_2$, $0000\ 1010_2$ です. 上の命令フォーマットと対応を確認してください.

このプログラムをフローチャートで表現すると, 次のようになります.



命令表
TeC でどんな機械語命令が使用できるかは，本体のケースに張り付けた命令表 (または，付録 C の命令表) により確認することができます. これまでに出てきた NO，HALT，LD 命令について，命令表を確認して下さい. ニーモニックと命令の名前，命令のフォーマットが分かりますね.
その他にも，まだ習っていない情報がたくさん掲載されています. 一通り勉強が終わったら，この命令表だけで TeC を自由にプログラミングできるようになります.

RUN ランプが点滅したら
CPU が機械語命令として解釈できない命令を実行しようとしたことを表します. PC が解釈できなかった命令の次の番地を指した状態で CPU が停止します. PC の値からおかしな命令を見付けて訂正してください.

5.4.2 ST(Store) 命令

Store は,「倉庫に保管する」と言う意味の英語です．ST は，Store の綴を縮めたものです．

意味： レジスタからメモリへデータを転送します．(レジスタの値は変化しません．)

フラグ： 変化しません．

ニーモニック： `ST GR,A[,XR]`

GR は，転送元レジスタを表します．A はデータを保管するメモリのアドレス (番地) を表します．[,XR] は，省略可能です．将来，きちんと説明するまで，[,XR] は常に省略してください．

命令フォーマット： ST 命令は 2 バイトの長さを持ちます．命令のフォーマットは次の通りです．各フィールドの意味は，LD 命令と同様です．

第 1 バイト		第 2 バイト
OP	GR XR	
0010_2	$GR\ XR$	aaaa aaaa

フローチャート： フローチャートでは，ST 命令を次のように描きます．[と] を書き忘れないように，注意して下さい．



使用例： 10 番地のデータを 11 番地にコピーして停止するプログラムの例を示します．番地と機械語の欄はいつも 16 進数で書くので，この例から小さく 16 と書くのを止めます．なお，ニーモニック欄の数値は 10 進数です．

番地	機械語	ラベル	ニーモニック
00	10 0A		LD GO,10
02	20 0B		ST GO,11
04	FF		HALT

問題

以下では番地が 16 進数です．注意して下さい．

1. 11_{16} 番地のデータを 12_{16} 番地に，10_{16} 番地のデータを 11_{16} 番地にコピーして停止するプログラムを，上の使用例のように書きなさい．
2. 10_{16} 番地のデータと 11_{16} 番地のデータを交換して停止するプログラムを，上の使用例のように書きなさい．
3. これらのプログラムを，実際に TeC で実行して正しく動くことを確認しなさい．

プログラムの作成手順

ST 命令の使用例のように, 表を作成しながらプログラムを書きます．手順は次の通りです．

1. 表の枠を書く．
2. ニーモニックでプログラムを書く．プログラムに間違えが無いか，この段階で良く考える．
3. 機械語命令の長さを考えながら番地欄を記入する．(これは，機械的な作業．ミスをしないよう慎重に．)
4. 機械語欄を記入する．(これも，機械的な作業．ミスをしないように．)
5. 全部記入したら，TeC に打ち込んで実行してみる．
6. うまく動かなかったら，全てのステップを再度確認し間違えを探す．
7. うまく動いたら完成 !!! (うまく動いたかどうかの確認も慎重に！)

表に記入したら終わりではありません．TeC に打ち込んで実際に動くことを確認してください．動くはずなのに動かないことが多い (まず，動かないと考えた方が良い) です．
実際に動かし正しく動作することを確認できるまでは，プログラムは完成していません．

RUN ランプに注意

RUN ランプはプログラム実行中に点灯しています．プログラムが暴走し終了しない時は，RUN ランプが点灯したままになります．プログラム実行中でも，コンソールパネルを操作してメモリやレジスタの値を表示できるので，プログラムが暴走していることに気づかないことがあります．RUN ランプを常に気にするようにして下さい．

5.5　算術演算命令

5.5.1　ADD(Add) 命令

ADD は名前の通り，加算をする命令です．

意味： レジスタの値とメモリデータの和を計算し，結果を元のレジスタに格納します．(メモリの値は変化しません．)

フラグ： 計算の結果により変化します．

ニーモニック： `ADD GR,A[,XR]`
GR は，足し算の対象になるレジスタを表します．このレジスタの値とメモリの値が足し合わされ，結果がこのレジスタに格納されます．A と [,XR] の意味は LD 命令と同様です．

命令フォーマット： ADD 命令は 2 バイトの長さを持ちます．命令の各フィールドの意味は，LD 命令と同様です．

第 1 バイト		第 2 バイト
OP	GR XR	
0011_2	$GR\ XR$	aaaa aaaa

フローチャート： フローチャートでは，ADD 命令を次のように描きます．[と] を書き忘れないように，注意して下さい．



使用例： 7 番地のデータと 8 番地のデータの和を計算し，9 番地に格納するプログラムの例を示します．

番地	機械語	ラベル	ニーモニック
00	10 07		LD GO,7
02	30 08		ADD GO,8
04	20 09		ST GO,9
06	FF		HALT

このプログラムを実行するときは，予め 7 番地，8 番地に足し合わせるデータを格納しておく必要があります．例えば，7 番地に 1，8 番地に 2 を格納してからこのプログラムを実行すると，計算結果としてプログラムにより 9 番地に 3 が格納されます．

5.5.2　SUB(Subtract) 命令

Subtract は，「引き算をする」と言う意味の英語です．SUB は Subtract の綴を縮めたものです．

意味： レジスタの値とメモリデータの差を計算し，結果を元のレジスタに格納します．(メモリの値は変化しません．)

フラグ： 計算結果により変化します．

ニーモニック： `SUB GR,A[,XR]`
GR は，引き算の対象になるレジスタを表します．このレジスタの値からメモリの値が引かれ，結果がこのレジスタに格納されます．A と [,XR] の意味は LD 命令と同様です．

命令フォーマット： SUB 命令は 2 バイトの長さを持ちます．命令の各フィールドの意味は，LD 命令と同様です．

第 1 バイト		第 2 バイト
OP	GR XR	
0100_2	$GR\ XR$	aaaa aaaa

フローチャート： SUB 命令は次のように描きます．



使用例： 7 番地のデータから 8 番地のデータを引いた結果を 9 番地に格納するプログラムの例を示します．

番地	機械語	ラベル	ニーモニック
00	10 07		LD GO,7
02	40 08		SUB GO,8
04	20 09		ST GO,9
06	FF		HALT

このプログラムを実行するときは，予め 7 番地に引かれる数，8 番地に引く数を格納しておく必要があります．

問題

10_{16} 番地のデータと 11_{16} 番地のデータの和を 12_{16} 番地に，差を 13_{16} 番地に格納するプログラムを作成し，TeC で実行して正しく実行できたことを確かめなさい．

フラグ
いくつかのプログラムを作成して動かして見ました．そのとき，TeC の C，S，Z ランプが点灯したり消灯したりしたのが分かったでしょうか？これらのランプは，対応したフラグ (Flag:旗) の値を表示しています．どんなときランプが点灯し (フラグが 1 になり)，どんなときランプが消灯した (フラグが 0 になった) のか説明します．

C(Carry) フラグ
Carry は「桁を繰り上げる」と言う意味です．C フラグは，計算中に 8bit の最上位桁からの「桁上がり」や，8bit の最上位桁での「桁借り」が発生したことを表します．つまり，計算結果が 255 を超えてしまったことや，0 より小さくなったことを表します．C フラグは，計算値が符号無しと考えたときのオーバーフローを表しています．

S(Sign) フラグ
Sign は，**「符号」**を意味します．計算の結果を符号付き 2 進数として解釈した場合，負の値になることを表します．つまり，計算結果の 8bit の最上位ビットが 1 のとき 1 になります．

Z(Zero) フラグ
Zero は，名前の通り計算の結果がゼロになったことを表します．つまり，計算結果の 8bit の全てのビットが 0 のとき 1 になります．

これら三つのフラグは，命令表で「フラグ変化」の欄に「○」印が付いている演算命令を実行する度に，計算結果によって変化します．つまり，直前の演算の結果を反映しています．これまでに出てきた命令では，ADD，SUB がフラグを変化させる命令です．

うまく動かない場合
プログラムが正しく動かない人は，1 命令ずつ実行しながら，メモリやレジスタの値が予想通りに変化するか調べて下さい．1 命令ずつ実行する方法は，4.2.2 中の「プログラムのステップ実行」に説明してあります．
面倒くさいと思わないで，地道に問題点を捜しましょう．問題点を見つけるコツが分かってきたら，演習がすごく楽になります．慣れるまで少し辛抱して下さい．

5.6 ジャンプ命令

ノイマン型コンピュータの特徴は逐次実行です．命令を番地の順番に順に実行します．プログラムの実行が進んで行く流れを**「プログラムの流れ」**と言います．プログラムの流れは PC(Program Counter) によって管理され，通常は PC が順次増加します．

しかし，プログラムの同じ部分を繰り返したり，条件によりプログラムの動きを変更する目的で，流れを別の場所に飛ばす (Jump) ことも必要です．

ジャンプ命令はこのような目的で PC の値を変更し，プログラムの流れを別の番地へ飛ばすものです．プログラムの流れは飛んで行った先にあるプログラムを順番に実行する流れになり，もとに戻って来ることはありません．

5.6.1 JMP(Jump) 命令

この命令を実行するとプログラムの流れが，必ず指定の番地にジャンプします．

意味： プログラムの流れをジャンプさせます．

フラグ： 変化しません．

ニーモニック： `JMP A[,XR]`
A はジャンプ先のアドレス (番地) を表します．`[,XR]` は，省略可能です．将来，きちんと説明するまで，`[,XR]` は常に省略 (この部分を書かない) してください．

命令フォーマット： JMP 命令は 2 バイトの長さを持ちます．GR フィールドは必ず 00 にします．第 2 バイトがジャンプ先のアドレスを示します．

第 1 バイト		第 2 バイト
OP	GR XR	
1010_2	00 XR	aaaa aaaa

フローチャート： JMP 命令はフローチャートの線に対応します．次の命令で詳しく説明します．

使用例： 00 番地の命令を，いつまでも繰り返す (止まらない) プログラムの例を示します．

番地	機械語	ラベル	ニーモニック
00	30 04		ADD GO,4
02	A0 00		JMP 0

ラベル

JMP 命令の使用例では，ジャンプする先のアドレスをニーモニックの中に直接書きました．もしも，プログラムに変更があり，ジャンプする先のアドレスが変化したらどうでしょうか．プログラムに直接アドレスが書いてあると，それを書き直す必要が生じます．

番地が変化してもニーモニックで書いたプログラムを変更しなくて良いように，アドレスの代わりにその場所に付けた記号を使うと便利です．

場所に付けた記号のことをラベルと言います．ラベルを用いてプログラムを書き換えると次のようになります．

番地	機械語	ラベル	ニーモニック
00	30 04	**LOOP**	ADD　G0,4
02	A0 00		JMP　**LOOP**

データの定義 (DC 命令)

ラベルを用いることにより，ニーモニックで書いたプログラムの取扱いが非常に便利になりました．しかし，データの部分はニーモニックに記述できていません．これでは，ニーモニックだけでプログラムの全体像が理解できません．

そこで，データを記述するための “DC(Define Constant)” 命令を追加します．DC 命令は，機械語の代わりにオペランドで指定した値のデータを生成します．また，DC 命令の行にラベルを付けることによりデータもラベルで参照できるようになります．

次は，JMP 命令の使用例を DC 命令を用いて書き直したものです．4 番地に値 “1” のデータがあることが，うまく記述できました．

番地	機械語	ラベル	ニーモニック
00	30 04	LOOP	ADD　G0,**ONE**
02	A0 00		JMP　LOOP
04	01	**ONE**	DC　1

領域の定義 (DS 命令)

DS 命令は DC 命令に良く似た命令です．DC 命令は機械語命令の代わりにデータを生成しました．DS 命令は結果を格納するための領域を生成します．指定された数値は，領域の大きさになります．

5.6.2　JZ(Jump on Zero) 命令

Z フラグが 1 のとき (計算結果が 0 だったとき) だけジャンプします．

意味： Z フラグが 1 ならジャンプします．

フラグ： 変化しません．

ニーモニック： `JZ A[,XR]`

JMP 命令と同様です．

命令フォーマット： JZ 命令は 2 バイトの長さを持ちます．GR フィールドは必ず 01 にします．第 2 バイトがジャンプ先のアドレスを示します．

第 1 バイト		第 2 バイト
OP	GR XR	
1010_2	01 XR	aaaa aaaa

フローチャート： JZ 命令のフローチャートは，次のように様々な描き方が考えられます．例には，JMP 命令と組合せて菱形一つに対応させたものもありますが，こだわる必要はありません．場合によって，柔軟にアレンジして下さい．



使用例： 結果がゼロだったら停止するプログラムの例を示します．定数の “1” を使用するために DC 命令でメモリの 07_{16} 番地にデータ 01_{16} を置きました．定数を使用するためにはメモリ上に定数データを置く必要があります．

番地	機械語	ラベル	ニーモニック	
00	30 07	LOOP	ADD	G0,ONE
02	A4 06		JZ	STOP
04	A0 00		JMP	LOOP
06	FF	STOP	HALT	
07	01	ONE	DC	1

5.6.3 JC(Jump on Carry) 命令

C フラグが 1 のときだけジャンプします.

意味： C フラグが 1 ならジャンプします.

フラグ： 変化しません.

ニーモニック： `JC A[,XR]`
JMP 命令と同様です.

命令フォーマット： JC 命令は 2 バイトの長さを持ちます. GR フィールドは必ず 10 にします. 第 2 バイトがジャンプ先のアドレスを示します.

第 1 バイト		第 2 バイト
OP	GR XR	
1010_2	10 XR	aaaa aaaa

フローチャート： JC 命令のフローチャートは，JZ 命令と同様な考えで描きます. JZ 命令を参考にして下さい.

5.6.4 JM(Jump on Minus) 命令

S フラグが 1 のとき (計算結果が負だったとき) だけジャンプします. 名前が，JS 命令ではなく JM 命令になっているので注意して下さい.

意味： S フラグが 1 ならジャンプします.

フラグ： 変化しません.

ニーモニック： `JM A[,XR]`
JMP 命令と同様です.

命令フォーマット： JM 命令は 2 バイトの長さを持ちます. GR フィールドは必ず 11 にします. 第 2 バイトがジャンプ先のアドレスを示します.

第 1 バイト		第 2 バイト
OP	GR XR	
1010_2	11 XR	aaaa aaaa

フローチャート： JM 命令のフローチャートも，JZ 命令と同様な考えで描きます. JZ 命令を参考にして下さい.

条件判断 1
ジャンプ命令を用いて，条件判断があるプログラムを作ることができます. 次のフローチャートとプログラムは，計算結果がゼロだった場合だけ処理をするものです. このように，ある条件の場合だけ処理をするプログラムを作ることができます.



条件判断 2
条件によって，二つの処理のどちらかを選んで実行するプログラムを作ることができます. 次のフローチャートとプログラムは，計算結果がゼロだった場合は「処理 1」を，ゼロ以外だった場合は「処理 2」を実行します.



繰り返し処理 1

次のように，ジャンプ命令を用いて同じ操作を繰り返すプログラムを作ることができます．次のフローチャートとプログラムは，点線部分を 10 回繰り返すためのものです．




```
        LD    G0,N
LOOP
        :
        処理
        :
        SUB   G0,ONE
        JZ    STOP
        JMP   LOOP
STOP    HALT
N       DC    10
ONE     DC    1
```

繰り返し処理 2

計算結果によって繰り返し回数が決まる場合等，繰り返し回数がゼロの場合も考慮しなければならないことがあります．そのような場合は，条件判断を前に移動すると，うまく処理できます．次のフローチャートとプログラムは，点線部分を N 回繰り返すためのものです．条件判断の前の引算は，G0 の値でフラグを強制的に変化させるためのものです．




```
        LD    G0,N
LOOP    SUB   G0,ZERO
        JZ    STOP
        :
        処理
        :
        JMP   LOOP
STOP    HALT
N       DS    1
ZERO    DC    0
```

例題 5-1 絶対値を求める．

問題： N 番地のデータの絶対値を計算し，M 番地に格納するプログラムを作りなさい．

考え方： 値が正なのか負なのかは，値からゼロを (SUB 命令で) 引き，そのときのフラグの変化で調べます．

負の値の絶対値は，ゼロからその値を引くことで求めることができます．

解答： 次のフローチャートのようなプログラムを作ります．



注意：[N番地]は、N番地に格納されているデータのこと

番地	機械語	ラベル	ニーモニック	
00	10 10	START	LD	G0,N
02	40 0F		SUB	G0,ZERO
04	AC 08		JM	L1
06	A0 0C		JMP	L2
08	10 0F	L1	LD	G0,ZERO
0A	40 10		SUB	G0,N
0C	20 11	L2	ST	G0,M
0E	FF		HALT	
0F	00	ZERO	DC	0
10	FF	N	DC	-1
11	00	M	DS	1

解説： プログラムの内容を説明します．

(1) G0 レジスタに値をロードします．

(2) G0 レジスタから 0 を引きます．値は変化しませんが，計算結果によりフラグが変化します．

(3) JM 命令は S フラグの値によりジャンプします．

(4), (5) この部分は，値が負だった場合のみ実行されます．$0 - [N$ 番地$]$ を計算し，値の絶対値を G0 に求めます．

(6) G0 の値を M 番地に格納します．

例題 5-2 $1+2+3+...+10$ を計算する．

問題： $1+2+3+...+10$ を計算し結果を SUM 番地に求めるプログラムを作りなさい．

考え方： G0 以外に，G1，G2，SP の三つのレジスタが自由に使用できるので，それぞれのレジスタに足す値を記憶する，合計を記憶する等の役割りを分担させ，繰り返し (10 回繰り返す) をうまく利用すると計算できます．

TeC にはレジスタの値同士を足し算する命令がありません．一方のレジスタ値をメモリに格納してから，足し算する必要があります．

解答： G0 を繰り返し回数のカウント，G1 をレジスタに足す数の記憶 (1,2,3,...,10)，G2 を合計の記憶に使用することにしました．フローチャートとプログラムは次のようになります．




```
        LD    G0,N
        LD    G1,ONE
        LD    G2,ZERO
LOOP    ST    G1,TMP
        ADD   G2,TMP
        ADD   G1,ONE
        SUB   G0,ONE
        JZ    STOP
        JMP   LOOP
STOP    ST    G2,SUM
        HALT
N       DC    10
ONE     DC    1
ZERO    DC    0
TMP     DS    1
SUM     DS    1
```

C フラグの詳しい説明

C フラグは，計算中に，8bit の最上位桁からの「桁上がり (Carry)」が発生したことや，8bit の最上位桁での「桁借り (Borrow)」が発生したことを表します．例えば，次の計算では最上位桁からの「桁上がり」は発生しませんので，C フラグは “0” になります．

	$0000\ 0001_2$	(1)
+	$0000\ 0011_2$	(3)
$\boxed{0}$	$0000\ 0100_2$	(4)
C		

次の計算では最上位桁からの「桁上がり」が発生し，C フラグが “1” になります．C フラグは，計算値が符号無しと考えたときのオーバーフローを表しています．

		符号無	符号付
	$1111\ 1111_2$	(255)	(-1)
+	$0000\ 0001_2$	(1)	(+1)
$\boxed{1}$	$0000\ 0000_2$	(オーバ	(0)
C		フロー)	

また，引き算でも C フラグが 1 になることがあります．例えば，次のように小さな数から大きな数を引いた場合です．機械的に引き算をすると，最上位桁で桁借りが発生します．この時も，C フラグが 1 になります．

		符号無	符号付
	$0110\ 0100_2$	(100)	(+100)
−	$0110\ 1110_2$	(110)	(+110)
$\boxed{1}$	$1111\ 0110_2$	(オーバ	(-10)
C		フロー)	

このように引き算の場合は，C フラグは，計算が符号無しと考えたときの，負へのオーバーフローを表しています．

5.6.5　JNZ(Jump on Not Zero) 命令

Z フラグが 0 のとき (計算結果がゼロ以外だったとき) だけジャンプします.
(**注意：**TeC7 で新たに追加された命令です．TeC6 では使用できません．)

意味： Z フラグが 0 ならジャンプします.

フラグ： 変化しません.

ニーモニック： `JNZ A[,XR]`
JMP 命令と同様です.

命令フォーマット： JNZ 命令は 2 バイトの長さを持ちます．GR フィールドは必ず 01 にします．第 2 バイトがジャンプ先のアドレスを示します.

第 1 バイト		第 2 バイト
OP	GR XR	
1011_2	01 XR	aaaa aaaa

フローチャート： JNZ 命令のフローチャートも，JZ 命令と同様な考えで描きます．JZ 命令を参考にして下さい.

使用例： JZ 命令の**使用例**で紹介した「結果がゼロだったら停止するプログラムの例」を JNZ 命令を使用して書き直しました．JNZ 命令を使用したほうが随分シンプルになります.

番地	機械語	ラベル	ニーモニック	
00	30 05	LOOP	ADD	GO,ONE
02	B4 00		JNZ	LOOP
04	FF		HALT	
05	01	ONE	DC	1

5.6.6　JNC(Jump on Not Carry) 命令

C フラグが 0 のときだけジャンプします.
(**注意：**TeC7 で新たに追加された命令です．TeC6 では使用できません．)

意味： C フラグが 0 ならジャンプします.

フラグ： 変化しません.

ニーモニック： `JNC A[,XR]`
JMP 命令と同様です.

命令フォーマット： JNC 命令は 2 バイトの長さを持ちます．GR フィールドは必ず 10 にします．第 2 バイトがジャンプ先のアドレスを示します.

第 1 バイト		第 2 バイト
OP	GR XR	
1011_2	10 XR	aaaa aaaa

フローチャート： JNC 命令のフローチャートも，JZ 命令と同様な考えで描きます．JZ 命令を参考にして下さい.

5.6.7　JNM(Jump on Not Minus) 命令

S フラグが 0 のとき (計算結果がゼロか正だったとき) だけジャンプします．名前が，JNS 命令ではなく JNM 命令になっているので注意して下さい.
(**注意：**TeC7 で新たに追加された命令です．TeC6 では使用できません．)

意味： S フラグが 0 ならジャンプします.

フラグ： 変化しません.

ニーモニック： `JNM A[,XR]`
JMP 命令と同様です.

命令フォーマット： JNM 命令は 2 バイトの長さを持ちます．GR フィールドは必ず 11 にします．第 2 バイトがジャンプ先のアドレスを示します.

第 1 バイト		第 2 バイト
OP	GR XR	
1011_2	11 XR	aaaa aaaa

フローチャート： JNM 命令のフローチャートも，JZ 命令と同様な考えで描きます．JZ 命令を参考にして下さい.

問題

1. かけ算プログラム
   N 番地と M 番地の値のかけ算を計算し，L 番地に結果を格納するプログラムを作りなさい.
2. かけ算プログラムの改良
   前のプログラムを [N] または，[M] がゼロのときでも正しく動くように改良しなさい.

5.7 比較命令

5.7.1 CMP(Compare) 命令

Compare は,「比較をする」と言う意味の英語です．CMP は Compare の綴を縮めたものです．比較は引き算により行います．レジスタの値からメモリの値を引いて，その結果によりフラグを変化させます．引き算結果そのものは，どこにも格納しないで捨てます．

意味： レジスタの値とメモリデータを比較します．引き算の結果により，フラグだけが変化します．

フラグ： 計算結果により変化します．

ニーモニック： `CMP GR,A[,XR]`
`GR` で指定されたレジスタの値と，メモリの値が比較されます．`A` と `[,XR]` の意味は LD 命令と同様です．

命令フォーマット： CMP 命令は 2 バイトの長さを持ちます．各フィールドの意味は，LD 命令と同様です．

第 1 バイト		第 2 バイト
OP	GR XR	
0101_2	$GR\ XR$	aaaa aaaa

フローチャート： CMP 命令は次のように描きます．



使用例： A，B 二つのデータで大きい方を選んで，C に格納するプログラムの例です．

番地	機械語	ラベル	ニーモニック	
00	10 0D		LD	G0,A
02	50 0E		CMP	G0,B
04	AC 08		JM	LB
06	A0 0A		JMP	LC
08	10 0E	LB	LD	G0,B
0A	20 0F	LC	ST	G0,C
0C	FF		HALT	
0D	64	A	DC	100
0E	C8	B	DC	200
0F	00	C	DS	1

例題 5-3 割算を計算する．

問題： M 番地の値を N 番地の値で割り，商を K 番地，余りを L 番地に求めるプログラムを作りなさい．

考え方： TeC には割算命令がありません．割算は引き算の繰り返しで計算できます．割られる数から，割る数を引くことを繰り返します．引くことができなくなったら終了します．このとき，引いた回数が商，引いた結果が余りになります．

解答： フローチャートとプログラムは次のようになります．




```
        LD    G0,M
        LD    G1,ZERO
LOOP    CMP   G0,N
        JM    STOP
        SUB   G0,N
        ADD   G1,ONE
        JMP   LOOP
STOP    ST    G1,K
        ST    G0,L
        HALT
M       DC    100
N       DC    10
ONE     DC    1
ZERO    DC    0
K       DS    1
L       DS    1
```

解説： CMP 命令と JM 命令で G0 が N 番地のデータより大きいか確認してから，引き算を実行します．

問題

以前の，絶対値を求めるプログラムを CMP 命令を用いて書き直しなさい．

5.8　シフト (桁ずらし) 命令

データの 2 進数を左右に桁移動する命令をシフト命令と言います．TeC は以下に説明する 4 種類のシフト命令を持っています．

5.8.1　SHLA(Shift Left Arithmetic) 命令

左算術 (算術= Arithmetic) シフトと言います．レジスタのデータを左方向に 1 ビットずらします．右側から常に “0” が入力されます．左にはみ出したビットは，C フラグの値になります．



意味： レジスタの値を，左に 1 ビット，算術シフトします．

フラグ： C フラグは上の説明のように，S，Z フラグは計算結果により変化します．

ニーモニック： `SHLA GR`
GR が，シフトされるレジスタを表します．GR の値とレジスタの対応は，LD 命令等と同じです．

命令フォーマット： SHLA 命令は 1 バイトの命令です．XR フィールドは必ず 00 にします．

第 1 バイト	
OP	GR XR
1001_2	GR 00

フローチャート： SHLA 命令は次のように描くことにします．なお，この書き方は，Java 言語や C 言語のシフト演算子を真似たものです．



使用例： A 番地のデータを 2 ビット左にシフトし，B 番地に格納する例です．左に 2 ビットシフトすることは，× 4 を計算したのと同じ結果になります．

番地	機械語	ラベル	ニーモニック	
00	10 07		LD	G0,A
02	90		SHLA	G0
03	90		SHLA	G0
04	20 08		ST	G0,B
06	FF		HALT	
07	01	A	DC	1
08	00	B	DS	1

5.8.2　SHLL(Shift Left Logical) 命令

左論理 (論理=Logical) シフトと言います．この命令は SHLA と全く同じ動作をします．(左シフトでは，算術と論理の差はない)

意味： レジスタの値を，左に 1 ビット，論理シフトします．

フラグ： SHLA 命令と同様です．

ニーモニック： `SHLL GR`
SHLA 命令と同様です．

命令フォーマット： SHLL 命令は 1 バイトの命令です．XR フィールドは必ず 01 にします．

第 1 バイト	
OP	GR XR
1001_2	GR 01

フローチャート： SHLL 命令は次のように描くことにします．SHLA 命令と全く同じ動作をするので，フローチャートの描き方も，SHLA 命令と全く同じです．



使用例： SHLA 命令と同じ動作をする命令なので，省略します．

5.8.3 SHRA(Shift Right Arithmetic) 命令

右算術 (算術＝ Arithmetic) シフトと言います．レジスタのデータを右方向に 1 ビットずらします．左側からシフト前のデータの最上位ビット (符号ビット) と同じ値が入力されます．右にはみ出したビットは，C フラグの値になります．



意味： レジスタの値を，右に 1 ビット，算術シフトします．

フラグ： SHLA 命令と同様です．

ニーモニック： `SHRA GR`
SHLA 命令と同様です．

命令フォーマット： SHRA 命令は 1 バイトの命令です．XR フィールドは必ず 10 にします．

第 1 バイト	
OP	GR XR
1001_2	*GR* 10

フローチャート： SHRA 命令は次のように描くことにします．



使用例： A，B 番地のデータを 1 ビット右にシフトし，C，D 番地に格納します．右 1 ビットシフトと，÷ 2 は同じ結果になります．算術シフトの場合は，負の数でも正しく 1/2 になります．

番地	機械語	ラベル	ニーモニック	
00	10 0B		LD	G0,A
02	92		SHRA	G0
03	20 0D		ST	G0,C
05	14 0C		LD	G1,B
07	96		SHRA	G1
08	24 0E		ST	G1,D
0A	FF		HALT	
0B	08	A	DC	8
0C	F8	B	DC	-8
0D	00	C	DS	1
0E	00	D	DS	1

5.8.4 SHRL(Shift Right Logical) 命令

右論理 (論理＝ Logical) シフトと言います．レジスタのデータを右方向に 1 ビットずらします．左側から “0” が入力されます．右にはみ出したビットは，C フラグの値になります．



意味： レジスタの値を，右に 1 ビット，論理シフトします．

フラグ： SHLA 命令と同様です．

ニーモニック： `SHRL GR`
SHLA 命令と同様です．

命令フォーマット： SHRL 命令は 1 バイトの命令です．XR フィールドは必ず 11 にします．

第 1 バイト	
OP	GR XR
1001_2	*GR* 11

フローチャート： SHRL 命令は次のように描くことにします．



使用例： A，B 番地のデータを 1 ビット右にシフトし，C，D 番地に格納します．右 1 ビットシフトと，÷ 2 は同じ結果になります．論理シフトの場合は，128 以上の大きな数でも正しく 1/2 の値になります．

番地	機械語	ラベル	ニーモニック	
00	10 0B		LD	G0,A
02	93		SHRL	G0
03	20 0D		ST	G0,C
05	14 0C		LD	G1,B
07	97		SHRL	G1
08	24 0E		ST	G1,D
0A	FF		HALT	
0B	7F	A	DC	127
0C	80	B	DC	128
0D	00	C	DS	1
0E	00	D	DS	1

例題 5-4 シフトの動作確認

問題： 4 種類のシフト命令について，G0 レジスタの値をシフトしながら動作を確認しなさい．

プログラム： 次のプログラムを作成します．

番地	機械語	ラベル	ニーモニック
00	90	L0	SHLA G0
01	FF		HALT
02	A0 00		JMP L0

フローチャート： フローチャートで描く場合は，次のように HALT を表現することにします．



実行方法： 次の手順で実行します．

1. PC を 00_{16} にする．
2. G0 レジスタに適当なデータを書き込む．
3. ロータリースイッチを G0 に合わせたまま，RUN ボタンを押す．
4. G0 の値がシフトされたことが，データランプで確認できる．
5. 再度，RUN ボタンを押す．
6. G0 の値が更にシフトされる．

一度 RUN ボタンを押すと HALT 命令で停止します．そのとき，PC は JMP 命令を指していますので，再度 RUN ボタンを押すと，もう一度，最初からプログラムが実行されます．

例題 5-5 ビットの回転

問題： 次の図のように，G0 レジスタの値を 1 ビット左回転するプログラムを作りなさい．



考え方： TeC には，ビットを回転する命令がありません．次の手順で目的を達成します．

1. シフト命令で G0 を 1 ビットシフトする．
2. はみ出したビットが C フラグに保存されているので，C フラグを調べる．
3. C フラグが 1 だった場合は，G0 最下位ビットを 1 にする．



プログラム： プログラムにすると次のようになります．例題 5-4 と同様に，最後に JMP 命令を追加しました．

番地	機械語	ラベル	ニーモニック
00	91	L0	SHLL G0
01	A8 05		JC L1
03	A0 07		JMP L2
05	30 0A	L1	ADD G0,ONE
07	FF	L2	HALT
08	A0 00		JMP L0
0A	01	ONE	DC 1

実行方法： 例題 5-4 と同様です．実行する度に，データランプの表示が回転します．

例題 5-6 シフトを用いた高速乗算

問題： シフトを用いて，A の 10 倍の値を B に求めるプログラムを作りなさい．

考え方： TeC には，かけ算命令がありません．そのため，繰り返しでかけ算をする方法を，以前，紹介しました．ここでは，繰り返しによる方法より高速なかけ算を紹介します．

× 10 は，次の式のように変形できます．

$$B = A \times 10 = A \times 8 + A \times 2$$

× 2 は，1 ビット左にシフトすることと同じです．× 8 は，3 ビット左にシフトすることと同じです．シフトした結果同士を足し合わせることにより，10 倍を計算することができます．

プログラム： プログラムにすると次のようになります．TMP は，一時的に A の 2 倍の値を記憶するために使用します．

番地	機械語	ラベル	ニーモニック	
00	10 0C		LD	G0,A
02	90		SHLA	G0
03	20 0E		ST	G0,TMP
05	90		SHLA	G0
06	90		SHLA	G0
07	30 0E		ADD	G0,TMP
09	20 0D		ST	G0,B
0B	FF		HALT	
0C	03	A	DC	3
0D	00	B	DS	1
0E	00	TMP	DS	1

問題

1. A の値の 7 倍を B に求めるプログラムを作りなさい．
2. A のデータが符号無し 2 進数の場合，A の 1/4 を B に求めるプログラムを作りなさい．

5.9 論理演算命令

5.9.1 AND(Logical AND) 命令

AND 命令は，レジスタとメモリデータのビット毎の論理積 (AND) を計算します．ビット毎の論理積とは，次の図のように，レジスタとメモリの対応するビット同士の論理積の計算のことです．計算結果はレジスタに格納されます．



意味： ビット毎の論理積を計算します．結果は元のレジスタに格納します．(メモリの値は変化しません．)

フラグ： C フラグは常に 0 になります．S，Z フラグは計算の結果により変化します．

ニーモニック： `AND GR,A[,XR]`

`GR` は，計算の対象になるレジスタを表します．このレジスタの値とメモリの値の論理積が計算されます．結果はこのレジスタに格納されます．`A` と `[,XR]` の意味は LD 命令と同様です．

命令フォーマット： AND 命令は 2 バイトの長さを持ちます．各フィールドの意味は，LD 命令と同様です．

第 1 バイト		第 2 バイト
OP	GR XR	
0110_2	$GR\ XR$	aaaa aaaa

フローチャート： AND 命令は次のように描くことにします．これも，Java 言語や C 言語の演算子を真似したものです．



使用例： A 番地のデータと B 番地のデータの論理積を計算し，C 番地に格納するプログラムの例を示します．

番地	機械語	ラベル	ニーモニック	
00	10 07		LD	G0,A
02	60 08		AND	G0,B
04	20 09		ST	G0,C
06	FF		HALT	
07	63	A	DC	63H
08	0F	B	DC	0FH
09	00	C	DS	1



AND 命令の応用： AND 命令は，データの特定のビットを 0 にクリアしたり，データの特定のビットの 1/0 を確かめたり，データの一部のビットだけを取り出したりするために使用できます．

1. G0 の最下位ビット (LSB) がゼロなら L1 へジャンプする．

ラベル	ニーモニック
	...
	AND G0,ONE
	JZ L1
	...
L1	...
	...
ONE	DC 01H

   01_{16} との AND の結果が 00_{16} になることから，LSB が 0 だったことが分かります．

2. G0 の，b3,b2 の 2 ビットを右詰めにして取り出す．

ラベル	ニーモニック
	...
	AND G0,MSK
	SHRL G0
	SHRL G0
	...
MSK	DC 0CH



3. G0 の値を 8 の倍数になるように切り捨てる．

ラベル	ニーモニック
	...
	AND G0,MSK
	...
MSK	DC F8H

   下 3 ビットをゼロにすると，8 の倍数になります．

4. G0 の値を 8 で割った余りを求める．

ラベル	ニーモニック
	...
	AND G0,MSK
	...
MSK	DC 07H

   下 3 ビットだけを取り出すと，8 で割った余りになります．

   このように，2 の累乗 (2,4,8,16,...) で割った商や余りは，シフトや論理演算で簡単に計算できます．また，2 の累乗の倍数に切捨てたり，切り上げたりする計算も，論理演算を使うと簡単にできます．

16 進数の表記
ニーモニック中で数値を書くとき，16 進数で書き表したいことがあります．そのときは，前のプログラム中の DC 命令のように，数値の後ろに “H” を付けます．

5.9.2 OR(Logical OR) 命令

OR 命令は，レジスタとメモリデータの対応するビット同士の論理和 (OR) を計算します．計算結果はレジスタに格納されます．

意味： ビット毎の論理和を計算します．結果は元のレジスタに格納します．(メモリの値は変化しません．)

フラグ： C フラグは常に 0 になります．S，Z フラグは計算の結果により変化します．

ニーモニック： `OR GR,A[,XR]`
`GR` は，計算の対象になるレジスタを表します．このレジスタの値とメモリの値の論理和が計算されます．結果はこのレジスタに格納されます．`A` と `[,XR]` の意味は LD 命令と同様です．

命令フォーマット： OR 命令は 2 バイトの長さを持ちます．各フィールドの意味は，LD 命令と同様です．

第 1 バイト		第 2 バイト
OP	GR XR	
0111_2	$GR\ XR$	aaaa aaaa

フローチャート： OR 命令は次のように描くことにします．これも，Java 言語や C 言語の演算子を真似したものです．



OR 命令の応用： OR 命令は，データの特定のビットを 1 にするために使用できます．次の例は，G0 の上位 4 ビットを全て 1 にします．

ラベル	ニーモニック
	...
	LD GO,DATA
	OR GO,MSK
	...
DATA	DC OAAH
MSK	DC OFOH



5.9.3 XOR(Logical XOR) 命令

XOR 命令は，レジスタとメモリデータの対応するビット同士の排他的論理和 (XOR) を計算します．計算結果はレジスタに格納されます．

意味： ビット毎の排他的論理和を計算します．結果は元のレジスタに格納します．(メモリの値は変化しません．)

フラグ： C フラグは常に 0 になります．S，Z フラグは計算の結果により変化します．

ニーモニック： `XOR GR,A[,XR]`
`GR` は，計算の対象になるレジスタを表します．このレジスタの値とメモリの値の排他的論理和が計算されます．結果はこのレジスタに格納されます．`A` と `[,XR]` は LD 命令と同様です．

命令フォーマット： XOR 命令は 2 バイトの長さを持ちます．各フィールドの意味は，LD 命令と同様です．

第 1 バイト		第 2 バイト
OP	GR XR	
1000_2	$GR\ XR$	aaaa aaaa

フローチャート： XOR 命令は次のように描くことにします．



XOR 命令の応用： XOR 命令を使用してデータの特定ビットの 0/1 を反転できます．TeC には NOT 命令がないので XOR 命令で代用します．次は G0 の上位 4 ビットの 0/1 を反転します．

ラベル	ニーモニック
	...
	LD GO,DATA
	XOR GO,MSK
	...
DATA	DC OAAH
MSK	DC OFOH



5.10　アドレッシングモード

LD，ST，ADD，SUB，CMP，AND，OR，XOR，JMP，JZ，JC，JM，JNZ，JNC，JNM 命令は，どれも次のような同じフォーマットでした．これまで，これらの命令の XR 部分は 00 にしてきました．

第 1 バイト		第 2 バイト
OP	GR XR	
OP	$GR\ XR$	aaaa aaaa

XR に 00 以外を指定することにより，メモリアドレス表現方法を変更することができます．このアドレス表現方法のことをアドレッシングモードと呼びます．TeC で使用できるアドレッシングモードは，XR の値により次の 4 種類があります．

XR	意味	
00	ダイレクトモード	(直接モード)
01	G1 インデクスドモード	(G1 指標モード)
10	G2 インデクスドモード	(G2 指標モード)
11	イミディエイトモード	(即値モード)

(同様な表が命令表にも掲載されているので，そちらも確認してください．)

以下では LD 命令と ST 命令を例に，4 種類のアドレッシングモードを説明します．ここでは説明しませんが，ADD，SUB，CMP，AND，OR，XOR 命令でも，LD 命令や ST 命令と同様にアドレッシングモードが使用できます．

5.10.1　ダイレクト (直接) モード

これまで使用してきたのは，このモードです．命令の第 2 バイトがメモリアドレスを直接に表します．

ニーモニックでは次のように書きました．ここで，A，B は，データを置いたメモリのアドレスです．

```
LD   G0,A
ST   G0,B
```

フローチャートでは，次のように描きました．



5.10.2　インデクスド (指標) モード

G1 または G2 レジスタの値と，命令の第 2 バイトの値の合計がメモリアドレスを表します．XR の値により，G1，G2 どちらのレジスタを使用するか決まります．G0 レジスタは使用できないので注意してください．

このモードは，配列データをアクセスするとき使用できます．また，ジャンプ命令でも使用できます．ジャンプ命令でインデクスドモードを使用する方法は，「6.7.3 ジャンプテーブル」で説明します．

ニーモニックでは次のように書きます．このプログラムは，$A + G1$ 番地のデータを G0 に読込み，$A + G2$ 番地に格納しています．

```
LD   G0,A,G1
ST   G0,A,G2
```

フローチャートでは，次のように描きます．



5.10.3 イミディエイト (即値) モード

第 2 バイトがデータそのものになるモードです. ST 命令やジャンプ命令では使用できません. (これらの命令で XR を 11 にすると，CPU が命令を実行しようとしたとき，命令コードのエラーになり RUN ランプが点滅します.)

ニーモニックでは次のように書きます.

```
LD   G0,#1
LD   G1,#A
```

最初の LD 命令は，G0 を 1 にします. 次の LD 命令は，G1 を A の番地 (内容ではない) にします. フローチャートでは，次のように描きます.



5.10.4 アドレッシングモードの使用例

A 番地のデータで，B 番地からの 10 バイトの配列を初期化するプログラムの例を示します.

G1，G2 レジスタを初期化する部分，G1，G2 レジスタの値を ± 1 する部分にイミディエイトモードを使用しました.

また，10 回の繰り返しで 10 バイトの領域を初期化できるよう，ST 命令の格納アドレスが G2 の値で 1 番地ずつずれていきます. ずらすためにインデクスモードを使用しました.

番地	機械語	ラベル	ニーモニック
00	10 11		LD G0,A
02	17 0A		LD G1,#10
04	1B 00		LD G2,#0
06	22 12	LOOP	ST G0,B,G2
08	3B 01		ADD G2,#1
0A	47 01		SUB G1,#1
0C	A4 10		JZ STOP
0E	A0 06		JMP LOOP
10	FF	STOP	HALT
11	AA	A	DC 0AAH
12	00 00	B	DS 10
14	00 00		
16	00 00		
18	00 00		
1A	00 00		

次に，このプログラムのフローチャートを示します. イミディエイトモードとインデクスドモードを，フローチャート上でどのように表現しているか，よく確認してください.



問題

1. これまでに出てきたプログラムを，イミディエイトモードを使用して書き換えなさい．
2. A 番地からの 5 バイトのデータの合計を，B 番地に求めるプログラムを作りなさい．
3. A 番地からの 5 バイトのデータを，B 番地からの 5 バイトにコピーするプログラムを作りなさい．

5.11　入出力

図 5.2 に示した TeC の構成のうち主記憶は，LD, ST 命令等を使用することでアクセスできることが分かりました．ここで説明する入出力 (Input Output= 略して I/O) 命令は，入出力インターフェース回路 (図 5.2「TeC の構成」を参照) をアクセスするための命令です．読み込みを行う IN 命令，書き込みを行う OUT 命令の 2 種類があります．

5.11.1　I/O マップ

いくつか存在するインターフェース回路の中の，どれをアクセスするかは I/O アドレス (メモリアドレスとは別のもの) により指定します．TeC の I/O アドレスは 0_{16}〜F_{16} までの 16 番地です．I/O アドレスの一覧を「I/O マップ」と言います．「I/O マップ」は次の表のようなものです．

I/O マップ		
番地	Read	Write
0	データスイッチ	ブザー
1	データスイッチ	スピーカ
2	SIO 受信データ	SIO 送信データ
3	SIO ステータス	SIO コントロール
4(*)	タイマ現在値	タイマ周期
5(*)	タイマステータス	タイマコントロール
6(*)	空き	INT3 コントロール
7(*)	入力ポート	出力ポート
8(*)	ADC CH0	空き
9(*)	ADC CH1	空き
A(*)	ADC CH2	空き
B(*)	ADC CH3	空き
C	空き	空き
D	空き	空き
E	空き	空き
F	空き	空き

注：(*) は TeC6 には搭載されていない機能を表す．

この表から，例えば次のことが分かります．(詳しくは，後の方で説明します．)

1. 0 番地の Read(読み込み) により，データスイッチの値を読み込むことができる．
2. 0 番地の Write(書き込み) により，ブザーにアクセスできる．
3. 1 番地の Read(読み込み) でも，データスイッチの値を読み込むことができる．

4. 1 番地の Write(書き込み) により，スピーカーにアクセスできる．
5. 2,3 番地は，SIO(シリアル入出力) を用いて PC と通信するために使用される．
6. 4,5 番地は，タイマを使用して時間を計るために使用される．
7. 6 番地は，コンソールパネルから発生する INT3 割込を制御する．
8. 7 番地は，入出力ポートの値を読み書きするために使用される．
9. 8,9,A,B 番地は，ADC(A/D コンバータ：電圧を計測して数値に変換する回路) の値を読み取るために使用される．
10. C,D,E,F 番地は使用されない．

I/O マップは命令表の中にも掲載されていますので，確認してください．

5.11.2 IN(Input) 命令

意味： 入出力インターフェース回路からデータをレジスタに入力します．I/O アドレスは 4 ビットで指定します．

ニーモニック： `IN GR,P`

`GR` は，値を入力するレジスタを表します．`P` は入出力インターフェース回路のアドレスです．

命令フォーマット： IN 命令は 2 バイトの長さを持ちます．XR フィールドは必ず 00 にします．第 2 バイトの上位 4 ビットも必ず 0000 にします．

第 1 バイト		第 2 バイト
OP	GR XR	
1100_2	*GR* 00	0000 pppp

フローチャート： IN 命令は次のように描くことにします．「動作の説明」は，各自が工夫して下さい．



使用例： 次のプログラムはデータスイッチの状態を I/O アドレス 0 番地から読み取り，G0 レジスタに格納するものです．ロータリースイッチを G0 に合わせた状態で，プログラムを実行してください．データスイッチを変化させると即座にデータランプの表示が変化します．

プログラムは無限ループになっているので，実行を開始したら STOP ボタンを押すまで止まりません．

ラベル	ニーモニック
START	IN G0,0
	JMP START

フローチャートで描くと，次のようになります．



DSWは、データスイッチのこと

IN 命令の応用： 次のプログラムは，データスイッチから数値を入力し，合計を G0 に求めるものです．ロタリースイッチを G0 に合わせて実行すると，2 進数入力，2 進数表示の簡単な電卓として使用できます．

操作手順は次の通りです．

1. プログラムを入力する．
2. PC にプログラムの実行開始番地をセットする．
3. ロータリースイッチを G0 に合わせる．
4. データスイッチにデータをセットする．
5. RUN ボタンを押す．
6. データの件数分，4，5 を繰り返す．
7. データランプに合計が表示されている．

ラベル	ニーモニック
START	LD G0,#0
LOOP	IN G1,00H
	ST G1,TMP
	ADD G0,TMP
	HALT
	JMP LOOP

5.11.3　OUT(Output) 命令

意味： レジスタのデータを，入出力インターフェース回路へ出力します．I/O アドレスは 4 ビットで指定します．

ニーモニック： `OUT GR,P`

`GR` は，値を出力するレジスタを表します．`P` は入出力インターフェース回路のアドレスです．

命令フォーマット： OUT 命令は 2 バイトの長さを持ちます．XR フィールドは必ず 11 にします．第 2 バイトの上位 4 ビットも必ず 0000 にします．

第 1 バイト		第 2 バイト
OP	GR XR	
1100_2	*GR* 11	0000 pppp

使用例： 次のプログラムはデータスイッチの値を読み取り，ブザー用のポートに出力するものです．

ブザーは次の図のような構造になっており，I/O アドレス 0 番地に書き込んだ値で鳴ったり止まったりします．プログラム実行中に，データスイッチを操作して最下位ビットを 1 にするとブザーが鳴ります．逆に最下位ビットを 0 にするとブザーの音が止まります．

プログラムは無限ループになっているので，実行を開始したら STOP ボタンを押すまで止まりません．ブザーが鳴っている状態でプログラムを停止しても，ブザーが鳴りっぱなしになります．その場合は，RESET スイッチを押して下さい．

ラベル	ニーモニック	
START	IN	G0,00H
	OUT	G0,00H
	JMP	START



5.12　TeC パラレル入出力 (PIO)

TeC7 には，基板上の入出力ポートコネクタ (図 4.1 参照) を通して 8 ビットデータを並列に入力・出力するパラレル入出力 (Parallel Input Output ： 略して **PIO**) 機能があります．次の写真は PIO を使用し，ブレッドボード上の LED をプログラムで点灯・消灯できるようにした例です．



下に入出力ポートコネクタのピン配置図を示します．I/O アドレスの 7 番地に OUT 命令で書込んだ 2 進数データがコネクタの O_7〜O_0 に出力されます．(対応するビットの “1” が 3.3V，“0” が 0V になります．) I/O アドレスの 7 番地を IN 命令で読むとコネクタの I_7〜I_0 に入力された電圧に対応したデータを入力できます．

入出力ポートコネクタの “0V” ピンは TeC7 の GND に接続されています．“3V” ピンは TeC7 の 3.3V 電源に接続されています．外部に接続した回路の電源として使用することができます．



上の写真では，ブレッドボードを用いて次図のように配線してあります．ポートに “0” を出力すると LED が光ります．



5.13 TeC アナログ入力 (AIN)

TeC7 には，基板上の入出力ポートコネクタ (図 4.1 参照) を通してアナログ電圧値を入力できる機能 (Analog INput：略して **AIN**) があります．TeC7 の AIN には，同時に 4 つのアナログ電圧が入力できます．

アナログ電圧は入出力ポートコネクタの I_3，I_2，I_1，I_0 ピンから入力します．入力電圧は，0V〜3.3V を 256 等分した数値 (0V=0，3.3V=255) に変換されます．数値に変換された結果は，I/O アドレス 8H〜BH で読み取ることができます．次の表に，入力ピンと I/O アドレスの対応をまとめます．

I/O 番地	名称	入力ピン
8	ADC CH0	I_0
9	ADC CH1	I_1
A	ADC CH2	I_2
B	ADC CH3	I_3

入出力ポートコネクタの I_3，I_2，I_1，I_0 ピンは，パラレル入力にもアナログ入力にも使用されます．パラレル入力ポートから見ると，入力電圧が 1.6V 付近を境界に “0” と “1” が切り替わります．

次にアナログ電圧を入力する配線例を示します．可変抵抗器を変化させることで入力値が変化することが確認できます．入力値をプログラムが読み取るためには，I/O アドレスの 8H 番地 (ADC CH0) を IN 命令で読みます．



PIO と AIN は TeC6 では使用できません．
TeC6 にも TeC7 の入出力ポートとよく似たコネクタがあります．このコネクタは「拡張コネクタ」と呼ばれ，TeC6 の内部バスが直接接続されたものです．拡張コネクタを使用すると外部に入出力インタフェース回路を増設でき TeC7 よりも柔軟に外部回路の追加ができますが，外部回路は TeC6 の内部バスと通信できる複雑なものになります．

5.14 TeC シリアル入出力 (SIO)

TeC には，パソコンと接続できるシリアル入出力 (Serial Input Output：略して **SIO**) インターフェースが備えてあります．

前出の IN，OUT 命令を使用して，SIO インターフェースにアクセスします．

5.14.1 シリアル入出力

名前の通り (Serial ＝直列) データを 1 ビット毎，転送する方式です．次の図のように，時間による電圧の変化としてデータを表現します．また，1 ビットの時間として 1/9600 秒を用います．1/9600 秒で 1 ビットを送る通信速度を 9600bau(ボー) と言います．



5.14.2 PC との接続 (TeC7 の場合)

PC と接続するときは，TeC7 の電源コネクタと PC の USB コネクタを接続します．これで，PC から電源が供給されると同時にシリアル通信も可能になります．なお，通信機能を使用するには PC の OS に適切なドライバソフトがインストールされている必要があります．(TeC7 には「USB シリアル変換 IC」として，FTDI 社の FT232RL が実装されています．PC に FT232RL 用のドライバソフトをインストールして下さい．)

次の図にシリアル通信をするときの PC と TeC7 の関係を示します．FT232RL と TeC7 の SIO インタフェース回路の間がシリアル通信部分になります．TeC7 のプログラムは SIO インタフェース回路を操作するだけなので，その先に FT232RL や USB ケーブルがあることは分かりません．TeC7 のプログラムからは，後述の TeC6 のシリアル入出力と全く同じに見えます．

PCとTeC7の接続



5.14.3　PC との接続 (TeC6 の場合)

PC と接続するときは，TeC6 の「RS-232C コネクタ」と PC の裏にある RS-232C コネクタを**クロスケーブル**で接続します．クロスケーブルで接続することにより，PC の送信データと TeC の受信データ，TeC の送信データと PC の受信データが接続され双方向の通信ができるようになります．次の図にシリアル通信をするときの PC と TeC6 の関係を示します．

PCとTeC6の接続



定数の定義 (EQU 命令)

IN 命令の使用例や応用例で示したプログラムでは，データスイッチの値を読み込むための I/O アドレス “0” が，プログラム中に直接書いてありました．データスイッチのことだと分かりやすい名前を用いるとプログラムが読みやすくなります．EQU 命令は，名前 (ラベル) を定義するための命令です．EQU 命令のオペランドの値を，ラベルに割り付けます．他の命令と異なり，機械語やデータを生成しません．

番地	機械語	ラベル	ニーモニック
00		DSW	EQU 00H
00	C0 00	START	IN G0,DSW
02	A0 00		JMP START

5.14.4　I/O ポート

SIO の I/O ポートは，2 番地，3 番地に配置されています (I/O マップ参照)．内容は次の通りです．

1. 受信データ
   2 番地を IN 命令で読むと受信データを読み出すことができます．

2. 送信データ
   2 番地に OUT 命令で送信データを書き出すことができます．

3. ステータス
   3 番地を IN 命令で読むと次図のようなデータを読み込むことができます．各ビットの意味も図の通りです．



プログラムは次の手順でステータスを調べ，データの入出力が可能か判断します．この判断は頻繁に使用するので，フローチャートを次のように簡単化して描くことにします．



4. コントロール
   3 番地へ OUT 命令で次図のようなデータを書き込みます．現在のところこの機能は使用しません．詳細は，6 章で説明します．



5.14.5 シリアル出力プログラム

シリアル出力を行うプログラムの，一般的なフローチャートを次の図に示します．I/O アドレス 3 番地の「T ビット」が “1” になり，送信可能になるのを待って出力データを 2 番地に書き込みます．T ビットが 1 か 0 か調べるために，AND 命令を使用しています．これで，(TeC が PC とケーブルで接続してあれば，) 2 番地に書き込んだデータが PC に送信されます．




```
SIOS  EQU  03H
SIOD  EQU  02H
TBIT  EQU  80H

      LD   G0,#'A'
WAIT  IN   G1,SIOS
      AND  G1,#TBIT
      JZ   WAIT
      OUT  G0,SIOD
      HALT
```

5.14.6 シリアル入力プログラム

シリアル入力を行うプログラムの，一般的なフローチャートを次の図に示します．I/O アドレス 3 番地の「R ビット」が “1” になり，受信データが届くのを待って 2 番地からデータを読み込みます．受信したデータは，G0 に格納されます．




```
SIOS  EQU  03H
SIOD  EQU  02H
RBIT  EQU  40H

WAIT  IN   G1,SIOS
      AND  G1,#RBIT
      JZ   WAIT
      IN   G0,SIOD
      HALT
```

ターミナルプログラム
UNIX または MacOS の **screen プログラム**や **tip プログラム**，Windows の**ハイパーターミナル**等がターミナルプログラムの一種です．ターミナルプログラムは，シリアルポートから受信したデータを ASCII 文字コードとみなし，対応する文字を画面に表示します．また，キーボードが押されると，押された文字の ASCII 文字コードをシリアルポートへ送信します．
TeC の SIO を使用するときは，TeC と PC をケーブルで接続するだけでなく，PC にターミナルプログラムを起動し，PC のシリアルポートが使用できるように準備する必要があります．

5.14.7 シリアル入出力データ

TeC が PC と通信するときは,「2.6 文字の表現」で勉強した ASCII コードを使用する約束とします．PC のキーボードを叩くと，対応する文字の文字コードが TeC に送信され，TeC が文字コードを送信すると，対応する文字が PC の画面に表示されます．(PC 側には，このように動作する**ターミナルプログラム**を動かしておく必要があります．)

ASCII 文字コード表
ASCII 文字コード表は「2.6 文字の表現」で勉強しました．確認のためコード表を再度掲載します．

(上位3ビット)

(下位4ビット)	0	1	2	3	4	5	6	7
0	NUL	DLE	(SP)	0	@	P	`	p
1	SOH	DC1	!	1	A	Q	a	q
2	STX	DC2	”	2	B	R	b	r
3	ETX	DC3	#	3	C	S	c	s
4	EOT	DC4	$	4	D	T	d	t
5	ENQ	NAK	%	5	E	U	e	u
6	ACK	SYN	&	6	F	V	f	v
7	BEL	ETB	’	7	G	W	g	w
8	BS	CAN	(	8	H	X	h	x
9	HT	EM	)	9	I	Y	i	y
A	LF	SUB	*	:	J	Z	j	z
B	VT	ESC	+	;	K	[	k	{
C	FF	FS	,	<	L	\	l	|
D	CR	GS	−	=	M	]	m	}
E	SO	RS	.	>	N	^	n	~
F	SI	US	/	?	O	_	o	DEL

ニーモニック中の文字データ表記

文字を扱うプログラムを書くとき，文字コードが必要になります．そのときは，ニーモニック中に数値の代わりに,「'A'」のように書くことができます．文字 A の文字コード (41H) と言う意味です．(文字コードには ASCII コードを用います．)
ニーモニック中の'A' は，41H または 65 と書いたのと全く同じ意味になります．
DC 命令では，連続した文字の定義に"文字列"を使用できます．下のプログラム中にあるように，"TeC"の表記で 3 バイトのデータが生成されます．

番地	機械語	ラベル	ニーモニック
00	13 41	START	LD G0,#'A'
02	FF		HALT
03	42 43		DC 'B','C'
05	45 65		DC "TeC"
07	43		

ターミナルプログラム画面の改行

ターミナルプログラムの表示を改行するには，復帰コード (CR) と改行コード (LF) の 2 文字を TeC から PC へ送るのが一般的です．
復帰コード (CR:文字コードは 0DH) は文字カーソルを左端に移動するための文字，改行コード (LF:文字コードは 0AH) は文字カーソルを一行下に移動するための文字です．
しかし，実際に UNIX の tip プログラムで試してみると，改行コードのみの送信で復帰と改行の両方ができてしまいました．(復帰コードを送っても害はありません．)
ターミナルプログラムにより，必要な復帰コードと改行コードの組合せが変化するようです．自分が実際に使用するシステムに合わせて，臨機応変に対応してください．

例題 5-7 文字列出力 1

問題： SIO へローマ字で自分の名前を出力するプログラムを作りなさい．

考え方： 筆者の場合，SHIGEMURA の 9 文字を出力するプログラムを作成します．データとして準備した 9 文字を，インデクスドモードのアドレッシングを使用し順に出力します．

フローチャート： 次のようになります．



プログラム：

番地	機械語	ラベル	ニーモニック
02		SIOD	EQU 02H
03		SIOS	EQU 03H
00			
00	17 00	START	LD G1,#0
02	1B 09		LD G2,#9
04	C0 03	LOOP	IN G0,SIOS
06	63 80		AND G0,#80H
08	A4 04		JZ LOOP
0A	11 17		LD G0,DATA,G1
0C	C3 02		OUT G0,SIOD
0E	37 01		ADD G1,#1
10	4B 01		SUB G2,#1
12	A4 16		JZ END
14	A0 04		JMP LOOP
16			
16	FF	END	HALT
17			
17	53 48	DATA	DC "SHIGEMURA"
19	49 47		
1B	45 4D		
1D	55 52		
1F	41		

例題 5-8 文字列出力 2

問題： SIO へローマ字で “Tokuyama Kousen”, “Shigemura” のような 2 行を出力するプログラムを作りなさい.

考え方： 繰り返しにより，1 文字ずつ SIO へ出力します．1 行目データと 2 行目データの間に，復帰コードと改行コードを置きます．これらのコードも通常の文字と同じように扱って構いませんので，プログラムは 2 行分を一気に出力するもので構いません.

プログラム： プログラムにすると次のようになります．データの最後の 00H は，データの終わりを示す目印です．文字コードが 00H の文字は通常使用しませんので，これを文字ではなく終わりの印に使用しました.

この方法だと，出力するデータの長さが変化してもプログラムを変更する必要がありません.

番地	機械語	ラベル	ニーモニック	
02		SIOD	EQU	02H
03		SIOS	EQU	03H
00				
00	17 00		LD	G1,#0
02	19 15	L0	LD	G2,DATA,G1
04	5B 00		CMP	G2,#0
06	A4 14		JZ	END
08	C0 03	L1	IN	G0,SIOS
0A	63 80		AND	G0,#80H
0C	A4 08		JZ	L1
0E	CB 02		OUT	G2,SIOD
10	37 01		ADD	G1,#1
12	A0 02		JMP	L0
14				
14	FF	END	HALT	
15				
15	54 6F	DATA	DC	"Tokuyama
17	6B 75			Kousen"
19	79 61			
1B	6D 61			
1D	20 4B			
1F	6F 75			
21	73 65			
23	6E			
24	0D 0A		DC	0DH,0AH
26	53 68		DC	"Shigemura"
28	69 67			
2A	65 6D			
2C	75 72			
2E	61			
2F	0D 0A		DC	0DH,0AH
31	00		DC	00H

例題 5-9 ‘A’〜‘Z’ の文字を表示

問題： SIO へ ‘A’〜‘Z’ の文字を連続して出力するプログラムを作りなさい.

考え方： ‘A’〜‘Z’ の文字コードは連続しています (ASCII 文字コード表参照)．最初に ‘A’ の文字コードをレジスタにロードし SIO へ出力します．次に，レジスタに 1 加えて文字コードを増やします．増やした結果は ‘B’ の文字コードですので，それを出力します．後は同様に ‘Z’ の文字コードまで出力します.

フローチャート： 次の通りです.



プログラム： G0 レジスタを文字コードの格納に，G1 レジスタを「T ビット」のチェックに使います．出力可能になったら G0 の文字コードを SIO へ出力します．出力した後で出力した文字を調べ，‘Z’ だっら終了します.

番地	機械語	ラベル	ニーモニック	
02		SIOD	EQU	02H
03		SIOS	EQU	03H
00				
00	13 41		LD	G0,#’A’
02	C4 03	WAIT	IN	G1,SIOS
04	67 80		AND	G1,#80H
06	A4 02		JZ	WAIT
08	C3 02		OUT	G0,SIOD
0A	53 5A		CMP	G0,#’Z’
0C	A4 12		JZ	OWARI
0E	33 01		ADD	G0,#1
10	A0 02		JMP	WAIT
12	FF	OWARI	HALT	

例題 5-10 echo プログラム

問題： SIO から入力した文字をそのまま SIO へ送り返す (こだま) プログラムを作りなさい.

考え方： STOP ボタンが押されるまで終了しないプログラムを作ります.

フローチャート： 次のようなアルゴリズムでできるはずです.



プログラム： G0 レジスタを文字コードの格納用に，G1 レジスタを「R ビット」,「T ビット」のチェック用に使います．IN 命令で G0 へ読み込んだ文字コードを，OUT 命令で書き戻すことにより文字を送り返します.

番地	機械語	ラベル	ニーモニック
02		SIOD	EQU 02H
03		SIOS	EQU 03H
00			
00	C4 03	START	IN G1,SIOS
02	67 40		AND G1,#40H
04	A4 00		JZ START
06	C0 02		IN G0,SIOD
08	C4 03	WAIT	IN G1,SIOS
0A	67 80		AND G1,#80H
0C	A4 08		JZ WAIT
0E	C3 02		OUT G0,SIOD
10	A0 00		JMP START

問題

SIO から 1 文字入力し，入力した文字がアルファベット小文字の場合大文字に変換し，そうでなければそのまま SIO へ出力するプログラムを作りなさい.

第6章　高度なプログラミング

6.1　クロス開発

第5章では，ハンドアセンブルにより機械語のプログラムを作成しました．プログラムをニーモニックから機械語に手作業で変換する際，ミスが多発しました．このようにミスが多発するので，実は，ハンドアセンブルによるプログラムの開発は，ほとんど行われません．

この授業の中では，原理をよく理解してもらうために，あえて面倒なハンドアセンブルをたくさんしてもらいました．そろそろ，機械語の原理はよく分かってきたので，楽にプログラミングする方法を教えます．その代わりに，これまでより，たくさんのプログラムを作成してもらいます．

プログラムを作成することを「プログラムを開発する」と言います．楽に「開発する」ために使用するコンピュータやプログラムのことを,「開発環境」と言います.「開発環境」には,「セルフ開発環境」と「クロス開発環境」があります．

「セルフ開発環境」は，パソコンでJava言語やC言語でプログラムを作成し実行するような，プログラムを開発するコンピュータと実行するコンピュータが同じ場合に用いる「開発環境」です．

「クロス開発環境」は，TeCのような小さなコンピュータのプログラム開発によく用いられます．他のコンピュータ上でプログラムを作成し，出来上がったプログラムを目的のコンピュータに転送して実行するような「開発環境」です.「クロス開発」環境を用いたプログラムの開発は,「クロス開発」と呼ばれます．

ここでは,「TeC クロス開発環境」について簡単に説明します．詳細は,「付録B TeC クロス開発環境」を参照してください．

6.1.1　アセンブラ

ニーモニックから機械語に変換する作業を，これまでは，手作業 (ハンドアセンブル) により行ってきました．このような単純で機械的な作業はコンピュータが得意なものです．この作業を行うプログラムを「アセンブラ」と言います．**アセンブラは，ニーモニックで記述したプログラムを，機械語に変換します．**

TeC 用アセンブラの使用方法

TeC用のアセンブラは，UNIXやMacOS，またはWindows上で動作する「クロスアセンブラ」です．パソコンへのインストール方法と使用方法の詳細は，TeC付属のCD-ROMに格納された`ReadMe.txt`ファイルを参照してください．以下では，UNIXまたはMacOSにインストールされていることを前提に説明します．

クロスアセンブラを使用したプログラムの作成手順は，次の通りです．

1. テキストエディタを用いて，プログラムのニーモニックを入力します．プログラムを格納するファイルの拡張子は“`.t7`”(TeC6の場合は“`.t6`”) でなければなりません．次に，プログラムを入力する例を示します．

```
$ emacs xxx.t7
1:  ; 入力例
2:      ld g0,a
3:      st g0,b
4:      halt
5:  a   dc 10
6:  b   ds 1
```

‘;’は，注釈 (コメント) の始まりを表します．‘;’から行末までが注釈になります．1行目が，注釈の例です．ラベルのない行 (2，3，4行) は，一つ以上の空白を書いた後，命令を書きます．ラベルのある行 (5，6) は，行の1文字目からラベルを書きます．なお，例では命令のインデントが揃っていますが，揃える必要はありません．

2. アセンブラを実行します．アセンブラのコマンド名は，`tasm7`(TeC6の場合は`tasm6`) です．実行例は，次の通りです．

```
$ tasm7 xxx.t7
アセンブル成功
結果は [xxx.lst] と [xxx.bin] に格納しました.

[xxx.lst]
ADR   CODE      Label   Instruction

00              ; 入力例
00    10   05           LD      G0,A
02    20   06           ST      G0,B
04    FF                HALT
05    0A        A       DC      10
06    00        B       DS      1
```

入力にミスがなければ，アセンブルリストと機械語プログラムが生成されます．

入力にミスがあった場合は，例えば次のようなエラーメッセージが表示され，エラーの場所と原因が分かるようになっています．

```
$ tasm7 yyy.t7
＊＊＊ エラー [未知のニーモニック] ＊＊＊
エラー発生場所 :  ファイル [yyy.t7] の 3 行で
エラー行の内容 :  [ SL G0,B]
エラートークン :  [SL]
```

TeC 用アセンブラの出力

アセンブルリスト： 画面と拡張子 “.lst” のファイルに，アセンブルリストが出力されます．アセンブルリストは，入力したニーモニックと，それから作られた機械語の対応を分かり易く表示したものです．ハンドアセンブルで行っていた機械語作成の結果と同じになっていることを確認してください．

機械語プログラムファイル： 拡張子 “.bin” のファイルに機械語が格納されます．機械語プログラムファイルの内容は次図の通りです．



ファイルの 1 バイト目は，機械語プログラムの TeC 主記憶上の配置アドレスです．機械語は，このバイトで表される番地を先頭にして主記憶に書き込まれます．ロードアドレスは，アセンブラの ORG 命令で指示することができます．ORG 命令を使用しない場合のロードアドレスは，0 番地になります．ファイルの 2 バイト目は，機械語プログラムの長さです．3 バイト目以降の機械語の長さをバイト単位で表します．

6.1.2　ダウンロード

アセンブラを実行することにより，拡張子 “.bin” のファイルに機械語が格納されました．次に，この機械語を TeC の主記憶にダウンロードします．

転送にはパソコンと TeC をケーブルで接続し通信できる状態にした上で，パソコンで tsend7 コマンド (TeC6 の場合は tsend6 コマンド)，TeC で IPL(Initial Program Loader:付録 B.8 参照) プログラムを実行します．tsend7(tsend6) コマンドと，IPL プログラムが通信してパソコンから TeC にプログラムが転送されます．

ケーブルの接続方法は，「5.14 TeC シリアル入出力」を参照してください．また，tsend7(tsend6) コマンドのインストール方法や Windows での使用方法は，TeC 付属の CD-ROM に格納された ReadMe.txt ファイルを参照してください．以下では，UNIX または MacOS にインストールされていることを前提に説明します．

ダウンロード手順は，次の通りです．

1. パソコンで tsend7(tsend6) プログラムを起動します．下の実行例のように，機械語が格納されたファイルを指定します．TeC を受信状態にするように指示が表示されます．
2. TeC で IPL プログラムを起動します．IPL は，E0H 番地から FFH 番地の ROM 領域 (「付録 C 命令表」のメモリマップを参照) に予め格納されています．PC に E0H をセットし，RUN ボタンを押すことで IPL を起動します．
3. パソコン側で Enter を入力します．

```
$ tsend7 xxx.bin TeC7      を 受 信 状 態 に し て Enter キ ー を 押 し て 下 さ い .
[00][07][10][05][20][06][ff][0a][00]
```

以上の操作で，コンソールパネルから機械語を打ち込むこと無く，機械語が TeC の主記憶の 00H 番地から始まる領域に書き込まれました．主記憶に書き込まれたプログラムを実行する方法は，コンソールパネルから機械語を打ち込んだ場合と同様です．

問題

1. 例題 5-5「ビットの回転」(P.44) をクロス開発環境を用いてやり直しなさい.
2. 例題 5-3「割算を計算する」(P.41) をクロス開発環境を用いてやり直しなさい.
3. 例題 5-8「文字列出力 2」(P.57) をクロス開発環境を用いてやり直しなさい.
   注意：同じシリアル通信機能を「クロス開発環境」と「文字列出力プログラム」の両方で使用します. プログラムのダウンロード時には，パソコンのターミナルプログラム (P.55 参照) を終了する必要があります.

ニーモニック入力上の注意

- TeC のアセンブラでは，アルファベット大文字と小文字は区別されません. 小文字で記述しても，アセンブルリストには大文字に変換されて表示されます.
- 'A'〜'F' の文字で始まる 16 進数は，前に '0' を付加する必要があります. アルファベットで始まる 16 進数は，ラベルと区別ができないためです. (例 F1H → 0F1H)
- 空白の表現には，スペースとタブのどちらも使用できます. インデントを揃えるにはタブを使用すると便利です. また，空白の文字数はいくつでも構いません. 読みやすいように，上手に空白を使用してください.
- ';' をニーモニックの中に書くと，次の文字から行末までが注釈 (コメント) になります. 上手に注釈を入れて，読みやすいプログラムにしてください.
- 文字や文字列の表記も，これまで使用してきた '文字' や "文字列" の記述が使用できます. 実際に使用する記号は，「'」(パソコンの日本語キーボードでは，シフトを押しながら「7」) と，「"」(シフトを押しながら「2」) です. 本書の記述の中では，左と右のクォートに異なる記号を用いている場合もありますが，実際には同じ記号です.

6.2 スタック

スタックは，一時的にデータを保管するために使用できます. 例えば，レジスタの個数が足らなくなったときレジスタの値をスタックに一時退避しておき，後でレジスタに復元することができます.

同じことを，ST，LD 命令と「名前を付けた領域」を用いて行うことも可能ですが，スタックを利用すると「名前を付けた領域」を使用する必要がありません. スタックもメモリ上の領域ですが，名前を付けたりすることなしに，プログラムのあちこちから同じ領域を繰り返し使用することができます.

6.2.1 仕組み

スタックは，CPU のスタックポインタ (SP レジスタ) により管理されます. 最初，SP は，スタック領域の最後の番地を記憶しています. スタックに値を保存するときは，SP の値から 1 を引き新しいアドレスを決め，そのアドレスに値を保存します. 値を保存する度に SP の値は減少していきます.

スタックから値を取り出すときは，SP の示すアドレスの値を取り出し，次に SP の値に 1 を加えます. 値を取り出す度に SP の値は増加していきます.



このように，スタックを用いると，保存したのとは逆の順番で値を取りだすようになります. 最後に入れたデータを最初に取り出すので，スタックは LIFO(Last In First Out) 方式のデータ構造です.

6.2.2　PUSH 命令

Push は,「押す」と言う意味の英語です．データをスタックに「押し込む」のに使用します．

意味： レジスタの値をスタックに押し込む．

ニーモニック： `PUSH GR`

`GR` で指定されたレジスタがスタックに押し込まれます．`GR` が SP を表す場合は，SP の値がスタックに押し込まれます．

命令フォーマット： PUSH 命令は，1 バイト長の命令です．XR フィールドは必ず 00 にします．

第 1 バイト	
OP	GR XR
1101_2	*GR* 00

動作の詳細： PUSH 命令は，まず，スタックポインタ (SP) の値を 1 減らします．次に，SP の値をアドレスとみなし，主記憶の SP 番地に指定されたレジスタの値を書き込みます．

フローチャート： 次のように描くことにします．



6.2.3　POP 命令

Pop は,「飛び出る」と言う意味の英語です．データをスタックから取り出すのに使用します．

意味： スタックのデータをレジスタに取り出す．

ニーモニック： `POP GR`

`GR` で指定されたレジスタにスタックからデータが取り出されます．`GR` が SP を表す場合は，スタックからデータが取り出され，SP に格納されます．

命令フォーマット： POP 命令は，1 バイト長の命令です．XR フィールドは必ず 10 にします．

第 1 バイト	
OP	GR XR
1101_2	*GR* 10

動作の詳細： POP 命令は，まず，主記憶の SP 番地からデータを取り出し，指定されたレジスタに格納します．次に，SP の値を 1 増やします．

フローチャート： 次のように描くことにします．



PUSH/POP 命令の応用例 (1)

PUSH 命令と POP 命令を組み合わせて，レジスタの値を一時的に保管することができます．まず，SP の値を初期化します．DCH 番地に初期化する理由は,「付録 C 命令表」のメモリマップを参照して考えてください．スタックは LIFO なので，**PUSH(保存) した順序と逆の順序で POP(復元) する必要があります．**

ラベル	ニーモニック
	`LD SP,#0DCH`
	`...`
	`G0, G1 を使用する処理 1`
	`...`
	`PUSH G0 ; G0 を保存`
	`PUSH G1 ; G1 を保存`
	`...`
	`G0, G1 を使用する処理 2`
	`...`
	`POP G1 ; G1 を復元`
	`POP G0 ; G0 を復元`
	`...`
	`処理 1 の続き`
	`...`

PUSH/POP 命令の応用例 (2)

PUSH 命令と POP 命令を組み合わせて，レジスタの値を別のレジスタにコピーすることができます．TeC にはレジスタ間のデータ転送命令がないので，このテクニックは役に立ちます．次の例では G0 の値を G1 にコピーします．

ラベル	ニーモニック
	`LD SP,#0DCH`
	`...`
	`PUSH G0 ; G0 を保存`
	`POP G1 ; G1 に復元`
	`...`

問題

G0 と G1 の値を交換する方法を考えなさい．

6.3 サブルーチン

プログラムの中に同じ処理が何度も出てくることがあります．また，一つのまとまった機能を実現している部分があり，他の部分と独立させた方がプログラムの見通しが良くなる場合があります．例えば，「5.14 TeC シリアル入出力 (SIO)」の例題プログラムの多くで，1 文字入力や 1 文字出力を行うプログラムの部分がありました．このような，一つのまとまった機能を実現している部分がそうです．

このような部分はサブルーチン (副プログラム) としてまとめることにより，繰り返し呼び出して使用したり，他の部分から独立させて見やすいプログラムにすることができます．ここでは，サブルーチンを作る方法を学びます．

6.3.1 仕組み

サブルーチンは，他のプログラムから呼び出されて実行されます．サブルーチンの実行が終わったら，呼び出したプログラムに戻り続きを実行します．

サブルーチンを呼び出すための CALL 機械語命令と，サブルーチンの最後から呼び出し元に戻るための RET 機械語命令を新たに導入します．

次のプログラムは，1 文字出力機能をサブルーチンとして独立させ，メインルーチン (主プログラム) から 2 回呼び出します．プログラムは先頭から実行が開始され CALL 命令で PUTCH サブルーチンへジャンプします．PUTCH サブルーチンを最後まで実行すると RET 命令でメインルーチンに戻り，CALL 命令の次の命令から実行を再開します．



6.3.2 CALL 命令

Call は,「呼び出す」と言う意味の英語です．サブルーチンを呼び出すために使用します．CALL 命令は JMP 命令に似ていますが，サブルーチンから戻る準備をしてからジャンプしなければなりません．CALL 命令は，自身の次の命令のアドレスをスタックに保存 (PUSH) してから、サブルーチンにジャンプします．

意味： サブルーチンを呼び出す．

フラグ： 変化しません．

ニーモニック： `CALL A[,XR]`

A 番地のサブルーチンを呼び出します．`[,XR]` は，インデックスドモードでジャンプアドレスを計算する場合に使用します．

命令フォーマット： CALL 命令は 2 バイトの長さを持ちます．各フィールドの意味は JMP 命令と同様です．この命令は TeC7 と TeC6 で異なる命令コードを持っています．

TeC7 の場合

第 1 バイト		第 2 バイト
OP	GR XR	
1011_2	00 XR	aaaa aaaa

TeC6 の場合

第 1 バイト		第 2 バイト
OP	GR XR	
1011_2	11 XR	aaaa aaaa

動作の詳細： CALL 命令は，まず，PC の値 (CALL 命令実行時には既に次の命令のアドレスになっている) を，スタックに保存 (PUSH) します．次に，サブルーチンにジャンプします．

フローチャート： 次のように描きます．箱の中には，呼び出すサブルーチンの名前を書きます．



使用上の注意： CALL 命令はスタックを使用するので，SP を初期化してから使用しなければなりません．

6.3.3　RET(Return) 命令

Return は,「戻る」と言う意味の英語です．RET は Return の綴を縮めたものです．サブルーチンから戻るために使用します．

意味： サブルーチンから戻る．

フラグ： 変化しません．

ニーモニック： `RET`

命令フォーマット： RET 命令は，1 バイト長の命令です．

第 1 バイト	
OP	GR XR
1110_2	11 00

動作の詳細： スタックから値を一つ取り出し PC にセットします．スタックには CALL 命令が保存した PC の値が保存されているので，プログラムの実行は，サブルーチンを呼び出した CALL 命令の次の命令に移ります．

フローチャート： 次のフローチャートの，最後の角を取った四角が RET 命令に対応します．フローチャートは，サブルーチン全体を示しています．



使用上の注意： RET 命令は POP 命令と同様な方法でスタックから値を取り出します．サブルーチンの中で PUSH 命令や POP 命令を使用した場合は，スタックの状態がサブルーチンの実行開始時と同じになるようにしてから，RET 命令を実行しなければなりません．サブルーチンの中で，PUSH 命令の実行回数と POP 命令の実行回数が同じになるようにして下さい．

例題 6-1 ‘A’〜‘Z’ の文字を表示 (改良版 1)

問題： SIO へ ‘A’〜‘Z’ の文字を連続して出力するプログラムを，前ページの 1 文字出力サブルーチン (PUTCH) を使用して作りなさい．

考え方： 基本的な考え方は，例題 5-9(P.57) と同じです．1 文字出力サブルーチンは，G0 レジスタにセットされた値を SIO へ出力します．メインルーチンは，G0 レジスタに ‘A’〜‘Z’ の文字の文字コードをセットして，サブルーチンを呼び出します．

フローチャート： サブルーチンの呼び出しと，サブルーチンは次のように描きます．



プログラム： CALL，RET 命令がスタックを使用するので，まず，SP の初期化をします．G0 レジスタを文字コードの格納に，G1 レジスタを「T ビット」のチェックに用います．

番地	機械語	ラベル	ニーモニック	
02		SIOD	EQU	02H
03		SIOS	EQU	03H
00				
00	1F DC		LD	SP,#0DCH
02	13 41		LD	G0,#'A'
04	B0 0F	LOOP	CALL	PUTCH
06	53 5A		CMP	G0,#'Z'
08	A4 0E		JZ	OWARI
0A	33 01		ADD	G0,#1
0C	A0 04		JMP	LOOP
0E	FF	OWARI	HALT	
0F				
0F	C4 03	PUTCH	IN	G1,SIOS
11	67 80		AND	G1,#80H
13	A4 0F		JZ	PUTCH
15	C3 02		OUT	G0,SIOD
17	EC		RET	

TeC6 では CALL 命令の機械語が BC になります．

例題 6-2 ‘A’〜‘Z’ の文字を表示 (改良版 2)

問題： SIO へ ‘A’〜‘Z’ の文字を連続して出力した後，SIO へ改行コードを出力するプログラムを作りなさい．

考え方： 例題 6-1(P.64) では，サブルーチンを使用したメリットが明確になりませんでした．同じ，サブルーチン (PUTCH) をメインルーチンの複数の箇所から呼び出すプログラムを作成します．

フローチャート： 改行するために，CR，LF を出力します．



プログラム： PUTCH ルーチンは，例題 6-1(P.64) と同様なので省略しました．

番地	機械語	ラベル	ニーモニック	
02		SIOD	EQU	02H
03		SIOS	EQU	03H
00				
00	1F DC		LD	SP,#0DCH
02	13 41		LD	G0,#'A'
04	B0 17	LOOP	CALL	PUTCH
06	53 5A		CMP	G0,#'Z'
08	A4 0E		JZ	OWARI
0A	33 01		ADD	G0,#1
0C	A0 04		JMP	LOOP
0E	13 0D	OWARI	LD	G0,#13
10	B0 17		CALL	PUTCH
12	13 0A		LD	G0,#10
14	B0 17		CALL	PUTCH
16	FF		HALT	
17				
17	C4 03	PUTCH	...	

TeC6 では CALL 命令の機械語が BC になります．

例題 6-3 文字列出力 (改良版 1)

問題： SIO へローマ字で自分の名前を出力するプログラムを作りなさい．

考え方： 例題 5-7(P.56) を，サブルーチンを使用して作りなおします．

フローチャート： G1 レジスタが PUTCH サブルーチンで破壊される (T ビットのチェックで使用します．) ので，メインルーチン側で G1 レジスタをスタックに保存しています．



プログラム： PUTCH ルーチンは省略しました．

番地	機械語	ラベル	ニーモニック	
02		SIOD	EQU	02H
03		SIOS	EQU	03H
00				
00	1F DC		LD	SP,#0DCH
02	17 00		LD	G1,#0
04	1B 09		LD	G2,#9
06	11 1E	LOOP	LD	G0,DATA,G1
08	D4		PUSH	G1
09	B0 15		CALL	PUTCH
0B	D6		POP	G1
0C	37 01		ADD	G1,#1
0E	4B 01		SUB	G2,#1
10	A4 14		JZ	END
12	A0 06		JMP	LOOP
14				
14	FF	END	HALT	
15				
15	C4 03	PUTCH	...	
1E				
1E	53 48	DATA	DC	"SHIGEMURA"
20	...			

TeC6 では CALL 命令の機械語が BC になります．

例題 6-4 文字列出力 (改良版 2)

問題： SIO へローマ字で自分の名前を出力するプログラムを作りなさい.

考え方： 例題 5-7(P.56) をサブルーチンを使用して作りなおします.

フローチャート： PUTCH サブルーチンが G1 レジスタを破壊しないよう，サブルーチン側で G1 レジスタをスタックに保存しています. 例題 6-3(P.65) の PUTCH サブルーチンより完成度の高い PUTCH サブルーチンです.



プログラム： PUTCH ルーチンも掲載します.

番地	機械語	ラベル	ニーモニック	
02		SIOD	EQU	02H
03		SIOS	EQU	03H
00				
00	1F DC		LD	SP,#0DCH
02	17 00		LD	G1,#0
04	1B 09		LD	G2,#9
06	11 1E	LOOP	LD	G0,DATA,G1
08	B0 13		CALL	PUTCH
0A	37 01		ADD	G1,#1
0C	4B 01		SUB	G2,#1
0E	A4 12		JZ	END
10	A0 06		JMP	LOOP
12				
12	FF	END	HALT	
13				
13	D4	PUTCH	PUSH	G1
14	C4 03	WAIT	IN	G1,SIOS
16	67 80		AND	G1,#80H
18	A4 14		JZ	WAIT
1A	C3 02		OUT	G0,SIOD
1C	D6		POP	G1
1D	EC		RET	
1E				
1E	53 48	DATA	DC	"SHIGEMURA"
20	49 47			
22	45 4D			
24	55 52			
26	41			

TeC6 では CALL 命令の機械語が BC になります.

問題

1. SIO から 1 文字を入力するサブルーチン (GETCH) を作成しなさい. GETCH は，入力した文字の文字コードを G0 に格納して戻るものとします.
2. 前出の 1 文字出力サブルーチン (PUTCH) と，GETCH を利用して例題 5-10「echo プログラム」(P.58) を作りなおしなさい.
3. 例題 5-3「割算を計算する」(P.41) を参考に，割算サブルーチン DIV を作成しなさい. DIV は，G0 に割られる数，G2 に割る数をセットして呼び出され，G1 に商，G0 に余りを格納して戻るものとします.
4. 例題 5-6「シフトを用いた高速乗算」(P.45) を参考に，G0 の数値の 10 倍を計算し，G0 に求めるサブルーチン MUL10 を作成しなさい.

6.4 時間の計測

コンピュータで，時間を計る必要がある場合があります．例えば，TeC を使って 3 分間のタイマを作ることを想像してください．

PC 等の本格的なコンピュータは，時間を計測するための専用の回路を持っています．しかし，TeC のような小規模マイコンは，専用の回路を持っていないことがあります (TeC は持っていません)．

そこで，プログラムだけで時間を計測する工夫が必要になります．

6.4.1 マシンステート

TeC で一定の時間が経過するのを待つには，各命令の実行に必要な時間を調べて，ちょうど目的の時間が必要な命令の組合せを実行させます．

命令表にステート数と書かれた欄があります．ここの数値が，命令の実行に要する時間を表します．例えば，NO 命令はステート数 3 の命令ですので，**3 マシンステート**の時間が必要になります．

TeC の場合，**1 マシンステート**は，回路の動く基準になるクロック信号の 1 周期分になります．TeC のクロック信号は，2.4576MHz ですので，**1 マシンステート**は，$1/(2.4576\times10^6) = 0.4069\times10^{-6} = 0.4069\mu s$ です．

例えば，NO 命令はステート数が 3 なので，実行に $0.4069\mu s \times 3 = 1.2207\mu s$ の時間がかかります．他の命令も，同様に実行時間が決まります．

6.4.2 1ms タイマ (マシンステートの応用)

マシンステートの合計を調節して，実行にちょうど 1ms の時間がかかるサブルーチンを作ります．

クロックの周波数から，1 秒が 2.4576×10^6 マシンステートになりますので，$1ms$ は，$(2.4576 \times 10^6) \times (1 \times 10^{-3}) = 2,457.6$ マシンステートになります．

実行に，2,458 マシンステートを要するサブルーチンを作ります．ステート数が多いので繰り返しを使用しないと，うまくいきそうにありません．繰り返しを作るには，次のようなプログラムが必要です．

ラベル	ニーモニック		ステート
MS1	PUSH	G0	6
	LD	G0,#n	5
MATU	SUB	G0,#1	5
	JZ	KAERU	4/5
	JMP	MATU	5
KAERU	POP	G0	6
	RET		6

このプログラムの実行に要するステート数は，繰り返し回数を n とすると，$6+5+(5+4+5)\times(n-1)+5+5+6+6$ となります．

ステート数が 2,458 になるように n を計算すると，n は 174 になります (3 ステート不足)．上のプログラムで n を 174 にしたものが，実行に約 $1ms$ かかるサブルーチンになります．

6.4.3 0.2 秒タイマ (入れ子サブルーチンの利用)

もう少し長いタイマを作って，実行に時間がかかっていることを実感しましょう．

$1ms$ タイマサブルーチンを利用すると，簡単にもっと長い時間のタイマサブルーチンを作ることができます．$200ms$ タイマルーチンは，$1ms$ タイマサブルーチンを 200 回呼び出すように作ります．(**「サブルーチンの中から別のサブルーチンを呼び出すこと」 =「サブルーチンの入れ子」を利用します．**)

次のプログラムは，実行に約 1 秒かかるようにしたものです．

ラベル	ニーモニック		
	; メインルーチン		
START	LD	SP,#0DCH	
	CALL	MS200	
	CALL	MS200	
	CALL	MS200	
	CALL	MS200	
	CALL	MS200	
	HALT		
	; 200ms サブルーチン		
MS200	PUSH	G0	
	LD	G0,#200	
L1	CALL	MS1	
	SUB	G0,#1	
	JZ	L2	
	JMP	L1	
L2	POP	G0	
	RET		
	; 1ms サブルーチン		
MS1	PUSH	G0	
	LD	G0,#174	
MATU	SUB	G0,#1	
	JZ	KAERU	
	JMP	MATU	
KAERU	POP	G0	
	RET		

6.4.4　1 秒タイマ (多重ループの利用)

約 $1ms$ のタイマを 200 回呼び出すサブルーチンを 5 回呼び出す方法では，誤差がたまって，正確な 1 秒タイマではなくなっている可能性があります．正確な 1 秒タイマを，1 から作りなおします．

1 秒タイマは，$2,457,600$ マシンステートのプログラムになります．$1ms$ タイマでは，繰り返し 1 回が 14 ステートでした．同じステート数だとすると，$2,447,600/14 = 174,828$ 回 ループを繰り返す必要があります．

カウンタに使用したレジスタは 8 ビットですから，256 回のカウントが限界です．そこで，下のフローチャートのように繰り返しを更に繰り返す (2 重ループ) プログラムを作成する必要があります．



ところが，256 回を 256 回繰り返しても，まだ，繰り返し回数の 174,828 回 に足りません．そこで，実際には「「繰り返し」の繰り返し」の繰り返し」(3 重ループ) を作ります．

プログラムは以下のようになります．繰り返し回数は，プログラムを書いてプログラム各部のステート数が定まった後で，試行錯誤で決めました．

番地	機械語	ラベル	ニーモニック		
00	13 6C	START	LD	G0,#108	; 5
02	17 CC	L1	LD	G1,#204	; 5
04	1B 07	L2	LD	G2,#7	; 5
06	4B 01	L3	SUB	G2,#1	; 5
08	A4 0C		JZ	L4	; 4/5
0A	A0 06		JMP	L3	; 5
0C	47 01	L4	SUB	G1,#1	; 5
0E	A4 12		JZ	L5	; 4/5
10	A0 04		JMP	L2	; 5
12	43 01	L5	SUB	G0,#1	; 5
14	A4 18		JZ	L6	; 4/5
16	A0 02		JMP	L1	; 5
18					
18	00	L6	NO		; 3
19	00		NO		; 3
1A	FF		HALT		; 3

問題

1. 0.2 秒タイマサブルーチンを使用した，1 秒タイマプログラムを入力して実行しなさい．
2. 0.2 秒タイマサブルーチンを利用して，ブザーを 1 秒間隔で 0.2 秒間鳴らすプログラムを作りなさい．このプログラムは STOP を押すまで繰り返し実行するものとします．
3. 多重ループを利用した 1 秒タイマプログラムを入力して実行しなさい．
4. 2 重ループの練習として，SIO に ‘A’ を 10 文字，5 行出力 (下の出力例参照) するプログラムを作りなさい．
5. 入れ子サブルーチンの練習として，SIO に ‘A’ を 10 文字，5 行出力するプログラムを作りなさい．(1 行出力サブルーチンを 5 回呼び出します．1 行出力サブルーチンは，1 文字出力サブルーチンを 10 回呼び出します．)
6. 最初の 3 分間は 1 秒毎にブザーを鳴らし，3 分経過したらブザーを鳴らしっぱなしするプログラム (3 分タイマ) を作りなさい．

‘A’ を 10 文字，5 行出力した例

```
A A A A A A A A A A
A A A A A A A A A A
A A A A A A A A A A
A A A A A A A A A A
A A A A A A A A A A
```

6.5 数値の入出力

これまでに，SIO を使用した文字の入出力を勉強しました．ここでは，数値の入出力を扱います．

6.5.1 2 進数の出力

データを 2 進数として表現したものを，人間に読める形式で SIO に出力します．ビットの 0 は，文字の ‘0’ に，ビットの 1 は，文字の ‘1’ に変換します．ここでは，1 バイトのデータを表示するプログラムの例を示します．

プログラムは，DATA 番地の内容を上位ビットから順に ‘0’，‘1’ の文字に変換し，SIO へ出力します．8 ビット分の出力が終わると，CR(0DH)，LF(0AH) を出力して画面を改行します．文字の出力には，前出の PUTCH サブルーチンを使用します．

このプログラムを実行すると，PC の画面に，DATA 番地のデータ CAH を 2 進数で表した，“11001010” が表示されます．

番地	機械語	ラベル	ニーモニック
02		SIOD	EQU 2
03		SIOS	EQU 3
00			
00	1F DC		LD SP,#0DCH
02	14 29		LD G1,DATA
04	1B 08		LD G2,#8
06	13 31	L1	LD G0,#’1’
08	95		SHLL G1
09	A8 0D		JC L2
0B	13 30		LD G0,#’0’
0D	B0 1E	L2	CALL PUTCH
0F	4B 01		SUB G2,#1
11	A4 15		JZ L3
13	A0 06		JMP L1
15			
15	13 0D	L3	LD G0,#0DH
17	B0 1E		CALL PUTCH
19	13 0A		LD G0,#0AH
1B	B0 1E		CALL PUTCH
1D	FF		HALT
1E			
1E	D4	PUTCH	PUSH G1
1F	C4 03	PL1	IN G1,SIOS
21	67 80		AND G1,#80H
23	A4 1F		JZ PL1
25	C3 02		OUT G0,SIOD
27	D6		POP G1
28	EC		RET
29			
29	CA	DATA	DC 0CAH

TeC6 では CALL 命令の機械語が BC になります．

6.5.2 16 進数の出力

DATA 番地のデータを 16 進数で SIO に出力するプログラムの例を，以下に示します．

まず，データの上位 4 ビットを取り出し，その値を用いて ‘0’〜‘F’ の 1 文字を選択し SIO へ出力します．次に，下位 4 ビットを用いて同様の出力をします．

最後に，CR(0DH)，LF(0AH) を出力して画面を改行します．このプログラムを実行すると，PC の画面に，DATA 番地のデータ “CA” が表示されます．

番地	機械語	ラベル	ニーモニック
02		SIOD	EQU 2
03		SIOS	EQU 3
00			
00	1F DC		LD SP,#0DCH
02			; 上位桁の出力
02	14 38		LD G1,DATA
04	97		SHRL G1
05	97		SHRL G1
06	97		SHRL G1
07	97		SHRL G1
08	11 28		LD G0,HSTR,G1
0A	B0 1D		CALL PUTCH
0C			; 下位桁の出力
0C	14 38		LD G1,DATA
0E	67 0F		AND G1,#0FH
10	11 28		LD G0,HSTR,G1
12	B0 1D		CALL PUTCH
14			; 改行の出力
14	13 0D		LD G0,#0DH
16	B0 1D		CALL PUTCH
18	13 0A		LD G0,#0AH
1A	B0 1D		CALL PUTCH
1C	FF		HALT
1D			; 1 文字出力
1D	D4	PUTCH	PUSH G1
1E	C4 03	WAIT	IN G1,SIOS
20	67 80		AND G1,#80H
22	A4 1E		JZ WAIT
24	C3 02		OUT G0,SIOD
26	D6		POP G1
27	EC		RET
28			; 16 進数字
28	30 31	HSTR	DC ’0’,’1’,’2’,’3’
2A	32 33		
2C	34 35		DC ’4’,’5’,’6’,’7’
2E	36 37		
30	38 39		DC ’8’,’9’,’A’,’B’
32	41 42		
34	43 44		DC ’C’,’D’,’E’,’F’
36	45 46		
38			; 出力するデータ
38	CA	DATA	DC 0CAH

TeC6 では CALL 命令の機械語が BC になります．

6.5.3　10 進数の出力

DATA 番地のデータを 10 進数で SIO に出力するプログラムを考えましょう．扱うデータは 8 ビットの符号無し 2 進数ですので，値の範囲は 0 〜 255(最大 3 桁) です．10 進数で表示するには，10 進数の一桁一桁に分解した後，一桁毎に対応する数字に変換して出力します．

各桁への分解

次のような手順で，数値を一桁一桁に分解します．下の，計算例を参照しながら確認して下さい．

1. データを 10 で割って商と余りを求めます．この時，余りが最下位桁 (1 位桁) の値です．
2. 商を更に 10 で割って，新しい商と余りを求めます．新しい余りが，下から 2 桁目 (10 位桁) の値です．
3. 新しい商は下から 3 桁目 (100 位桁) の値になります．

計算例

10) 123
10) 12 ･･･3　(余り 3 が 1 位桁の値)
1 ･･･2　(余り 2 が 10 位桁の値)
(商 1 が 100 位桁の値)

以上の手順をプログラムにするためには，割算機能が必要です．ここでは，**例題 5-3**(P.41) を参考に割算サブルーチンを準備し，DIV と言う名前で作成済みと仮定します．

次は，割算サブルーチン DIV を使用した桁への分解プログラム (部分) です．各桁の値は，D1，D2，D3 の 3 バイトに格納されます．後で，これらの値を上の桁から順に出力します．

各桁に分解する部分

```
ラベル    ニーモニック      コメント
PUTDEC    LD    SP,#0DCH    ; SP 初期化
          LD    G0,DATA     ; 割られる数
          LD    G2,#10      ; 割る数
          CALL  DIV         ; 割算
          ST    G0,D1       ; 余り (1 位桁)
          PUSH  G1          ; 商を
          POP   G0          ; 割られる数に
          CALL  DIV         ; 割算
          ST    G0,D2       ; 余り (10 位桁)
          ST    G1,D3       ; 商 (100 位桁)
```

各桁の出力

桁への分解ができたら，分解した各桁を上の桁から順に数字に変換して SIO へ出力します．数字の文字コードが ‘0’ 〜 ‘9’ まで連続していることを利用すると，数値から数字への変換が簡単にできます．

数字 ＝ 数値 ＋ ‘0’ の文字コード

変換したら，前出の PUTCH サブルーチンを利用して SIO へ出力します．次は，100 位桁の値を数字に変換し出力するプログラム (部分) です．

一桁目を数字に変換し出力する部分

```
ラベル    ニーモニック      コメント
          LD    G0,D3       ; 100 位桁の数値
          ADD   G0,#'0'     ; 文字コードへ変換
          CALL  PUTCH       ; SIO へ出力
```

6.5.4　10 進数の入力

10 進数の出力ができるようになりましたので，今度は入力を考えましょう．

10 進数の出力と，ちょうど逆の手順で行います．まず，一桁毎に入力し，数字を数値に変換します．次に，各桁の値を合成して一つの数値にします．

各桁の入力

入力は符号無し 8bit 2 進数で表現できる 0 〜 255 の数値です．今回はプログラムの作りやすさを優先して，3 桁の数字しか入力されない前提で考えます．100 未満の数値を入力するときは，“012” のように上の 1 桁に ‘0’ を入力します．10 未満の数値を入力するときは，“001” のように上の 2 桁に ‘0’ を入力します．

1 文字入力し，入力した文字の文字コードから ‘0’ の文字コードを引きます．これで，数字に対応した数値が求まります．入力された各桁の値は，メモリ上の領域に保存しておきます．

数値 ＝ 数字 − ‘0’ の文字コード

ここまでの手順をプログラムにすると，次のリストのようになります．プログラム中で 1 文字入力 (GETCH) 毎に 1 文字出力 (PUTCH) しているのは，入力した文字が端末 (PC) の画面に表示されるようにするためです．

3 桁入力し数値に変換してメモリに保存する部分

ラベル	ニーモニック		コメント
GETDEC	LD	SP,#0DCH	; SP 初期化
	CALL	GETCH	; 1 文字入力
	CALL	PUTCH	; エコーバック
	SUB	G0,#'0'	; 文字コードから値に
	ST	G0,D3	; 100 位桁
	CALL	GETCH	; 1 文字入力
	CALL	PUTCH	; エコーバック
	SUB	G0,#'0'	; 文字コードから値に
	ST	G0,D2	; 10 位桁
	CALL	GETCH	; 1 文字入力
	CALL	PUTCH	; エコーバック
	SUB	G0,#'0'	; 文字コードから値に
	ST	G0,D1	; 1 位桁

各桁の値の合成

下の式のような計算で，目的の値を求めることができます．2 行目の式のように計算すれば，10 倍を計算するサブルーチンを 2 回呼び出せば良いことになります．

$$\begin{aligned} \text{値} &= 100\text{ 位桁} \times 100 + 10\text{ 位桁} \times 10 + 1\text{ 位桁} \\ &= ((100\text{ 位桁} \times 10) + 10\text{ 位桁}) \times 10 + 1\text{ 位桁} \end{aligned}$$

この計算をプログラムにすると，下のリストのようになります．ここで MUL10 は，G0 値の 10 倍を計算するサブルーチンです．**例題 5-6**(P.45) を参考に，予め MUL10 を作成しておく必要があります．

各桁の値を合成する部分

ラベル	ニーモニック		コメント
	LD	G0,D3	; 100 位桁
	CALL	MUL10	; × 10
	ADD	G0,D2	; 10 位桁
	CALL	MUL10	; × 10
	ADD	G0,D1	; 1 位桁
	ST	G0,DATA	; 結果を保存

掛算の回数

一般に，掛算や割算は時間のかかる計算です．TeC のような掛算命令や割算命令を持たない小さなコンピュータにとっては，特に時間がかかります．そこで，掛算や割算の回数が少なくなるように，プログラムするよう心掛ける必要があります．上で掛算の回数が少なくるなるように式を変形したのは，そのような意味があるからです．

問題

1. SIO から入力した文字の文字コードを 16 進数で出力するプログラムを作成しなさい．(1 文字入力サブルーチン (GETCH)，16 進数出力サブルーチン (PUTHEX) を作成し，これらを用いてプログラムを完成しなさい．)
2. 本文中のプログラムを参考に，DATA 番地のデータを 10 進数で出力するプログラムを作成しなさい．
3. 10 進数の出力プログラムを，出力する数値の桁数に合わせて先行する余分な '0' が出力されないよう改良しなさい．
4. 10 進数の出力プログラムを，D1，D2，D3 領域の代わりにスタックを使用するように改造しなさい．
5. 本文中のプログラムを参考に，10 進数を入力し DATA 番地に格納するプログラムを作成しなさい．
6. 10 進数を入力するプログラムを，数字以外が入力されるまで入力を受け付けるように改造しなさい．(3 桁固定でなく，10[エンター] や 1[エンター] で入力できるプログラムにする．)
7. 10 進数を入力し，10 加えた値を 10 進数で出力するプログラムを作成しなさい．(10 進数入力サブルーチン (GETDEC)，10 進数出力サブルーチン (PUTDEC) を作成し，これらを用いてプログラムを完成しなさい．)
8. 二つの 10 進数を入力し，和を 10 進数で出力するプログラムを作成しなさい．
9. ゼロが入力されるまで 10 進数を入力し，合計を 10 進数で出力するプログラムを作成しなさい．

6.6　符号付数の入出力

「2.3 負数の表現」で学んだように，負の数を 2 の補数表現を用いて 2 進数で表現することができます．TeC も，負の数を 2 の補数表現を用いて表現します．TeC は 8bit 符号付 2 進数を扱いますので，–128 〜 +127 の数値を扱うことができます．

データが符号付であるか符号無しであるかは，そのデータをプログラムがどのように扱うかで決まります．特に，データの外部表現を左右する入出力ルーチンは，符号付の場合とそうでない場合で，処理内容がはっきり区別されます．

6.6.1　符号付 10 進数の出力

以下に G0 に格納された符号付 2 進数の値を SIO へ出力するサブルーチンのフローチャートを示します．なお，G0 に格納した 1 文字を出力する PUTCH サブルーチン，G0 に格納した値を 10 進数として出力する PUTDEC サブルーチンが別に必要です．

まず，G0 の値と 0 を比較し，G0 の値が正か負かを判断します．負の場合は，‘–’ を出力した後，G0 の値の符号を反転します．符号の反転は，「2.3.6 2 の補数から元の数を求める手順」で学習したように，**「ビット反転した後，1 を加える」**ことで実行できます．ビットの反転には XOR 命令を使用します．

最後に，PUTDEC サブルーチンを用いて G0 の値を 10 進数で出力します．



6.6.2　符号付 10 進数の入力

次のような形式の 10 進数を入力します．

正の数：空白 数字 数字 数字
負の数：‘–’ 数字 数字 数字

このような形式で数値が入力される場合の，符号付数値の入力サブルーチンのフローチャートを示します．なお，SIO から 1 文字を入力し G0 に格納するサブルーチン GETCH，SIO から 10 進数を入力し G0 に格納するサブルーチン GETDEC が，予め準備してあるものとします．

まず，符号を入力しスタックに退避します．続いて 10 進数を G0 に入力します．退避しておいた符号を G1 に取り出し，文字 ‘–’ と比較します．同じなら，G0 の符号を反転します．



問題

1. 符号付の 10 進数出力サブルーチン (PUTSDEC) を作成しなさい．
2. 符号付の 10 進数入力サブルーチン (GETSDEC) を作成しなさい．
3. ゼロが入力されるまで符号付の 10 進数を入力し，合計を符号付 10 進数で出力するプログラムを作成しなさい．

6.7 アドレスデータ

番地をデータとして扱えると便利なことがあります．番地 (＝アドレス) のデータですので，ここではアドレスデータと呼びます．C 言語を勉強したことのある人は，ポインタのことだ言えばピンとくると思います．アドレスデータを使用すると便利な例としては，サブルーチンにデータの置き場所を教える場合が考えられます．

6.7.1 TeC のアドレスデータ

TeC では，インデクスドモードのアドレッシングを用いて，レジスタで指定された，特定アドレスのデータをアクセスすることが可能です．例えば，12H 番地にあるデータを G0 レジスタにロードするプログラムは，次のようになります．

```
LD   G1,#12H
LD   G0,0,G1
```

G1 インデクスドモードでは，データのアドレスを 「第 2 バイト+G1」 で求めます．このプログラムの場合，第 2 バイトが 0 ですから，G1 の値がそのままデータのアドレスになります．つまり，G1 に格納されたアドレスデータを用いて，そのアドレスをアクセスします．

6.7.2 文字列出力サブルーチン (アドレスデータの使用例)

文字列出力サブルーチンを作ってみましょう．サブルーチンには，出力すべき文字列を渡す必要があります．ここでは文字列を，例題 5-8(P.57) で用いたのと同じ，文字コード 0 で終わるバイト列として表現することにします．(この文字列表現方式は，C 言語と同じです．)

サブルーチンにデータを渡す方法として，これまではレジスタを使用してきました．例えば，1 文字出力サブルーチン (PUTCH) は，出力したい文字の文字コードを G0 に格納して呼び出されるようになっていました．文字列の場合はデータの量が多いので，レジスタにデータ全体を格納してサブルーチンに渡すことができません．そこで，文字列データの置いてあるアドレスを渡すことにします．

次のプログラムは，文字列出力サブルーチン (PUTSTR) が，出力する文字列のアドレスを受け取るようにした例です．G1 に文字列のアドレスをセットして，PUTSTR を呼び出します．なお，プログラム中「#STR1」のような記述は，STR1 ラベルの値を意味します．どのような機械語へ変換されているか，確認してください．

番地	機械語	ラベル	ニーモニック
02		SIOD	EQU 02H
03		SIOS	EQU 03H
00			
00	1F DC	START	LD SP,#0DCH
02	17 2F		LD G1,#STR1
04	B0 13		CALL PUTSTR
06	17 41		LD G1,#CRLF
08	B0 13		CALL PUTSTR
0A	17 39		LD G1,#STR2
0C	B0 13		CALL PUTSTR
0E	17 41		LD G1,#CRLF
10	B0 13		CALL PUTSTR
12	FF		HALT
13			
13			; 文字列出力ルーチン
13	D0	PUTSTR	PUSH G0
14	D4		PUSH G1
15	11 00	LOOP	LD G0,0,G1
17	53 00		CMP G0,#0
19	A4 21		JZ RETURN
1B	B0 24		CALL PUTCH
1D	37 01		ADD G1,#1
1F	A0 15		JMP LOOP
21	D6	RETURN	POP G1
22	D2		POP G0
23	EC		RET
24			
24			; 1 文字出力ルーチン
24	D4	PUTCH	PUSH G1
25	C4 03	WAIT	IN G1,SIOS
27	67 80		AND G1,#80H
29	A4 25		JZ WAIT
2B	C3 02		OUT G0,SIOD
2D	D6		POP G1
2E	EC		RET
2F			
2F	53 48	STR1	DC "SHIGEMURA",0
31	49 47		
33	45 4D		
35	55 52		
37	41 00		
39	54 45	STR2	DC "TETSUJI",0
3B	54 53		
3D	55 4A		
3F	49 00		
41	0D 0A	CRLF	DC 0DH,0AH,0
43	00		

TeC6 では CALL 命令の機械語が BC になります．

6.7.3　ジャンプテーブル

これまで，データのアドレスばかり考えてきましたが，プログラムのアドレスも同様に扱うことができます．インデクスドモードのアドレッシングは，ジャンプ命令や CALL 命令でも使用できます．

このことを利用すると，表に登録してあるいくつかのサブルーチンから一つを選んで実行する，プログラムをつくることができます．サブルーチンを登録してある表のことを，ここでは，ジャンプテーブルと呼びます．

ジャンプ命令を表にする場合

ジャンプ命令や CALL 命令のインデクスドアドレッシングは，ジャンプ先アドレスをインデクスレジスタを用いて計算します．例えば次の JMP 命令は，A+G2 番地にジャンプします．

```
      JMP   A,G2
```

つぎのプログラムは，G2 の値を使用して TBL 以下のどれか一つの JMP 命令を選択し，PRG0, PRG1, ... のいずれかにジャンプする例です．つまり，Java 言語や C 言語の switch 文に似た動きをします．JMP 命令は 2 バイトの命令なので，SHLL を用いて G2 の値を 2 倍にしています．

```
        LD    G2,N
        SHLL  G2
        JMP   TBL,G2
TBL     JMP   PRG0
        JMP   PRG1
          ...
```

CALL 命令を用いてサブルーチンを呼び出すこともできます．CALL 命令の次の命令にサブルーチンから戻ってき来るので，注意して下さい．

```
      LD    G2,N
      SHLL  G2
      CALL  TBL,G2
サブルーチンから戻る場所
          ...
TBL   JMP   PRG0
      JMP   PRG1
          ...
```

アドレスを表にする場合

ジャンプ命令を表にすると表の要素が 2 バイトになります．アドレスだけを表にすると表の要素が 1 バイトで済み，メモリの節約になります．

次のプログラムは，アドレスを表にした例です．G1 の値を使用し，表から PRG0, PRG1, ... のどれかのアドレスを選び G2 にロードします．次に，JMP 命令で G2 の内容が表す番地にジャンプします．

```
        LD    G1,N
        LD    G2,TBL,G1
        JMP   0,G2
TBL     DC    PRG0
        DC    PRG1
          ...
```

次のプログラムは，CALL 命令を用いてサブルーチンを呼び出す例です．

```
      LD    G1,N
      LD    G2,TBL,G1
      CALL  0,G2
サブルーチンから戻る場所
          ...
TBL   DC    PRG0
      DC    PRG1
          ...
```

問題

1. G1 レジスタで指定された番地から始まる G0 レジスタで指定されたバイト数の領域を，00H で埋めつくすサブルーチンを作りなさい．
2. C 言語の switch 文が，TeC の機械語でどのように表現できるか考察しなさい．

実効アドレス

本文では，演算，転送やジャンプの対象となるメモリのアドレスのことを,「データのアドレス」,「ジャンプ先アドレス」と書いてありますが，正しい用語では「実効アドレス (Effective Address)」と呼びます．また，Effective Address を EA と略して書くこともあります．命令表で命令の動作説明の欄にさかんに EA とか [EA] の記述が出てきますが，これは実効アドレスそのものや，実効アドレスに置いてあるデータのことを指しています．

6.8 割込み (Interrupt)

ノイマン型のコンピュータはプログラムを逐次実行しますので，仕事を順番に処理するのが基本です．しかし緊急の仕事が発生したとき等は，処理中の仕事を後回しにして，緊急の仕事を先に処理しなければならない場合があります．割込みは，そのような緊急の事象をコンピュータに伝えて，優先される処理を割り込ませる仕組です．英語では割り込みのことを，Interrupt と言います．

割込みは，CPU の外部で発生した事象をハードウェアにより CPU に伝えることにより発生します．割込みにはいくつかの種類があり，CPU には割込みの種類毎に入力が用意されています．

割込みが発生すると，CPU は現在実行中のプログラムを中断し，予め登録された割込み処理プログラム (「割込み処理ルーチン」とも呼びます) の実行を開始します．割込み処理ルーチンは，割込みの種類毎に登録することができます．



6.8.1 TeC の割込み種類

TeC の場合，割込みの種類は 4 種類です．四つの割込みには，“INT0”，“INT1”，“INT2”，“INT3” という名前が付いています．各割込みの意味は TeC7 と TeC6 で少し異なっており，TeC7 の場合と TeC6 の場合について以下で別々に解説します．

割込みの一覧表には割込みベクタアドレスが書いてあります．ベクタには，割込み毎に割込み処理ルーチンを登録します．例えば “INT0” の割込み処理ルーチンが 12H 番地から始まる場合は，メモリの DCH 番地に，12H を書き込みます．

TeC7 の割込み

TeC7 で使用できる割込の一覧は次の表の通りです．“INT0” はインターバルタイマに接続されています．インターバルタイマは指定された時間間隔で繰返し割込みを発生します．指定できる間隔は 1/75 秒〜256/75 秒の範囲です．“INT1” は SIO に接続されており SIO が 1 文字を受信完了する毎に割込みが発生します．“INT2” も SIO に接続されており，SIO が 1 文字を送信完了する毎に割込みが発生します．“INT3” はコンソールパネルに接続されています．プログラム実行中にユーザがコンソールパネルの WRITE スイッチを押すと割込みが発生します．

TeC7 の割込み種類

名前	意味	ベクタ
INT0	インターバルタイマ割込み	DCH
INT1	SIO 受信割込み	DDH
INT2	SIO 送信割込み	DEH
INT3	コンソール割込	DFH

TeC6 の割込み

TeC6 で使用できる割込の一覧は次の表の通りです．“INT0” は拡張ポートに接続されています．拡張ポートにハンダ付け練習基板を接続すると，約 1 秒毎に割込みが発生するタイマ割込みとして機能します．“INT1” は SIO に接続されており，SIO が 1 文字の受信を完了する毎に割込みが発生します．“INT2” も SIO に接続されており，SIO が 1 文字の送信を完了する毎に割込みが発生します．“INT3” も拡張ポートに接続されています．拡張ポートに接続した回路により割込みの意味が決まります．

TeC6 の割込み種類

名前	意味	ベクタ
INT0	外部割込み (タイマ)	DCH
INT1	SIO 受信割込み	DDH
INT2	SIO 送信割込み	DEH
INT3	外部割込み	DFH

6.8.2 TeC の割込み動作

TeC の CPU は，新しい命令を実行する直前に割込みが発生していないかチェックします．割込みが発生していた場合は次のような動作をします．

1. PC をスタックに保存 (PUSH) する．
2. 割込み許可フラグをリセットする．(割込み禁止にする)
3. 割込みの種類 (INT0〜INT3) に対応するベクタから，割込み処理ルーチンのアドレスを読む．
4. 割込み処理ルーチンのアドレスを PC にセットする．

6.8.3　EI(Enable Interrupt) 命令

Enable は「可能にする」という意味の英語です．EI 命令は割込み許可フラグをセットし割り込みを「可能にする」します．

EI 命令は CPU が割込みを受け付ける準備ができたとき，割込みを許可するために使用します．CPU は，リセットされると割込み禁止状態になります．プログラムは，SP の初期化，割込みベクタの設定等をしたあと，EI 命令を用いて割込みを許可します．

割込みが許可状態かそうでないかは,「割込み許可フラグ」の状態によって決まります．EI 命令は「割込み許可フラグ」をセットします．

意味： 割込みを許可する．(割込み許可フラグをセットする．)

フラグ： C，S，Z フラグは変化しません．割込み許可フラグがセットされます．

ニーモニック： `EI`

命令フォーマット： EI 命令は，1 バイト長の命令です．

第 1 バイト	
OP	GR XR
1110_2	00 00

フローチャート： 次のように描くことにします．



6.8.4　DI(Disable Interrupt) 命令

Disable は,「できなくする」という意味の英語です．DI 命令は割込み許可フラグをリセットし割り込みを「できなく」します．

CPU が割込みを受け付けると都合が悪いとき，DI 命令を使用して割込みを禁止します．EI 命令を用いて割込みを許可したあと，割込みが発生すると都合が悪くなったときに使用します．

割り込みが発生し，割り込み処理ルーチンの実行が始まった時は，自動的に割り込み禁止状態になります．それ以外で，割り込み禁止にしたいとき使用する命令です．

意味： 割込みを禁止する．(割込み許可フラグをリセットする．)

フラグ： C，S，Z フラグは変化しません．割込み許可フラグがリセットされます．

ニーモニック： `DI`

命令フォーマット： DI 命令は，1 バイト長の命令です．

第 1 バイト	
OP	GR XR
1110_2	00 11

フローチャート： 次のように描くことにします．



6.8.5　RETI(Return from Interrupt) 命令

割込み処理ルーチンから通常のプログラムへ戻るための，特別な RET 命令です．RETI 命令を実行して割込み処理ルーチンが終了すると，CPU は割込み許可状態になり，割込みが発生したとき実行中だったプログラムに戻ります．

意味： 割込み処理ルーチンから戻る．

フラグ： C，S，Z フラグは変化しません．割込み許可フラグがセットされます．

ニーモニック： `RETI`

命令フォーマット： RETI 命令は，1 バイト長の命令です．

第 1 バイト	
OP	GR XR
1110_2	11 11

動作の詳細： スタックから値を一つ取り出し PC にセットし，割込み許可フラグをセットします．

フローチャート： 次のフローチャートの，最後の角を取った四角が RETI 命令に対応します．フローチャートは，割り込み処理ルーチン全体を示しています．



6.8.6 PUSHF(Push Flag) 命令

フラグをスタックに保存する命令です．主に割込み処理ルーチンで使用します．割込みは，いつ発生するか分かりません．割込み処理ルーチンが終了したあと，以前のプログラムに戻って処理を続行するには，CPU の状態を完全に元通りに復元する必要があります．CPU の状態とはレジスタ (PC，SP を含む) とフラグの値のことです．

レジスタの値は，PUSH 命令でスタックに退避することができます．フラグの値は，PUSHF 命令でスタックに退避します．

意味： フラグをスタックに退避する．

フラグ： フラグは変化しません．

ニーモニック： `PUSHF`

命令フォーマット： PUSHF 命令は，1 バイト長の命令です．

第 1 バイト	
OP	GR XR
1101_2	11 01

動作の詳細： スタックに退避されるデータは，次のような形式の 1 バイトになります．ここで，E が割込み許可フラグです．

E	0	0	0	0	C	S	Z

フローチャート： 次のように描くことにします．



6.8.7 POPF(Pop Flag) 命令

フラグをスタックから復元する命令です．主に割込み処理ルーチンで使用します．割込み処理ルーチンは，POPF 命令を使用してフラグの値を復元してから，RETI 命令で以前のプログラムに戻ります．

意味： フラグをスタックから復元します．

フラグ： フラグは，スタックから取り出した値により変化します．

ニーモニック： `POPF`

命令フォーマット： POPF 命令は，1 バイト長の命令です．

第 1 バイト	
OP	GR XR
1101_2	11 11

動作の詳細： スタックから取り出すデータは，PUSHF 命令と同じ形式です．

フローチャート： 次のように描くことにします．



6.9　タイマ割込み

一般に，コンピュータには，割込みを発生するタイマが内蔵されています．時間を計ったり，複数の仕事を切替えるきっかけとして，タイマから発生する割込みを使用するからです．

パソコン上では，複数のプログラムが同時に実行されているように見えますが，これは 1/10 〜 1/100 秒程度の間隔で実行するプログラムを切替えることにより，複数のプログラムを同時に実行しているように見せているのです．このとき，プログラムを切替えるきっかけになるのは，タイマからの割込みです．

TeC7 と TeC6 でタイマ割込みのハードウェアが大きく異なります．そこで，以下では TeC7 の場合と TeC6 の場合に分けて説明します．

6.9.1　TeC7 のタイマ割込み

TeC7 にはプログラムから周期を設定できるインターバルタイマが一つ内蔵されています．インターバルタイマは指定した時間間隔で繰返し割込みを発生します．インターバルタイマには 8 ビットのタイマ・カウンタと 8 ビットの周期レジスタが内蔵されています．

I/O ポート (TeC7 の場合)

インターバルタイマの I/O ポートは，4 番地，5 番地に配置されています (I/O マップ参照)．

1. タイマ・カウンタ
   4 番地を IN 命令で読むとタイマ・カウンタの現在値を読み出すことができます．タイマ動作中，この値は 1/75 秒毎にカウントアップされます．そして値が周期レジスタの値と一致したら 0 に戻ります．その後，カウントアップが再開されます．



2. 周期レジスタ
   周期レジスタ値によって割込みの発生周期が決まります．4 番地に OUT 命令で書き込んだ値が周期レジスタに書き込まれます．書き込む値を n とすると周期は次の式で計算できます．

   周期 $= (n + 1)/75$ 秒

   周期レジスタの初期値は 1 秒に相当する 74 です．「周期」を変更するとタイマはカウントを停止します．下記のコントロールを使用してスタートし直す必要があります．



3. ステータス
   5 番地を IN 命令で読むとタイマがスタートしてから一周期の時間が経過したかどうかを割込みを用いることなく知ることができます．一周期時間が経過すると b_7 の値が “1” に変化します．b_7 は「ステータス」を一度読みだすと “0” にクリアされます．その後一周期時間が経過する度に b_7 の値が “1” に変化します．



4. コントロール
   タイマ割込みの (許可／不許可)，タイマカウンタのカウント (スタート／ストップ) の情報を 5 番地に OUT 命令で書き込みます．b_7 が割込みの許可ビット，b_0 がタイマカウンタのスタートビットです．タイマカウントをスタートすると，タイマカウンタの値が “00H” にリセットされてからカウントが始まります．



プログラミング例 (TeC7 の場合)

インターバルタイマを用いたプログラムの例として，1 秒間隔でブザーの ON/OFF を繰り返すプログラムを紹介します．

タイマ割込みプログラムの例 (TeC7 用)

```
ラベル    ニーモニック          コメント
BUZ       EQU     00H         ; ブザーポート
TMRVCT    EQU     0DCH        ; 割込みベクタ
TMRCNT    EQU     4           ; カウンタ
TMRCTR    EQU     5           ; コントロール
START     ; メインプログラム
          LD      SP,#0DCH    ; SP 初期化
          LD      G0,#TMRINT  ; 割込ルーチンを
          ST      G0,TMRVCT   ; ベクタに登録
          LD      G0,#74      ; タイマ初期化
          OUT     G0,TMRCNT   ; 周期=1 秒,
          LD      G0,#81H     ; 割込み許可,
          OUT     G0,TMRCTR   ; カウント開始
          EI                  ; CPU の割込許可
WAIT      LD      G0,FLG      ; フラグを
          CMP     G0,#0       ; チェック
          JZ      WAIT        ; 0 の間は待つ
          LD      G0,VAL      ; ブザーポートの
          XOR     G0,#01H     ; 値を反転して
          ST      G0,VAL      ; 記録すると
          OUT     G0,BUZ      ; 同時に出力
          LD      G0,#0       ; フラグをクリア
          ST      G0,FLG      ; 次の割込みを
          JMP     WAIT        ; 待つ

          ; 共有変数
FLG       DC      0           ; フラグ
VAL       DC      0           ; ブザーポートの
                              ; 現在値ひかえ
          ; 割込みルーチン
TMRINT    PUSHF               ; コンテキストを
          PUSH    G0          ; 保存する
          LD      G0,#1       ; フラグに
          ST      G0,FLG      ; 1 を書き込む
          POP     G0          ; コンテキストを
          POPF                ; 復元し
          RETI                ; 割込み終了
```

上のプログラムの内容を説明します．**メインプログラム**は，まず，スタックポインタの値を初期化します．割込みが発生すると CPU が自動的に PC をスタックに保存するので，割込みが発生する前に，**必ず，スタックポインタの値を初期化する必要があります．**

次に，INT0 の割込みベクタに割込みルーチンの番地を書き込みます．DCH 番地が INT0 の割込みベクタですので，この番地に割込み処理ルーチンの先頭番地を書き込みます．

更に，インターバルタイマの初期化を行います．1 秒ごとの割込みが必要なので 74 を周期レジスタに書き込みます．続けてコントロールに 81H を書き込むことで割込み出力が可能な状態でタイマがスタートします．

最後に，EI 命令を実行して CPU を割込み許可状態にします．これで，割込みを受け付ける準備ができました．ループを実行して割込みの発生を待ちます．TeC は，CPU が停止しているときは割込みを受け付けませんので，プログラムを終了しないで割込みの発生を待つ必要があります．

割込みが発生すると CPU は PC の値をスタックに保存した後，割込みベクタに書いてある番地にジャンプします．メインプログラムが INT0 の割込みベクタに割込みルーチンの先頭番地を書き込んでいるので，タイマ割込み (INT0) が発生すると，割込みルーチンへジャンプします．

割込みルーチンは，メインプログラムを実行中の CPU の状態 (コンテキスト) を保存した後，割込み処理を行います．このプログラムの場合，割込み処理の内容は FLG 変数に “1” を書き込みむことです．処理が終わったら CPU の状態 (コンテキスト) を元に戻してメインプログラムに RETI 命令で戻ります．

メインプログラムは FLG 変数の値が “1” に変化するとブザーの ON/OFF を変更した後，FLG 変数を “0” に戻します．そして，割込み待ちループに戻ります．

結局，割込みルーチンが「1 秒に一度」FLG 変数をセットし，そのたびにメインプログラムがブザーの ON/OFF を変更することになります．

6.9.2 TeC6 のタイマ割込み

TeC6 付属のハンダ付け練習基板を TeC6 の拡張コネクタに接続すると，約 1 秒間隔で割込みを発生するインターバルタイマとして動作します．練習基板から拡張コネクタを通して INT0 割込み信号が入力されます．TeC6 が割込み許可命令 (EI) を実行するだけで 1 秒間隔で INT0 割込みが発生します．

割込みの発生間隔を変更することはできません．また，タイマ以外の割込みを使用するために EI 命令で CPU を割込許可状態にすると，タイマ割込みも発生してしまいます．タイマ割込みだけを禁止するには基板を取り外す以外に方法がありません．

次に TeC6 のタイマ割込みの応用例として，ブザーを 1 秒間隔で ON/OFF するプログラムを紹介します．実は，このプログラムは前出の TeC7 用のものと比較して，タイマ初期化部分 4 行が無いだけです．このプログラムの解説は前出の TeC7 のものを参照して下さい．

タイマ割込みプログラムの例 (TeC6 用)

```
ラベル    ニーモニック          コメント
BUZ       EQU     00H         ; ブザーポート
TMRVCT    EQU     0DCH        ; 割込みベクタ
START     ; メインプログラム
          LD      SP,#0DCH    ; SP 初期化
          LD      G0,#TMRINT  ; 割込ルーチンを
          ST      G0,TMRVCT   ; ベクタに登録
          EI                  ; CPU の割込許可
WAIT      LD      G0,FLG      ; フラグを
          CMP     G0,#0       ; チェック
          JZ      WAIT        ; 0 の間は待つ
          LD      G0,VAL      ; ブザーポートの
          XOR     G0,#01H     ; 値を反転して
          ST      G0,VAL      ; 記録すると
          OUT     G0,BUZ      ; 同時に出力
          LD      G0,#0       ; フラグをクリア
          ST      G0,FLG      ; 次の割込みを
          JMP     WAIT        ; 待つ

          ; 共有変数
FLG       DC      0           ; フラグ
VAL       DC      0           ; ブザーポートの
                              ; 現在値ひかえ
          ; 割込みルーチン
TMRINT    PUSHF               ; コンテキストを
          PUSH    G0          ; 保存する
          LD      G0,#1       ; フラグに
          ST      G0,FLG      ; 1 を書き込む
          POP     G0          ; コンテキストを
          POPF                ; 復元し
          RETI                ; 割込み終了
```

問題

1. 「タイマ割込みプログラムの例」の動作を確認しなさい．
2. “LD G0,#TMRINT” は，どんな機械語に変換されたか確認しなさい．

例題 6-5 3 分間タイマ

問題： タイマ割込みを使用した，3 分間タイマを作りなさい．タイマは，3 分経過したらブザーを鳴らしっぱなしにするものとします．

考え方： 1 秒間隔でタイマ割込みが発生するようにし割込みの回数で時間を計ります．割込みルーチンでカウンタの値をインクリメントし，180 回目の割込みが発生したらブザーを鳴らしたままプログラムを終了します．ブザーはリセットボタンを押して止めます．

プログラム： 割込みルーチンは，180 をカウントしたらフラグ (FLAG) セットします．メインルーチンは，フラグがセットされたらブザーを ON にして，プログラムを終了します．なお，プログラムは TeC7 用ですが，TeC6 で実行することも可能です．

```
ラベル    ニーモニック          コメント
TMRVCT    EQU     0DCH
TMRCNT    EQU     4           ; カウンタ
TMRCTR    EQU     5           ; コントロール
          ; メインプログラム
START     LD      SP,#0DCH    ; SP 初期化
          LD      G0,#0       ; フラグと
          ST      G0,FLAG     ; カウンタ
          ST      G0,CNT      ; リセット
          LD      G0,#TMRINT
          ST      G0,TMRVCT   ; ベクタ設定
          LD      G0,#74      ; タイマ初期化
          OUT     G0,TMRCNT   ; 周期=1 秒
          LD      G0,#81H     ; 割込み許可
          OUT     G0,TMRCTR   ; カウント開始
          EI                  ; 割込み許可
L1        LD      G0,FLAG     ; フラグが 0
          CMP     G0,#0       ; の間
          JZ      L1          ; 繰り返す
          LD      G0,#01H
          OUT     G0,00H      ; ブザー ON
          HALT                ; 終了
          ; 割込み処理ルーチン
TMRINT    PUSHF               ; コンテキス
          PUSH    G0          ; スト保存
          LD      G0,CNT      ; 回数のカウ
          ADD     G0,#1       ; ント
          ST      G0,CNT
          CMP     G0,#180     ; 3 分経過
          JZ      TINT1
          JMP     TINT2
TINT1     ST      G0,FLAG     ; フラグ ON
TINT2     POP     G0          ; コンテキ
          POPF                ; スト復元
          RETI
          ; 作業領域
FLAG      DS      1           ; フラグ
CNT       DS      1           ; カウンタ
```

6.10 コンソール割込み (TeC7)

TeC7 では，コンソールパネルの操作で INT3 割込みを発生することが可能です．(**この機能は TeC6 にはありません．**) コンソール割込みを許可する手順は次の通りです．

1. スタックポインタを初期化する．
2. INT3 割込みベクタに，コンソール割込み処理ルーチンの番地を書き込む．
3. コンソール割込み (INT3) 許可ビットを ON にする．割込み許可ビットは，I/O 6 番地の b_0 です．(「5.11.1 I/O マップ」参照)
4. EI 命令で CPU の割込みを許可する．

この状態で，プログラム実行中にコンソールパネルの WRITE ボタンを押すと割込が発生します．

コンソール割込みの応用例

以下にコンソール割込みを応用したプログラムの例を示します．このプログラムはコンソール割込みとタイマ割込みの両方を使用した例になっています．

このプログラムは，コンソールパネルの WRITE ボタンが押されるたびに 0.3 秒間ブザーを鳴らします．以下の手順で，機能が実現されています．

1. 初期化
   メインプログラムはスタックポインタとタイマ等のハードウェアを初期化し，CPU を割込み許可状態にします．初期化が完了したら割込み待ちの無限ループに入ります．
2. コンソール割込み
   ユーザが WRITE ボタンを押して割込みが発生すると，コンソール割込みルーチンが起動されます．コンソール割込みルーチン中でブザーを ON にした上でタイマをスタートします．
3. タイマ割込み
   タイマがスタートして 0.3 秒後にタイマ割込が発生します．タイマ割り込みルーチンではブザーを OFF にしてタイマを停止します．

コンソール割込みプログラムの例 (TeC7 用)

```
ラベル    ニーモニック          コメント
BUZ       EQU     00H         ; ブザーポート
TMRVCT    EQU     0DCH        ; TMR 割込みベクタ
CONVCT    EQU     0DFH        ; CON 割込みベクタ
TMRCNT    EQU     4           ; TMR(周期)
TMRCTR    EQU     5           ; TMR コントロール
CONCTR    EQU     6           ; CON コントロール
START     ; メインプログラム
          LD      SP,#0DCH    ; SP 初期化
          LD      G0,#CONINT  ; コンソール割込み
          ST      G0,CONVCT   ; ルーチン登録
          LD      G0,#01H     ; コンソール割込み
          OUT     G0,CONCTR   ; 許可
          LD      G0,#TMRINT  ; タイマ割込み
          ST      G0,TMRVCT   ; ルーチン登録
          LD      G0,#25      ; 周期=0.3SEC
          OUT     G0,TMRCNT   ; タイマ周期セット
          EI
WAIT
          JMP     WAIT        ; 無限ループ

          ; コンソール割込みルーチン
CONINT    PUSHF               ; コンテキストを
          PUSH    G0          ; 保存する
          LD      G0,#01H     ;
          OUT     G0,BUZ      ; BEEP ON
          LD      G0,#81H     ;
          OUT     G0,TMRCTR   ; TIMER START
          POP     G0          ; コンテキストを
          POPF                ; 復旧する
          RETI                ; 戻る

          ; タイマ割込みルーチン
TMRINT    PUSHF               ; コンテキストを
          PUSH    G0          ; 保存する
          LD      G0,#00H     ;
          OUT     G0,BUZ      ; BEEP OFF
          LD      G0,#00H     ;
          OUT     G0,TMRCTR   ; TIMER STOP
          POP     G0          ; コンテキストを
          POPF                ; 復旧する
          RETI                ; 戻る
```

6.11　入出力割込み

一般に，CPU の処理速度と入出力装置の動作速度には大きな差があるので，入出力装置を動かすとき CPU は長い待ち状態になります (囲み記事「入出力装置と CPU の速度差」参照)．入出力装置の**動作完了**を待って，CPU がループを回り続けるだけでは時間が勿体ないので，その間に他の仕事ができる仕組みが望まれます．

そこで，入出力装置は**動作が完了**したことを，CPU に割込みで知らせることができるように設計してあります．キーボード，マウス，ハードディスク，ネットワーク等のほとんどの入出力装置が，動作完了の割込みを発生する機構を持っています．このような割込みを**「入出力割込み」**と呼びます．

6.11.1　TeC の入出力割込み

TeC は SIO から入出力割込みを発生することができます．(TeC6 の場合は，拡張ポートに接続した拡張回路から入出力割込みを発生することもできます．)

SIO からの割込みは，文字を受信したとき発生する「受信割込み」と，文字を送信可能になったとき発生する「送信割込み」の 2 種類があります．これらの割込みを使用するかしないかは，割込み許可ビットにより，別々に制御できるようになっています．

SIO 受信割込み

TeC の SIO は文字を受信したとき，INT1 割込みを発生することが可能です．割込みの発生を許可する手順は次の通りです．

1. スタックポインタを初期化する．
2. INT1 割込みベクタに，SIO 受信割込み処理ルーチンの番地を書き込む．
3. SIO の受信割込み許可ビットを ON にする．割込み許可ビットは，I/O 3 番地の b6 です．(「5.14.4 I/O ポート」参照)
4. EI 命令で CPU の割込みを許可する．

文字を受信すると INT1 割込みが発生します．受信した文字を，I/O マップ 3 番地の SIO 受信データレジスタ (「5.11.1 I/O マップ」参照) から読み出して下さい．データを読み出さないで割込み処理ルーチンから戻ると，再度，割込み処理ルーチンが呼び出されるので注意してください．

SIO 送信割込み

TeC の SIO は文字を送信することが可能なとき，INT2 割込みを発生することができます．割込みの発生を許可する手順は次の通りです．

1. スタックポインタを初期化する．
2. INT2 割込みベクタに，SIO 送信割込み処理ルーチンの番地を書き込む．
3. SIO の送信割込み許可ビットをに ON する．割込み許可ビットは，I/O 3 番地の b7 です．(「5.14.4 I/O ポート」参照)
4. EI 命令で CPU の割込みを許可する．

次の文字を送信することが可能になると，INT2 割込みが発生します．送信する文字を，I/O マップ 3 番地の SIO 送信データレジスタ (「5.11.1 I/O マップ」参照) に書き込んでください．データを書き込まないで割込み処理ルーチンから戻ると，再度，割込み処理ルーチンが呼び出されるので注意してください．これ以上，送信する文字が無い場合は，割込み許可ビットを OFF にしてください．

6.11.2　入出力割込みの使用例

SIO を割込み駆動で使用する例を示します．

SIO 送信割込みの使用

ローマ字で自分の名前を出力するプログラムを，割込みを使用して作成します．(例題 5-7(P.56)，例題 6-3(P.65)，例題 6-4(P.66) と同様なプログラムです．) プログラムは下の通りです．プログラムの動きは次の通りです．

1. メインルーチンで割込みを許可すると，すぐに最初の割込みが発生する．

2. 割込み処理ルーチンの中で，最初の文字を SIO データレジスタに書き込む．
3. その後は，1 文字の送信が終了する毎に割込みが発生するので，割込み処理ルーチンの中で SIO データレジスタに次の文字を書き込む．
4. 全ての文字を送信したら，SIO の送信割込み許可ビットを OFF にし，それ以上の割込みが発生しないようにする．

```
ラベル      ニーモニック        コメント
INT2V       EQU     0DEH
SIOD        EQU     02H
SIOC        EQU     03H
            ;メインプログラム
START       LD      SP,#0DCH    ; SP 初期化
            LD      G0,#SIOXINT ; 割込みルー
            ST      G0,INT2V    ; チンを登録
            LD      G0,#80H     ; SIO 送信割
            OUT     G0,SIOC     ; 込み許可 ON
            LD      G0,#DATA    ; データの位
            ST      G0,POINTER  ; 置をセット
            LD      G0,#1       ; 出力中 Flag
            ST      G0,FLAG     ; セット
            EI                  ; 割込み許可
LOOP        LD      G0,FLAG     ; 出力中フラ
            CMP     G0,#0       ; グがクリア
            JZ      END         ; で終了
            JMP     LOOP        ; 割込み待ち
END         HALT
DATA        DC      "SHIGEMURA"
            DC      0
            ; SIO 送信割込み処理ルーチン
SIOXINT     PUSHF
            PUSH    G0
            PUSH    G1
            LD      G1,POINTER  ; 一文字
            LD      G0,0,G1     ; 取り出す
            CMP     G0,#0       ; 文字列の
            JZ      OWARI       ; 終わりか？
            OUT     G0,SIOD     ; 送信する
            ADD     G1,#1       ; 次の文字
            ST      G1,POINTER  ; に進める
            JMP     KAERI
OWARI       LD      G0,#00H     ; SIO 送信割
            OUT     G0,SIOC     ; 込み許可 OFF
            ST      G0,FLAG     ; Flag クリア
KAERI       POP     G1
            POP     G0
            POPF
            RETI
            ; 割込み処理ルーチンのデータ
POINTER     DS      1           ; 文字列位置
FLAG        DS      1           ; 動作中 FLAG
```

SIO 受信割込みの使用

次は，繰り返しブザーを鳴らすプログラムです．ブザーを鳴らす間隔を SIO から受信した文字 (アルファベット小文字) によって切替えることができます．リストでは省略していますが,「6.4.3 0.2 秒タイマ」で作成した MS200 サブルーチンを利用しています．実際に実行する場合は MS200 サブルーチンも打ち込んで下さい．

```
ラベル      ニーモニック        コメント
INT1V       EQU     0DDH
BUZZ        EQU     00H
SIOD        EQU     02H
SIOC        EQU     03H
            ;メインプログラム
START       LD      SP,#0DCH    ; SP 初期化
            LD      G0,#SIORINT ; 割込みルーチン
            ST      G0,INT1V    ; を登録
            LD      G0,#40H     ; SIO 受信割込み
            OUT     G0,SIOC     ; 許可ビット ON
            LD      G0,#'a'     ; 最後に入力した
            ST      G0,IKEY     ; 文字='a'
            EI                  ; 割込み許可
LOOP        LD      G0,IKEY     ;
            ST      G0,KEY
            CMP     G0,#'z'     ; 'z' が入力され
            JZ      END         ; たら終了
            LD      G0,#'a'
L1          CALL    MS200       ; 200MS 待ち
            CMP     G0,KEY
            JZ      L2
            ADD     G0,#1
            JMP     L1          ; 割込み待ち
L2          LD      G0,#1
            OUT     G0,BUZZ     ; ブザー ON
            LD      G0,#'a'
L3          CALL    MS200
            CMP     G0,KEY
            JZ      L4
            ADD     G0,#1
            JMP     L3
L4          LD      G0,#0
            OUT     G0,BUZZ     ; ブザー OFF
            JMP     LOOP
END         HALT
KEY         DS      1
IKEY        DS      1
            ;SIO 受信割込み処理ルーチン
SIORINT     PUSHF
            PUSH    G0
            IN      G0,SIOD     ; 入力した文字を
            ST      G0,IKEY     ; IKEY に格納する
            POP     G0
            POPF
            RETI
```

入出力装置と CPU の速度差
一般に，CPU に比較して入出力装置はデータの処理に長い時間がかかります．例えば TeC の SIO (「5.14 TeC のシリアル入出力」参照) は $9,600bau$ で動作するので，1 文字を入出力するのに約 1ms の時間が必要です．1ms は TeC の 2,458 ステートに相当します (「6.4.2 1ms タイマ」参照)．1 機械語命令が平均 5 ステートだと仮定すると，約 491 命令を実行できる時間になります．491 命令分の時間を SIO の**動作が完了**したか調べるだけのループで費すのは勿体ないことです．その間に別の仕事をできる工夫が望まれます．

6.12　マシンステートとスピーカ

TeC には，音を鳴らすためのスピーカが準備されています．スピーカには OUT 命令でアクセスします．ここでは，マシンステートを意識したプログラミングを行い，スピーカを鳴らします．

6.12.1　スピーカーの仕組み

次の図に示すように，I/O アドレス 1 番地 (I/O マップ参照) の LSB(b0) には 1 ビットのレジスタが存在します．このレジスタの 1/0 によりスピーカに加わる電圧が変化するようになっています．なお，図では省略しましたが,「5.11.3 OUT(Output) 命令」で紹介した「ブザー音源」も，同じスピーカに接続されています．



スピーカの構造は次のようになっていて，加えた電圧により振動板が前後に移動します．



レジスタの値を，$1, 0, 1, 0, 1, 0, 1, 0$ と変化させることにより，振動板が前後に振動して音がでます．

1kHz の音を出したい時は，プログラムで $1, 0$ 書き込みを 1 秒間に 1000 回 (1000 サイクル) 行う必要があります．

6.12.2　スピーカ駆動プログラム

プログラムで 1 秒間に 1000 回のペースで $1, 0$ の書き換えをするためには，時間を計って 1/2000 秒毎に 1 度，I/O ポートの書き換えが行われるようにする必要があります．

マシンステートを使用して時間を計り，一定の時間毎にスピーカポートの 0/1 を書き換えるプログラムを作ります．書き換えの間隔によって音の高さが決まります．

例題 6-6 1kHz ブザー

問題： スピーカーから 1kHz の音を出すプログラムを作りなさい．

考え方： 「6.4.2 1ms タイマ」では，$1ms$ のタイマを作りました．今回の問題では，半分の $0.5ms$ 毎にスピーカ出力の 0/1 を書き換えるようにします．

フローチャート： 次のようになります．



プログラム： 若干のステート数の増減は誤差と考えて無視します．プログラムは，次のようになります．

番地	機械語	ラベル	ニーモニック
00			; 1kHz ブザー
01		SPK	EQU 01H
00	1F DC	START	LD SP,#0DCH
02	B0 10	LOOP	CALL MS5
04	17 01		LD G1,#01H
06	C7 01		OUT G1,SPK
08	B0 10		CALL MS5
0A	17 00		LD G1,#00H
0C	C7 01		OUT G1,SPK
0E	A0 02		JMP LOOP
10			; 0.5ms サブルーチン
10	D0	MS5	PUSH G0
11	13 57		LD G0,#87
13	43 01	MATU	SUB G0,#1
15	A4 19		JZ KAERU
17	A0 13		JMP MATU
19	D2	KAERU	POP G0
1A	EC		RET

TeC6 では CALL 命令の機械語が BC になります．

問題

$200Hz$ ブザーを作りなさい．

6.12.3 電子オルゴールプログラム

スピーカ出力プログラムの応用として，電子オルゴールプログラムを紹介します．電子オルゴールプログラムでは，曲の楽譜をデータとして準備します．データは，音の高さを決めるためのループの繰り返し回数と，音の長さを決めるためのループの繰り返し回数を組み合わせたものです．

フローチャート

プログラムは 3 重のループになっています．最も内側の二つのループが，スピーカに 1 を出力している時間と，0 を出力している時間を計っています．2 番目のループの繰り返しにより，音の継続時間を制御します．最も外側のループは曲のデータを順番に取り出しています．



音符データ

最も内側のループは 20 ステートあります．このループの繰り返し回数で時間を計って，スピーカへの出力を音波波形の半サイクル毎に書き換えます．

例えば「ラ」の音は $440Hz$ ですので，半サイクルの時間は $2,457,600/440/2 = 2,792$ ステート になります．半サイクルの時間をループで計っているわけですから，ループの繰り返し回数は，$2,792/20 = 139 = 8B_{16}$回 となります．この値が，音の高さを決めます．



音の長さは，音の出力を何サイクル行うかにより決めます．例えば「ラ」の音は $440Hz$ なので，$220 = DC_{16}$サイクル で 0.5 秒 になります．このプログラムでは，0.5 秒 を 4 分音符一つの時間としました．「ラ」以外の音も，同様に対応するデータ値を次の表のように決めました．

音	周波数 (Hz)	長さ	高さ
ド	262	83	E8
ド#	277	8B	DD
レ	294	93	D1
レ#	311	9C	C5
ミ	330	A5	BA
ファ	349	AF	AF
ファ#	370	B9	A6
ソ	392	C4	9C
ソ#	415	D0	93
ラ	440	DC	8B
ラ#	466	E9	83
シ	494	F7	7C
ド	523	FF	75

プログラム

完成したプログラムとデータを次ページに示します．プログラム中の ORG 命令は，プログラムのアドレスを指定する命令です．この場合，データを 30H 番地に生成するために使用しました．

このプログラムは，休符が表現できない，同じ高さの音が連続すると一つの音になってしまう等の問題があります．改良を考えて下さい．

電子オルゴールプログラム

番地	機械語	ラベル	ニーモニック	
01		SPK	EQU 01H	
00				
00			; 電子オルゴール	
00	17 2E		LD G1,#TABLE-2	
02	37 02	L9	ADD G1,#2	
04	19 00		LD G2,0,G1	
06	5B 00		CMP G2,#0	
08	A4 2C		JZ L16	
0A	11 01	L10	LD G0,1,G1	; 7
0C	00	L11	NO	; 3
0D	00		NO	; 3
0E	43 01		SUB G0,#1	; 5
10	A4 14		JZ L12	; 4/5
12	A0 0C		JMP L11	; 5
14	13 01	L12	LD G0,#01H	; 5
16	C3 01		OUT G0,SPK	; 8
18	11 01	L13	LD G0,1,G1	; 7
1A	00	L14	NO	; 3
1B	00		NO	; 3
1C	43 01		SUB G0,#1	; 5
1E	A4 22		JZ L15	; 4/5
20	A0 1A		JMP L14	; 5
22	13 00	L15	LD G0,#00H	; 5
24	C3 01		OUT G0,SPK	; 8
26	4B 01		SUB G2,#1	; 5
28	A4 02		JZ L9	; 4/5
2A	A0 0A		JMP L10	; 5
2C	FF	L16	HALT	
2D				
2D			; ドレミの歌	
30		TABLE	ORG 30H	
30	C5 E8		DC 0C5H,0E8H	; ド
32	49 D1		DC 049H,0D1H	; レ
34	F7 BA		DC 0F7H,0BAH	; ミ
36	41 E8		DC 041H,0E8H	; ド
38	A5 BA		DC 0A5H,0BAH	; ミ
3A	83 E8		DC 083H,0E8H	; ド
3C	A5 BA		DC 0A5H,0BAH	; ミ
3E	A5 BA		DC 0A5H,0BAH	; ミ
40	DC D1		DC 0DCH,0D1H	; レ
42	52 BA		DC 052H,0BAH	; ミ
44	57 AF		DC 057H,0AFH	; ファ
46	57 AF		DC 057H,0AFH	; ファ
48	52 BA		DC 052H,0BAH	; ミ
4A	49 D1		DC 049H,0D1H	; レ
4C	AF AF		DC 0AFH,0AFH	; ファ
4E	AF AF		DC 0AFH,0AFH	; ファ
50	AF AF		DC 0AFH,0AFH	; ファ
52	AF AF		DC 0AFH,0AFH	; ファ
54	F7 BA		DC 0F7H,0BAH	; ミ
56	57 AF		DC 057H,0AFH	; ファ
58	C4 9C		DC 0C4H,09CH	; ソ
5A	62 9C		DC 062H,09CH	; ソ
5C	52 BA		DC 052H,0BAH	; ミ
5E	C4 9C		DC 0C4H,09CH	; ソ
60	A5 BA		DC 0A5H,0BAH	; ミ
62	C4 9C		DC 0C4H,09CH	; ソ
64	C4 9C		DC 0C4H,09CH	; ソ
66	00		DC 000H	; 終了
67				

付 録A　電子オルゴールプログラムの実行例

以下は，図 A.1 の電子オルゴールプログラムを入力して実行する手順です．少し，音痴なオルゴールですが TeC から音楽が流れます．

A.1　プログラムの打ち込み

電子オルゴールプログラムは予め作ってあります．(図 A.1 の 16 進数のリストです．) これを打ち込みます．リストは，00_{16} 番地に 17_{16}，01_{16} 番地に $2E_{16}$，...，$0F_{16}$ 番地に 01_{16}，10_{16} 番地に $A4_{16}$，... を打ち込むことを表わしています．

打ち込み手順は次の通りです．操作方法は，4.2.2 を参照してください．

1. ロータリースイッチを MM の位置に合わせます．
2. アドレスランプに 00_{16}(2 進数で $0000\ 0000_2$) をセットします．
3. 16 進数を順に打ち込みます．
   最初は，17_{16} ですから，データスイッチに $0001\ 0111_2$ をセットして WRITE スイッチを押します．次は，$2E_{16}$ ですから，今度はデータスイッチに $0010\ 1110_2$ をセットして WRITE スイッチを押します．以降，同様にリストの最後まで打ち込みます．
4. 最後まで打ち込んだら，最初からもう一度データを確認します．
   アドレスランプに 00_{16} をセットしデータを確認します．合っていたら，INCA スイッチを押して次の番地に進みデータを確認します．間違っていた場合は，正しい値をデータスイッチにセットして WRITE スイッチを押します．

A.2　プログラムの実行

このプログラムの実行開始番地は 00_{16} です．PC に 00_{16} をセットし，プログラムの実行を開始します．手順は次の通りです．

1. ロータリースイッチを PC の位置に合わせます．
2. データスイッチに $0000\ \ 0000_2$ をセットし WRITE スイッチを押します．
3. BREAK スイッチ，STEP スイッチが下に倒れていることを確認します．
   (下に倒れていなかったら，下に倒します．)
4. RUN ボタンを押し，プログラムの実行を開始します．

A.3　残念ですが...

苦労して打ち込んだプログラムは保存しておきたいのですが，TeC にはその機能がありません．電源を切るとプログラムが消えてしまいます．残念ですが，毎回，打ち込む必要があります．

A.4　曲データの変更

打ち込んだ 16 進数のリストの構成は，00_{16} 番地から $2F_{16}$ 番地までにプログラム，30_{16} 番地から 65_{16} 番地までに「ドレミの歌」(途中まで) の音楽データ，66_{16} 番地に音楽データの終了を表す 00_{16} となっています (図 A.1 参照)．音楽データの部分を他の曲のデータに置き換えると別の曲を鳴らすことができます．

音楽データは 2 バイトで一つの音を表すようになっています．四分音符の各高さの音のデータは下表の通りです．1 バイト目が音の長さを調整するパラメータ，2 バイト目が音の高さを制御するパラメータです．1 バイト目の値は音の高さ毎に意味が変化します．

例えば，30_{16} 番地と 31_{16} 番地の $C5_{16}$ と $E8_{16}$ は，付点四分音符のドの音を表わしています．四分音符のドのデータは 83_{16} と $E8_{16}$ ですので，付点四分音符になるように 1 バイト目の値を 1.5 倍にして，このデータができました．



図 A.1: 電子オルゴールプログラムの 16 進数リスト

FF_{16} より長い値は指定できませんので，指定できない長い音は同じデータの繰り返しで表現して下さい．(休符は表現できません．ˆˆ;;;) 少々(かなり?)，制限が多いですがプログラムの打ち込みの練習になりますので，是非，別の曲のデータを作って打ち込んでみて下さい．

電子オルゴールデータの意味

音	長さ	高さ
ド	83	E8
ド#	8B	DD
レ	93	D1
レ#	9C	C5
ミ	A5	BA
ファ	AF	AF
ファ#	B9	A6
ソ	C4	9C
ソ#	D0	93
ラ	DC	8B
ラ#	E9	83
シ	F7	7C
ド	FF	75

付 録B　TeCクロス開発環境

B.1　始めに

TeC クロス開発環境は，教育用コンピュータ TeC(Tokuyama kousen Educational Computer) の機械語プログラムを開発するためのパソコン上で動作するプログラム群である．クロス開発環境は，クロスアセンブラ tasm(tasm7:TeC7 用，tasm6:TeC6 用)，ダウンロードプログラム tsend(tsend7:TeC7 用，tsend6:TeC6 用) よりなる．

tasm は，ニーモニックで記述されたソースプログラムを解釈し，アセンブルリストと機械語をファイルに出力する. 出力された機械語は tsend プログラムにより，シリアル通信を用いて TeC にダウンロードする．tasm と tsend を使用することにより，TeC のプログラムをパソコン上でクロス開発することができる．その様子を図に示す．



処理の流れ

B.2　アセンブルコマンド

tasm は次の形式のコマンド行から起動され，アセンブラソースプログラムをアセンブルリスト (拡張子 .lst のファイルに出力) と機械語 (拡張子 .bin のファイルに出力) に変換 (アセンブル) する．なお，アセンブラソースプログラムは，TeC7 では拡張子 “.t7” のファイルに，TeC6 では拡張子 “.t6” のファイルに格納する．

`tasm7 [-l nn] [-w nn]` *source_file*
(`tasm6 [-l nn] [-w nn]` *source_file*)

source_file は B.4 章で説明する文法で記述されたアセンブラソースファイルの名前である．ソースファイル名の拡張子は “.t7”(“.t6”) である必要がある．オプションの意味は次の通りである．

オプション	意味
`-l nn`	アセンブルリストの 1 ページを nn 行にする．
`-w nn`	アセンブルリストの 1 ページを nn 桁にする．

B.3 転送コマンド

tsend は tasm が出力した機械語プログラムをシリアル通信を用いて TeC に転送するコマンドである．次の形式のコマンド行から起動する．

`tsend7` *bin_file* *[com_port]*
(`tsend6` *bin_file* *[com_port]*)

bin_file は tasm が出力した機械語プログラムファイルの名前である．*[com_port]* は，TeC と接続されたシリアルポートを表すデバイスファイルの名前である．tsend を起動すると画面に「TeC を受信状態にする」ように指示が表示されるので，TeC の E0H 番地に格納されている IPL プログラム (B.8 IPL プログラム 参照) を起動する．

B.4 アセンブラの文法

以下では tasm の入力となるソースプログラムの文法について説明する．

B.4.1 行フォーマット

ソースファイルは，行の連なりからなるテキストファイルである．なお，tasm のソースファイルでは文字定数・文字列中を除き英字大文字と小文字は区別されない．空白はトークンの前後にいくつでも挿入することができる．空白は，スペース (文字コード 20H) とタブ (文字コード 09H) の 2 種類である．但し，行頭に空白を置いた場合は「ラベルが存在しない」ことを表す．

行＝ [ラベル][命令とオペランド][; コメント]

B.4.2 ラベル

行の最初の文字が英字の場合はその行にラベルがあるものとみなされる．ラベルは英字で始まりその後に任意の長さの英数字が続く文字列である．ラベルが無い行の場合は行を空白文字で始める．ラベルの長さに制限はないが，アセンブルリストの見栄えから 7 文字以内を推奨する．EQU 命令のラベルを除き，その行のアドレスがラベルの値になる．

B.4.3 疑似命令

以下では，tasm が処理できる 4 種類の疑似命令について説明する．

EQU 命令

EQU 命令はラベルを定義するために使用する疑似命令である．この命令にはラベルが必ず必要である．命令行のラベルが式値で定義される．式では，ラベルの前方参照はできない．EQU 命令の書式は次の通りである．

EQU 命令 = *EQU* 式

EQU 命令の使用例

ラベル	命令	オペランド
START	EQU	03H
SIZE	EQU	100
END	EQU	START+SIZE

ORG 命令

ORG 命令は機械語を生成するアドレスを変更する疑似命令である．機械語 (データも含む) が生成される全ての行より前で使用された場合は，単に次の命令のアドレスを変更する．機械語が生成される行の途中で使用された場合は，指定アドレスに達するまで，ゼロで初期化された領域を生成する．現在のアドレスより前のアドレスを指定することはできない．ORG 命令の式では，ラベルの前方参照はできない．ORG 命令の書式は次の通りである．

$$ORG\ 命令 = ORG \quad 式$$

ORG 命令の使用例

ラベル	命令	オペランド
START	ORG	03H

DS 命令

DS 命令は領域を定義するために使用する疑似命令である．領域の値はゼロで初期化される．DS 命令の式では，ラベルの前方参照はできない．DS 命令の書式は次の通りである．

$$DS\ 命令 = DS \quad 式$$

DS 命令の使用例

ラベル	命令	オペランド
WORK	DS	10

DC 命令

DC 命令はメモリ上に任意の定数を出力するために使用する疑似命令である．DC 命令の書式は次の通りである．

$$DC\ 命令 = DC \quad \left\{ \begin{array}{c} 式 \\ 文字列 \end{array} \right\} [, \left\{ \begin{array}{c} 式 \\ 文字列 \end{array} \right\}, \cdots, \left\{ \begin{array}{c} 式 \\ 文字列 \end{array} \right\}]$$

DC 命令の使用例

ラベル	命令	オペランド
DATA	DC	1,2,3,4
	DC	1+1
CHA	DC	'a'
STR	DC	"abc"

B.4.4 機械語命令

以下では，tasm が処理できる 6 グループの機械語命令について説明する．

機械語命令 1

オペランドの無い 1 バイトの機械語命令である．下表の命令がこのグループに属する．

機械語命令 1 = 命令

機械語命令 *1* に属す命令	
命令	意味
NO	何もしない
PUSHF	フラグをスタックに退避
POPF	フラグをスタックから復旧
EI	割り込み許可
DI	割り込み禁止
RET	サブルーチンから復帰
RETI	割り込みから復帰
HALT	プログラム終了

機械語命令 *1* の使用例		
ラベル	命令	オペランド
END	HALT	

機械語命令 2

レジスタ指定付きの 1 バイトの機械語命令である．レジスタとして．G0，G1，G2，SP が使用可能である．下表の命令がこのグループに属する．

機械語命令 2 = 命令　レジスタ

機械語命令 *2* に属す命令	
命令	意味
SHLA	左算術シフト
SHLL	左論理シフト
SHRA	右算術シフト
SHRL	右論理シフト
PUSH	スタックにレジスタを退避
POP	スタックからレジスタを復旧

機械語命令 *2* の使用例		
ラベル	命令	オペランド
L1	SHRA	SP
	PUSH	G0
	POP	G1

機械語命令 3

オペランドにレジスタと式を書く 2 バイトの機械語命令である．レジスタとして．G0，G1，G2，SP が使用可能である．下表の命令がこのグループに属する．

機械語命令 3 = 命令　レジスタ，式

機械語命令 *3* に属す命令	
命令	意味
IN	入出力ポートから読み込む
OUT	入出力ポートに書き込む

機械語命令 *3* の使用例		
ラベル	命令	オペランド
SIO	EQU	03H
	IN	G0,SIO

機械語命令 4

レジスタとメモリをオペランドに指定できる命令で，メモリ指定にイミディエイトモードが使用できるグループである．レジスタとして．G0，G1，G2，SP が使用可能である．インデクスレジスタとして．G1，G2 が使用可能である．下表の命令がこのグループに属する．

機械語命令 4 = 命令 $\left\{ \begin{array}{c} \text{レジスタ，式　[，インデクスレジスタ]} \\ \text{レジスタ，\#式} \end{array} \right\}$

機械語命令 *4* に属す命令	
命令	意味
LD	レジスタにメモリオペランド値をロード
ADD	レジスタとメモリオペランド値の和
SUB	レジスタとメモリオペランド値の差
CMP	レジスタとメモリオペランド値を比較
AND	レジスタとメモリオペランド値の論理積
OR	レジスタとメモリオペランド値の論理和
XOR	レジスタとメモリオペランド値の排他的論理和

機械語命令 *4* の使用例		
ラベル	命令	オペランド
L1	LD	G0,DATA
	LD	G1,#0
	SUB	G0,DATA,G1
	ADD	G1,#1

機械語命令 5

機械語命令 4 と，ほぼ，同様であるが，イミディエイトモードが使用できない機械語命令のグループである．レジスタとして．G0，G1，G2，SP が使用可能である．インデクスレジスタとして．G1，G2 が使用可能である．下表のように ST 命令だけがこのグループに属する．

機械語命令 5 = 命令　レジスタ，式　[，インデクスレジスタ]

機械語命令 *5* に属す命令	
命令	意味
ST	レジスタの値をメモリオペランドに格納

機械語命令 *5* の使用例		
ラベル	命令	オペランド
	ST	G0,WORK
	ST	G0,WORK,G1

機械語命令 6

メモリオペランドだけの機械語命令グループである．下表のようにジャンプ・コール命令がこのグループに属する．インデクスレジスタとして．G1，G2 が使用可能である．

機械語命令 6 = 命令　式　[，インデクスレジスタ]

機械語命令 *6* に属す命令	
命令	意味
JMP	必ずジャンプ
JZ	Z フラグが'1' ならジャンプ
JC	C フラグに'1' ならジャンプ
JM	S フラグに'1' ならジャンプ
JNZ	Z フラグが'0' ならジャンプ
JNC	C フラグに'0' ならジャンプ
JNM	S フラグに'0' ならジャンプ
CALL	サブルーチン呼び出し

(JNZ,JNC,JNM 命令は TeC6 では使用できない．)

機械語命令 *6* の使用例		
ラベル	命令	オペランド
	JMP	LOOP
	CALL	SUB1

B.5 アセンブラの文法まとめ

以下に文法を BNF 風にまとめる.

```
＜プログラム＞   ::=  { ＜行＞ } EOF
＜行＞           ::=  ＜命令行＞ | ＜空行＞ | ＜コメント行＞
＜命令行＞       ::=  ＜ラベル欄＞ [＜命令欄＞] [＜コメント＞] ＜改行＞
＜空行＞         ::=  ＜改行＞
＜コメント行＞   ::=  ＜コメント＞ ＜改行＞
＜ラベル欄＞     ::=  ＜ラベル＞ | ＜空白＞
＜命令欄＞       ::=  ＜命令記述１＞ | ＜命令記述２＞ | ＜命令記述３＞
                      | ＜命令記述４＞ | ＜命令記述５＞ | ＜命令記述６＞
                      | ＜命令記述７＞ | ＜命令記述８＞
＜命令記述１＞   ::=  ＜命令１＞
＜命令記述２＞   ::=  ＜命令２＞ ＜レジスタ＞
＜命令記述３＞   ::=  ＜命令３＞ ＜レジスタ＞, ＜式＞
＜命令記述４＞   ::=  ＜命令４＞ ＜レジスタ＞, ＜アドレス１＞
＜命令記述５＞   ::=  ＜命令５＞ ＜レジスタ＞, ＜アドレス２＞
＜命令記述６＞   ::=  ＜命令６＞ ＜アドレス２＞
＜命令記述７＞   ::=  ＜命令７＞ ＜式＞
＜命令記述８＞   ::=  ＜命令８＞ ＜値列＞
＜アドレス１＞   ::=  # ＜式＞ | ＜式＞ [, ＜インデクス＞]
＜アドレス２＞   ::=  ＜式＞ [, ＜インデクス＞]
＜値列＞         ::=  ＜拡張式＞ { , ＜拡張式＞ }
＜拡張式＞       ::=  ＜式＞ | ＜文字列＞
＜式＞           ::=  ＜項＞ |＜項＞ + ＜式＞ | ＜項＞ − ＜式＞
＜項＞           ::=  ＜因子１＞ | ＜因子１＞ * ＜項＞ | ＜因子１＞ ／ ＜項＞
＜因子１＞       ::=  ＜因子＞ | +＜因子＞ | −＜因子＞
＜因子＞         ::=  ＜数値＞ | ＜ラベル＞ | ( ＜式＞ )
＜コメント＞     ::=  ; { ＜文字１＞ }
＜数値＞         ::=  ＜１０進数＞ | ＜１６進数＞ | ＜文字定数＞
＜１０進数＞     ::=  ＜数字＞ { ＜数字＞ }
＜１６進数＞     ::=  ＜数字＞ { ＜１６進数字＞ } H
＜文字定数＞     ::=  '＜文字２＞'
＜文字列＞       ::=  " { ＜文字３＞ } "
＜ラベル＞       ::=  ＜英字＞ { ＜英数字＞ }
＜英数字＞       ::=  ＜英字＞ | ＜数字＞
＜文字１＞       ::=  ＜改行＞以外の文字
＜文字２＞       ::=  ＜改行＞, ' 以外の文字
＜文字３＞       ::=  ＜改行＞, " 以外の文字
＜１６進数字＞   ::=  0 | 1 | 2 | 3 | 4 | 5 | 6 | 7 | 8 | 9
                      | A | B | C | D | E | F
＜英字＞         ::=  A | B | C | D | E | F | G | H | I | J
                      | K | L | M | N | O | P | Q | R | S | T
                      | U | V | W | X | Y | Z | _
＜数字＞         ::=  0 | 1 | 2 | 3 | 4 | 5 | 6 | 7 | 8 | 9
＜命令１＞       ::=  NO | PUSHF | POPF | EI | DI | RET |RETI | HALT
＜命令２＞       ::=  SHLA | SHLL | SHRA | SHRL | PUSH | POP
＜命令３＞       ::=  IN | OUT
＜命令４＞       ::=  LD | ADD | SUB | CMP | AND | OR |XOR
＜命令５＞       ::=  ST
＜命令６＞       ::=  JMP | JZ | JC | JM | JNZ | JNC | JNM | CALL
＜命令７＞       ::=  EQU | DS | ORG
＜命令８＞       ::=  DC
＜インデクス＞   ::=  G1 | G2
＜レジスタ＞     ::=  G0 | G1 | G2 | SP
＜改行＞         ::=  '\n'
＜空白＞         ::=  ' ' |'\t'
```

B.6 ローマ字で名前を SIO に出力するプログラムの例

```
ADR  CODE          Label   Instruction             Comment              Page(1)

02                 SIOD    EQU     02H
03                 SIOS    EQU     03H
00
00  17 00          START   LD      G1,#0
02  19 15          L0      LD      G2,DATA,G1
04  5B 00                  CMP     G2,#0
06  A4 14                  JZ      END
08  C0 03          L1      IN      G0,SIOS
0A  63 80                  AND     G0,#80H
0C  A4 08                  JZ      L1
0E  CB 02                  OUT     G2,SIOD
10  37 01                  ADD     G1,#1
12  A0 02                  JMP     L0
14  FF             END     HALT
15
15  73 69 67 65    DATA    DC      "sigemura"
19  6D 75 72 61
1D  0D 0A                  DC      0DH,0AH
1F  00                     DC      0               ; 終わりの印
```

B.7 アセンブル実行例

TeC7 を用い，ローマ字で名前を SIO に出力するプログラムをアセンブル・実行した例を示す.

```
$ tasm7 sigemura.t7                              <--- アセンブル
アセンブル成功
結果は [sigemura.lst] と [sigemura.bin] に格納しました。

... 省略...

$ tsend7 sigemura.bin                            <--- ダウンロード
TeC7 を受信状態にして Enter キーを押して下さい。
[00][20][17][00][19][15][5b][00][a4][14][c0][03][63][80][a4][08][cb][02][37]
[01][a0][02][ff][73][69][67][65][6d][75][72][61][0d][0a][00]
$ screen /dev/tty.usbserial-xxxxxxxxx            <--- 通信ソフト起動
sigemura                                         <--- 実行結果
^A^\                                             <--- 通信ソフトの終了
[EOT]
```

B.8 IPL プログラム

TeC の ROM に格納されている IPL プログラムのアセンブルリストを参考のため以下に示す．TeC のメモリ空間は E0H〜FFH 番地が ROM になっており，以下の IPL プログラムが予め格納されている．この IPL プログラムは，パソコン上で tasm を使用してクロス開発された機械語プログラムをシリアル入出力から受信してメモリに格納する．

```
ADR  CODE          Label   Instruction             Comment              Page(1)

02                  SIOD    EQU     02H
03                  SIOS    EQU     03H
E0                          ORG     0E0H
E0   1F DC          IPL     LD      SP,#0DCH        ; 割込みベクタの直前がスタック
E2   B0 F6                  CALL    READ            ; ロードアドレスを入力
E4   D0                     PUSH    G0
E5   D6                     POP     G1
E6   B0 F6                  CALL    READ            ; 長さを入力
E8   D0                     PUSH    G0
E9   DA                     POP     G2
EA   A4 FF          LOOP    JZ      STOP
EC   B0 F6                  CALL    READ
EE   21 00                  ST      G0,0,G1
F0   37 01                  ADD     G1,#1
F2   4B 01                  SUB     G2,#1
F4   A0 EA                  JMP     LOOP
F6                  ;
F6   C0 03          READ    IN      G0,SIOS
F8   63 40                  AND     G0,#40H
FA   A4 F6                  JZ      READ
FC   C0 02                  IN      G0,SIOD
FE   EC                     RET
FF                  ;
FF   FF             STOP    HALT                    ; PC がゼロになって止まる
```

(TeC6 では CALL の機械語が BC になる．)

B.9 機械語プログラムファイル形式

tasm が出力する機械語プログラムの形式は次の通りである．



付 録C　命令表

次のページに TeC7 の機械語命令表，その次のページに TeC6 の機械語命令表を掲載します．これらはケース蓋の裏に張り付けてあるものと同じ内容です．

TEC7命令表

Ver. 3.0 (2012.4.6)

※ステート数：イミディエイト／ダイレクト／インデクスド
(ジャンプ命令では、条件不成立／ダイレクト／インデクスド)

ニーモニック	命令	第１バイト OP	第１バイト GRXR	第２バイト	フラグ変化	ステート数※	動作
NO	No Opration	0000	00 00	————	×	3	何もしない
LD	Load	0001	GR XR	aaaa aaaa	×	5/7/7	GR <- [EA]
ST	Store	0010	GR XR	aaaa aaaa	×	-/7/7	[EA] <- GR
ADD	Add	0011	GR XR	aaaa aaaa	○	5/7/7	GR <- GR + [EA]
SUB	Subtract	0100	GR XR	aaaa aaaa	○	5/7/7	GR <- GR - [EA]
CMP	Compare	0101	GR XR	aaaa aaaa	○	5/7/7	GR - [EA]
AND	Logical And	0110	GR XR	aaaa aaaa	○	5/7/7	GR <- GR & [EA]
OR	Logical Or	0111	GR XR	aaaa aaaa	○	5/7/7	GR <- GR | [EA]
XOR	Logical Xor	1000	GR XR	aaaa aaaa	○	5/7/7	GR <- GR ^ [EA]
SHLA	Shift Left Arithmetic	1001	GR 00	————	○	4	GR <- GR << 1
SHLL	Shift Left Logical	1001	GR 01	————	○	4	GR <- GR << 1
SHRA	Shift Right Arithmetic	1001	GR 10	————	○	4	GR <- GR >> 1
SHRL	Shift Right Logical	1001	GR 11	————	○	4	GR <- GR >> 1
JMP	Jump	1010	00 XR	aaaa aaaa	×	-/5/6	PC <- EA
JZ	Jump on Zero	1010	01 XR	aaaa aaaa	×	4/5/6	if Zero PC <- EA
JC	Jump on Carry	1010	10 XR	aaaa aaaa	×	4/5/6	if Carry PC <- EA
JM	Jump on Minus	1010	11 XR	aaaa aaaa	×	4/5/6	if Sign PC <- EA
CALL	Call subroutine	1011	00 XR	aaaa aaaa	×	-/6/7	[--SP]<-PC, PC<-EA
JNZ	Jump on Not Zero	1011	01 XR	aaaa aaaa	×	4/5/6	if !Zero PC <- EA
JNC	Jump on Not Carry	1011	10 XR	aaaa aaaa	×	4/5/6	if !Carry PC <- EA
JNM	Jump on Not Minus	1011	11 XR	aaaa aaaa	×	4/5/6	if !Sign PC <- EA
IN	Input	1100	GR 00	0000 pppp	×	8	GR <- IO[P]
OUT	Output	1100	GR 11	0000 pppp	×	8	IO[P] <- GR
PUSH	Push Register	1101	GR 00	————	×	6	[--SP] <- GR
PUSHF	Push Flag	1101	11 01	————	×	6	[--SP] <- FLAG
POP	Pop Register	1101	GR 10	————	×	6	GR <- [SP++]
POPF	Pop Flag	1101	11 11	————	○	6	FLAG <- [SP++]
EI	Enable Interrupt	1110	00 00	————	×	4	割り込み許可
DI	Disable Interrupt	1110	00 11	————	×	4	割り込み禁止
RET	Return from subroutine	1110	11 00	————	×	6	PC <- [SP++]
RETI	Return from Interrupt	1110	11 11	————	×	6	PC <- [SP++], STI
HALT	Halt	1111	11 11	————	×	4	停止

GR	意味
00	G0
01	G1
10	G2
11	SP

XR	意味
00	ダイレクトモード
01	G1インデクスドモード
10	G2インデクスドモード
11	イミディエイトモード

メモリマップ	
Addr	内容
00 | DB	RAM
DC	Tmr 割り込みベクタ
DD	SIO 受信割り込みベクタ
DE	SIO 送信割り込みベクタ
DF	Console 割り込みベクタ
E0 | FF	ROM(IPL)

I/Oマップ	
Addr	Read/Write
0	Data-Sw/b0:Beep
1	Data-Sw/b0:Speaker
2	SIO-Data/SIO-Data
3	b7:Tx Ready/b7:Tx STI b6:Rx Ready/b6:Rx STI
4	TMR現在値/TMR周期
5	b7:TMR Poll / b7:TMR STI b0:TMR Enb
6	空き/b0:Console STI
7	Input/Output
8	ADC CH0/空き
9	ADC CH1/空き
A	ADC CH2/空き
B	ADC CH3/空き
.... F	空き/空き



PIO : Parallel Input Output
SIO : Serial Input Output
TMR : Timer
TMR周期 : 75=1s
STI : Set Interrupt

図 C.1: TeC7 命令表

TEC6命令表 Ver. 2.3 (2006.3.25)

※ステート数：イミディエイト／ダイレクト／インデクスド
(ジャンプ命令では、条件不成立／ダイレクト／インデクスド)

ニーモニック	命令	第１バイト OP	第１バイト GRXR	第２バイト	フラグ変化	ステート数※	動作
NO	No Opration	0000	00 00	————	×	3	何もしない
LD	Load	0001	GR XR	aaaa aaaa	×	5/7/7	GR <- [EA]
ST	Store	0010	GR XR	aaaa aaaa	×	-/7/7	[EA] <- GR
ADD	Add	0011	GR XR	aaaa aaaa	○	5/7/7	GR <- GR + [EA]
SUB	Subtract	0100	GR XR	aaaa aaaa	○	5/7/7	GR <- GR - [EA]
CMP	Compare	0101	GR XR	aaaa aaaa	○	5/7/7	GR - [EA]
AND	Logical And	0110	GR XR	aaaa aaaa	○	5/7/7	GR <- GR & [EA]
OR	Logical Or	0111	GR XR	aaaa aaaa	○	5/7/7	GR <- GR | [EA]
XOR	Logical Xor	1000	GR XR	aaaa aaaa	○	5/7/7	GR <- GR ^ [EA]
SHLA	Shift Left Arithmetic	1001	GR 00	————	○	4	GR <- GR << 1
SHLL	Shift Left Logical	1001	GR 01	————	○	4	GR <- GR << 1
SHRA	Shift Right Arithmetic	1001	GR 10	————	○	4	GR <- GR >> 1
SHRL	Shift Right Logical	1001	GR 11	————	○	4	GR <- GR >>> 1
JMP	Jump	1010	00 XR	aaaa aaaa	×	-/5/6	PC <- EA
JZ	Jump on Zero	1010	01 XR	aaaa aaaa	×	4/5/6	if Zero PC <- EA
JC	Jump on Carry	1010	10 XR	aaaa aaaa	×	4/5/6	if Carry PC <- EA
JM	Jump on Minus	1010	11 XR	aaaa aaaa	×	4/5/6	if Sign PC <- EA
CALL	Call subroutine	1011	11 XR	aaaa aaaa	×	-/6/7	[--SP]<-PC, PC<-EA
IN	Input	1100	GR 00	0000 pppp	×	8	GR <- IO[P]
OUT	Output	1100	GR 11	0000 pppp	×	8	IO[P] <- GR
PUSH	Push Register	1101	GR 00	————	×	6	[--SP] <- GR
PUSHF	Push Flag	1101	11 01	————	×	6	[--SP] <- FLAG
POP	Pop Register	1101	GR 10	————	×	6	GR <- [SP++]
POPF	Pop Flag	1101	11 11	————	○	6	FLAG <- [SP++]
EI	Enable Interrupt	1110	00 00	————	×	4	割り込み許可
DI	Disable Interrupt	1110	00 11	————	×	4	割り込み禁止
RET	Return from subroutine	1110	11 00	————	×	6	PC <- [SP++]
RETI	Return from Interrupt	1110	11 11	————	×	6	PC <- [SP++], EI
HALT	Halt	1111	11 11	————	×	4	停止

GR	意味
00	G0
01	G1
10	G2
11	SP

XR	意味
00	ダイレクトモード
01	G1インデクスドモード
10	G2インデクスドモード
11	イミディエイトモード

メモリマップ	
Addr	内容
00 | DB	RAM
DC	INT0 割り込みベクタ
DD	INT1 割り込みベクタ
DE	INT2 割り込みベクタ
DF	INT3 割り込みベクタ
E0 | FF	ROM(IPL)

INT1 : SIO 受信割り込み
INT2 : SIO 送信割り込み

I/Oマップ	
Addr	Read/Write
0	Data-Sw/b0:Beep
1	Data-Sw/b0:Speaker
2	SIO-Data/SIO-Data
3	b7:Tx Ready, b7:Tx STI b6:Rx Ready, b6:Rx STI
4	空き/空き
....	....
F	空き/空き

図 C.2: TeC6 命令表

付 録D　参考資料

TeC をもっと活用してもらうために，拡張ボード，TeC の回路図，FPGA のピン配置表等を掲載します．次のページから以下の資料を掲載します．

TeC7 基板回路図

TeC7 本体のプリント基板回路図です．

TeC7 ピン配置表

TeC7 に搭載されている Xilinx Spartan-6 FPGA (XC6LX9-2TQG144C) のピンがどのように利用されているか示す一覧表です．



図 D.1: TeC7 基板回路図

TeC7a Spartan-6 ピン配置

V1.1 2012/10/05

ピン番号	VHDL名称	VHDL属性	注釈
1	ADC_REF(7)	OUT	ADC ラダー MSB
2	ADC_REF(6)	OUT	ADC ラダー
3			GND
4			VCCO
5	ADC_REF(5)	OUT	ADC ラダー
6	ADC_REF(4)	OUT	ADC ラダー
7	ADC_REF(3)	OUT	ADC ラダー
8	ADC_REF(2)	OUT	ADC ラダー
9	ADC_REF(1)	OUT	ADC ラダー
10	ADC_REF(0)	OUT	ADC ラダー LSB
11	JP_IN(0)	INOUT(PULLUP)	ジャンパー上段
12	JP_IN(1)	INOUT(PULLUP)	ジャンパー下段
13			GND
14	BREAK_SW	IN(PULLUP)	BREAK スイッチ
15	STEP_SW	IN(PULLUP)	STEP スイッチ
16	RUN_SW	IN(PULLUP)	RUN スイッチ
17	STOP_SW	IN(PULLUP)	STOP スイッチ
18			VCCO
19			VCCINT
20			VCCAUX
21	SETA_SW	IN(PULLUP)	SETA スイッチ
22	INCA_SW	IN(PULLUP)	INCA スイッチ
23	DECA_SW	IN(PULLUP)	DECA スイッチ
24	WRITE_SW	IN(PULLUP)	WRITE スイッチ
25			GND
26	DATA_SW(7)	IN(PULLUP)	D7 スイッチ
27	DATA_SW(6)	IN(PULLUP)	D6 スイッチ
28			VCCINT
29	DATA_SW(5)	IN(PULLUP)	D5 スイッチ
30	DATA_SW(4)	IN(PULLUP)	D4 スイッチ
31			VCCO
32	DATA_SW(3)	IN(PULLUP)	D3 スイッチ
33	DATA_SW(2)	IN(PULLUP)	D2 スイッチ
34	DATA_SW(1)	IN(PULLUP)	D1 スイッチ
35	DATA_SW(0)	IN(PULLUP)	D0 スイッチ
36			VCCAUX

ピン番号	VHDL名称	VHDL属性	注釈
37			PRGM
38			NC
39			INIT
40	LEFT_SW	IN(PULLUP)	← スイッチ
41	RIGHT_SW	IN(PULLUP)	→ スイッチ
42			VCCO
43	RUN_LED	OUT	RUN ランプ
44	Z_LED	OUT	Z ランプ
45	S_LED	OUT	S ランプ
46	C_LED	OUT	C ランプ
47	MM_LED	OUT	MM ランプ
48	PC_LED	OUT	PC ランプ
49			GND
50	SP_LED	OUT	SP ランプ
51	G2_LED	OUT	G2 ランプ
52			VCCINT
53			VCCAUX
54			GND
55	CLK_IN	IN	9.8304MHz
56	G1_LED	OUT	G1 ランプ
57	G0_LED	OUT	G0 ランプ
58	DATA_LED(0)	OUT	D0 ランプ
59	DATA_LED(1)	OUT	D1 ランプ
60			M1(GND)
61	DATA_LED(2)	OUT	D2 ランプ
62	DATA_LED(3)	OUT	D3 ランプ
63			VCCO
64			NC
65			DIN
66			NC
67			NC
68			GND
69			M0(VCC)
70			CCLK
71			DONE
72			NC

ピン番号	VHDL名称	VHDL属性	注釈
73			NC
74			NC
75			NC
76			VCCO
77			GND
78	DATA_LED(4)	OUT	D4 ランプ
79	DATA_LED(5)	OUT	D5 ランプ
80	DATA_LED(6)	OUT	D6 ランプ
81	DATA_LED(7)	OUT	D7 ランプ
82	SPK_OUT	OUT	スピーカ
83	ADDR_LED(0)	OUT	A0 ランプ
84	ADDR_LED(1)	OUT	A1 ランプ
85	ADDR_LED(2)	OUT	A2 ランプ
86			VCCO
87	ADDR_LED(3)	OUT	A3 ランプ
88	ADDR_LED(4)	OUT	A4 ランプ
89			VCCINT
90			VCCAUX
91			GND
92	ADDR_LED(5)	OUT	A5 ランプ
93	ADDR_LED(6)	OUT	A6 ランプ
94	ADDR_LED(7)	OUT	A7 ランプ
95	-	-	μSD 挿入検出
96			GND
97	-	-	μSD DAT1
98	SPI_DIN	IN	SPI データ入力
99	SPI_SCLK	OUT	SPI クロック出力
100	SPI_DOUT	OUT	SPI データ出力
101	SPI_CS	OUT	SPI セレクト出力
102	-	-	μSD DAT2
103			VCCO
104	ACC_LED	OUT	μSD アクセスランプ
105	SIO_TXD	OUT	SIO 送信データ
106			TDO
107			TMS
108			GND

ピン番号	VHDL名称	VHDL属性	注釈
109			TCK
110			TDI
111	SIO_RXD	IN	SIO 受信データ
112	PS2_DAT	INOUT(PULLUP)	キーボードデータ
113			GND
114	PS2_CLK	INOUT	キーボードクロック
115	VGA_VSYNC	OUT	ディスプレイ垂直同期
116	VGA_HSYNC	OUT	ディスプレイ水平同期
117	VGA_BLUE	OUT	ディスプレイ青
118	VGA_GREEN	OUT	ディスプレイ緑
119	VGA_RED	OUT	ディスプレイ赤
120	RESET_SW	OUT	RESET スイッチ
121	EXT_OUT(7)	OUT	拡張ポート4番ピン
122			VCCO
123	EXT_OUT(6)	OUT	拡張ポート6番ピン
124	EXT_OUT(5)	OUT	拡張ポート8番ピン
125			VCCO
126	EXT_OUT(4)	OUT	拡張ポート10番ピン
127	EXT_OUT(3)	OUT	拡張ポート12番ピン
128			VCCINT
129			VCCAUX
130			GND
131	EXT_OUT(2)	OUT	拡張ポート14番ピン
132	EXT_OUT(1)	OUT	拡張ポート16番ピン
133	EXT_OUT(0)	OUT	拡張ポート18番ピン
134	EXT_IN(7)	IN(PULLUP)	拡張ポート3番ピン
135			VCCO
136			GND
137	EXT_IN(6)	IN(PULLUP)	拡張ポート5番ピン
138	EXT_IN(5)	IN(PULLUP)	拡張ポート7番ピン
139	EXT_IN(4)	IN(PULLUP)	拡張ポート9番ピン
140	EXT_IN(3)	IN(PULLUP)	コンパレータ OUT1
141	EXT_IN(2)	IN(PULLUP)	コンパレータ OUT2
142	EXT_IN(1)	IN(PULLUP)	コンパレータ OUT3
143	EXT_IN(0)	IN(PULLUP)	コンパレータ OUT4
144			HS(GND)

専用ピン、電源ピン、未配線ピン

図 D.2: TeC7 ピン配置図

TeC 教科書

発行年月　２０１５年１１月　Ver.3.2.2
発　　行　独立行政法人国立高等専門学校機構
徳山工業高等専門学校
情報電子工学科　重村哲至
〒 745-8585 山口県周南市学園台
sigemura@tokuyama.ac.jp